SYBEX

MORE USES FOR YOUR

TIMEX/SINCLAIR 1000: ASTRONOMY

ON YOUR COMPUTER

ERIC BURGESS • HOWARD J. BURGESS



More Uses for Your Timex/Sinclair 1000™

Astronomy On Your Computer

Eric Burgess, F.R.A.S.
Howard Burgess

SYBEX®

Berkeley • Paris • Düsseldorf

Cover photo courtesy of NASA
Inset cover photo by John Horvers
Book design and layout by Ingrid Owen

Library of Congress Card Number: 83-61386
ISBN 0-89588-112-8
Printed in the United States of America

10 9 8 7 6 5 4 3 2 1

Contents

Acknowledgements

The authors gratefully acknowledge comments made by readers and users of *Celestial BASIC,* and particularly those from some astronomical groups. Where possible, these suggestions have been included in these new programs for the T/S 1000 computer.

We also express our thanks to the staff at Sybex, Inc., and particularly to our editor, Barbara Gordon.

Eric and Howard Burgess
Sebastopol, California
May 1983

Introduction

With the programs in this book and a Timex/Sinclair 1000™ home computer, amateur and armchair astronomers and students are equipped with important aids to observing and understanding the configurations, motions, and apparitions of our Sun and Moon, the planets, and the stars.

Planetary positions, as well as the rising and setting times of all the planets, Sun, and Moon, can be quickly found for any locality in the world, at any time, on any date. Astronomical (star) times can be converted to local times and vice versa. Circles can be set on equatorially mounted telescopes, and true north can be determined with the aid of Polaris. You can find out when the next lunar eclipse will occur and how much of our satellite will be covered by Earth's shadow. You can learn when Mars will be in opposition next and how large it will appear in a telescope, or when Venus or Mercury are favorably placed for observation in the morning or evening sky. In the comfort of your own living room, you can see how the midnight sun appears on the oil fields of Alaska's North Slope. And if you are not up to steam in recognizing the constellations, use a tutorial to learn how to identify the major constellations and their brightest stars.

Educators, who are increasingly using computers at all levels in teaching, will find these programs useful to illustrate the dynamics of observational astronomy. In addition to the tutorial on recognizing the constellations, other tutorials aid in converting astronomical measurements and in taking photographs of the planets.

The book is divided into four major sections. Part 1 covers times, dates, and conversions among them, updating of coordinates, and using Polaris to determine latitude and true north. Part 2 concentrates on our Moon, and Part 3 concentrates on the planets. Part 4 provides general and tutorial programs. A Bibliography is provided to refer you to further information on astronomical computations.

Originally written for an Exidy Sorcerer™ machine, the programs were rewritten for an Apple II™ to make them available to a wider range of users, and they were published by Sybex in the book *Celestial BASIC* in 1982. Most of the important programs have been rewritten for this book in a form suitable to the BASIC of the Timex/Sinclair 1000 home computer (with 16K of memory). Thus these programs are now available to the many hundreds of thousands of users of the Sinclair ZX81™, the Timex/Sinclair 1000, and the new Sinclair Spectrum™. Readers are advised to purchase *Celestial BASIC* for background information on the subjects covered by the programs and for alternative ways of writing the programs.

The programs are written in the Sinclair BASIC language, and complex programming techniques have been avoided. This has been done to minimize complications of program structure, so that the programs are readily understandable to the beginner and adaptable to other computers. Again, the reader is advised to refer to *Celestial BASIC* for more details of how to use the programs in other BASIC languages.

Readers will notice when typing in the programs that some of the longer lines cannot be displayed as one line on the screen. This should not cause concern; the enter key should be pressed only when the line has been completed.

The graphics peculiar to the T/S 1000 generally are avoided in these programs, in order to allow the programs to be adapted more easily to the graphic displays of different computers. However, the COPY feature of the T/S 1000 has been incorporated into the program listings. This was done to allow those readers with a ZX81 or similar printer to copy the screen data to a printer. If you do not have a printer connected or do not want a printed copy of the screen data, you merely answer the computer's 'Want Copy' query with 'N'.

Most programs can be adapted relatively easily for use in the Southern Hemisphere, where constellations will need to be inverted. Several programs have the capability of offering Northern or Southern Hemisphere viewing without modification—SKYPT, for example. All programs have been carefully edited, run, and checked against actual astronomical observations over many years.

The programs are designed so that large memories are not required. In general, they require the 16K memory for the T/S

1000. They can be placed conveniently in one cassette when needed concurrently, or they can be selected by menus. The programs are available on cassette for the T/S 1000 and the Spectrum, on diskettes for the Apple II, and on cassettes and diskettes for the Sorcerer and some other computers. Information can be obtained by writing to the authors.

As mentioned in the introduction to the Bibliography, it is possible to add further steps within the structure of the programs to increase the accuracy of the computer's readout. The references cited in the Bibliography provide the information necessary to achieve higher orders of accuracy. (Larger random access memories may be required to incorporate these refinements.)

We are grateful to the many people who have derived the equations forming the basic algorithms of the computer programs, many of which have been used in their fundamental forms for a long time. Our intention is to provide a useful service to amateur and armchair astronomers, educators, and students by presenting the equations in a form that allows them to be manipulated on a small and inexpensive microprocessing system.

PART 1

TIME

The science of astronomy is an outgrowth of man's need to measure time. Early studies of the sky resulted in finding a relationship between the motions of the heavenly bodies and the recurrence of seasonal and other natural phenomena. Time was measured, calendars were derived, and religious functions were set from observations of the motions of the Sun, Moon, and stars. As astronomy progressed and greater precision of observations became possible, more precise systems of time measurement were needed.

The following programs provide easy access to and conversions between the various systems for measuring time. Because the celestial bodies are not fixed in space relative to Earth-centered coordinates, one program also provides a means to update the positions of the stars over the years.

Program 1: EASTR

Date of Easter Sunday for Any Year

In 1966 Thomas H. O'Beirne published an algorithm to calculate the date of Easter for any year between 1900 and 2099. The O'Beirne algorithm, which works for one century only in the form used in this book, divides the year by 19 and a multiple of the remainder by 19. Then it uses a multiple of that remainder for another division and further manipulates the remainders. This short program uses the O'Beirne algorithm (instructions 160 through 320). It quickly provides the date of Easter Sunday for any year between 1900 and 2099 (see Figure 1.1). However, even this algorithm needs to be corrected for certain years (1974, 1984, and 1994). The date given by the program has been adjusted by adding 7 days to the date for Easter 1974 and 1984. For Easter 1994 the date has been adjusted by adding 7 days, subtracting 31 days, and changing the month to April.

The listing for the EASTR program follows.



Figure 1.1: A display produced by the EASTR program

EASTR

```
 10 PRINT
 20 REM EASTR 4/15/83
 30 PRINT
 40 PRINT
 50 PRINT
 60 PRINT "THIS PROGRAM CALCULATES THE DATE"
 70 PRINT "OF EASTER SUNDAY FOR ANY YEAR"
 80 PRINT "    BETWEEN 1900 AND 2099"
 90 PRINT
100 PRINT
110 PRINT
120 PRINT "    BY ERIC BURGESS F.R.A.S."
130 PRINT
140 PRINT
150 PRINT
160 PRINT  "ALL RIGHT RESERVED BY"
170 PRINT "S AND T SOFTWARE SERVICES"
180 PAUSE 300
190 CLS
200 PRINT
210 PRINT
220 PRINT
230 PRINT
```

EASTR (continued)

```
240 PRINT "YEAR REQUIRED ";
250 INPUT Y$
260 LET Y= VAL (Y$)
270 IF Y<1900  OR Y>2099 THEN GOTO 290
280 GOTO 390
290 CLS
300 PRINT
310 PRINT
320 PRINT "INVALID ENTRY"
330 PAUSE 150
340 CLS
350 PRINT
360 PRINT
370 PRINT
380 GOTO 240
390 PRINT Y
400 LET N=Y-1900
410 LET A=N/19
420 LET A=19*(A- INT (A))
430 LET B= INT ((7*A+1)/19)
440 LET M=0
450 LET M=(11*A+4.00001-B)/29
460 LET X=M- INT (M)
470 IF X=1 THEN GOTO 500
480 IF X <> 1 THEN LET M=29*X
490 GOTO 510
500 LET M=0
510 LET Q= INT (N/4)
520 LET W=(N+Q+31-M)/7
530 LET W=7*(W- INT (W))
540 LET W= INT (W)
550 LET DE= INT (25-M-W)
560 CLS
570 PRINT
580 PRINT
590 PRINT
600 PRINT
610 PRINT
620 PRINT "EASTER SUNDAY ";Y;" IS"
630 PRINT
640 PRINT
650 PRINT
660 IF DE>0 THEN LET M$="APRIL"
670 IF DE<0 THEN LET M$="MARCH"
680 IF DE=0 THEN PRINT "MARCH 31"
690 IF DE=0 THEN GOTO 780
700 IF DE<-9 THEN LET DE=DE+9
710 IF DE<-9 THEN GOTO 700
720 IF DE<0 THEN LET D=31- ABS (DE)
730 IF DE>0 THEN LET D=DE
740 IF Y=1974 OR Y=1984 THEN LET D=D+7
750 IF Y=1994 THEN LET D=D+7-31
760 IF Y=1994 THEN LET M$="APRIL"
770 PAUSE 250
780 PRINT " ";M$;" ";D
790 PRINT
800 PRINT
```


EASTR (continued)

```
810 PRINT
820 PRINT
830 PRINT "WANT ANOTHER YEAR Y/N ";
840 INPUT A$
850 PRINT A$
860 IF A$="Y" THEN PAUSE 100
870 IF A$="Y" THEN GOTO 190
880 IF A$="N" THEN GOTO 920
890 PRINT
900 PRINT "INVALID REPLY"
910 GOTO 830
920 CLS
930 STOP
```

Program 2: TIMES

Time Conversion: Local Mean to Sidereal, or Sidereal to Local Mean

This program computes local sidereal time (LST) given local clock time (LMT), or vice versa. It applies worldwide; you can enter location parameters (time zone, town name, longitude) each time you run the program, or you can set up your local parameters when you key in the program (instructions 240 through 280) and change them when asked during a run, still leaving your own locality permanently in the program. This would be useful if you wish to compute time for some other locality, such as a vacation observing site. A typical display produced by this program is shown in Figure 2.1.

The program asks for your time zone and longitude and applies a correction to clock time to take into account your location within your time zone. You can convert between universal time (U.T.) and sidereal time at 0 longitude by entering time zone 0 and longitude 0 when asked. The program can be modified easily if you wish to accept longitude only, ignoring the time zone. Alternatively, you can have it ignore longitude and calculate times more approximately from the time zone input only. This also applies to other programs in this book where both time zone and longitude are shown as inputs.

Times are computed within a few seconds' accuracy, suitable for setting circles on equatorial mounts. You can use decimal hours or hours, minutes, and seconds for inputs. The program does not correct for daylight savings time; you must adjust inputs and outputs when this is in force at your locality or at the locality for which you are computing data.

The program is written for and tested on a T/S 1000. If you are using other computers you will have to make some substitutions to suit their particular BASIC instructions, as discussed in *Celestial BASIC*.

The listing of the TIMES program follows.


```
LST TIME(S) REQUESTED
DATE 1983 10 1
LOCATION SEBASTOPOL ZONE 8
FOR LONGITUDE 122.49
INTERVAL FOR TIME CALCS. 24 HRS

                LMT                 LST
 DEC.HR  HR/MI/SE  DEC.HR  HR/MI/SE
1  23.00  23/0/0    7.8294   7/49/46
2  23.00  23/0/0    7.8951   7/53/42
3  23.00  23/0/0    7.968    7/57/39
4  23.00  23/0/0    8.265    8/1/35
WANT COPY Y/N
```

Figure 2.1: This figure shows a display produced by the TIMES program, giving conversions for a series of three local mean times separated by 24 hours. Times are expressed in decimal hours and hours/minutes/seconds, irrespective of how you input the first time in the series.

TIMES

```
 10 REM TIMES 4/12/83
 20 LET FLG=0
 30 PRINT
 40 PRINT
 50 PRINT
 60 PRINT
 70 PRINT "       TIME CONVERSIONS"
 80 PRINT
 90 PRINT
100 PRINT "      A COMPUTER PROGRAM"
110 PRINT
120 PRINT "        FOR ASTRONOMERS"
130 PRINT
140 PRINT
150 PRINT "   BY ERIC BURGESS F.R.A.S."
160 PRINT
170 PRINT
180 PRINT "    ALL RIGHTS RESERVED BY"
190 PRINT
```

TIMES (continued)

```
200 PRINT "   S AND T SOFTWARE SERVICE"
210 PAUSE 200
220 CLS
230 REM SET INITIAL CONDITIONS
240 REM ZN=TIME ZONE
250 REM MINUS ZN IF EAST OF GREENWICH
260 REM L$=NAME OF YOUR LOCALITY
270 REM LO=LONGITUDE
280 LET ZN=8
290 LET L$="SEBASTOPOL"
300 LET Z$="8"
310 LET O$="122.49"
320 LET LO=122.49
330 PRINT
340 PRINT
350 PRINT "INITIAL CONDITION SET FOR"
360 PRINT L$
370 PRINT
380 PRINT
390 PRINT
400 PRINT "TIME ZONE ";
410 PRINT ZN
420 PRINT
430 PRINT "LONGITUDE ";
440 PRINT LO
450 PRINT
460 PRINT
470 PRINT "DO YOU WANT TO CHANGE"
480 PRINT "CONDITIONS? Y/N ";
490 INPUT A$
500 PRINT A$
510 IF A$="N" THEN GOTO 720
520 IF A$="Y" THEN GOTO 560
530 PAUSE 100
540 PRINT "INVALID REPLY"
550 GOTO 470
560 CLS
570 PRINT
580 PRINT
590 PRINT "LOCATION NAME ";
600 INPUT L$
610 PRINT L$
620 PRINT
630 PRINT "TIME ZONE ";
640 INPUT Z$
650 PRINT Z$
660 LET ZN= VAL (Z$)
670 PRINT
680 PRINT "LONGITUDE ";
690 INPUT O$
700 PRINT O$
710 LET LO= VAL (O$)
720 CLS
730 PRINT
740 PRINT
750 PRINT
```


TIMES (continued)

```
760 LET Z=ZN
770 IF FLG=1 THEN GOTO 1620
780 LET A$=""
790 LET C$=""
800 LET E$=""
810 PRINT "PROGRAM COMPUTES LOCAL MEAN"
820 PRINT "TIME (LMT) WHEN GIVEN LOCAL"
830 PRINT "SIDEREAL TIME (LST), OR LOCAL"
840 PRINT "SIDEREAL TIME WHEN GIVEN"
850 PRINT "LOCAL MEAN TIME"
860 PRINT
870 PRINT "YOU CAN REQUEST CALCULATIONS"
880 PRINT "FOR A NUMBER OF TIME INTERVALS"
890 PRINT
900 PRINT "TIME MAY BE ENTERED AND"
910 PRINT "DISPLAYED IN DECIMAL HOURS"
920 PRINT "OR HRS. MIN. SEC. USING"
930 PRINT "24-HOUR CLOCK"
940 PRINT
950 PRINT "TIME INTERVALS MUST BE IN"
960 PRINT "DECIMAL HOURS"
970 PRINT
980 PRINT "::::::::::::::::::::::::::::"
990 PRINT "PRESS ANY KEY"
1000 IF INKEY$ ="" THEN GOTO 1000
1010 CLS
1020 PRINT
1030 PRINT
1040 PRINT
1050 PRINT
1060 PRINT "DO YOU WANT TO CALCULATE LMT'
1070 PRINT TAB 22;"OR LST"
1080 PRINT
1090 PRINT "TYPE LMT OR LST"
1100 INPUT T$
1110 PRINT T$
1120 PRINT
1130 IF T$="LMT" THEN GOTO 1200
1140 IF T$="LST" THEN GOTO 1200
1150 PRINT "INVALID REPLY"
1160 PRINT
1170 GOTO 1060
1180 PRINT
1190 PRINT
1200 PRINT "ENTER THE DATE"
1210 PRINT
1220 PRINT "THE YEAR ";
1230 INPUT Y
1240 PRINT Y
1250 PRINT
1260 PRINT "THE MONTH ";
1270 INPUT M
1280 PRINT M
1290 IF M<1 OR M>12 THEN GOTO 1250
1300 PRINT
1310 PRINT "THE DAY ";
1320 INPUT D
```

TIMES (continued)

```
1330 PRINT D
1340 IF D<1 OR D>31 THEN GOTO 1300
1350 IF M=2 AND D>29 THEN GOTO 1300
1360 IF D>30 AND (M=4 OR M=6 OR M=9) THEN GOTO 1300
1370 IF M=11 AND D>30 THEN GOTO 1300
1380 PAUSE 100
1390 CLS
1400 PRINT
1410 PRINT
1420 PRINT "WANT TO SEE LOCATION"
1430 PRINT "CONDITIONS? Y/N ";
1440 INPUT B$
1450 PRINT B$
1460 IF B$="Y" THEN GOTO 1510
1470 IF B$="N" THEN GOTO 1550
1480 PRINT "INVALID REPLY"
1490 PRINT
1500 GOTO 1420
1510 PRINT "LOCATION CONDITIONS ARE..."
1520 PRINT "  LOCATION NAME.. ";L$
1530 PRINT "  TIME ZONE..... ";Z$
1540 PRINT "  LONGITUDE..... ";O$
1550 PRINT
1560 PRINT "DO YOU WANT TO CHANGE THEM? Y/N ";
1570 INPUT X$
1580 PRINT X$
1590 IF X$="N" THEN GOTO 1620
1600 IF X$="Y" THEN LET FLG=1
1610 GOTO 560
1620 LET FLG=0
1630 GOSUB 3060
1640 IF T$="LMT" THEN GOTO 1670
1650 IF T$="LST" THEN LET U$="LMT"
1660 GOTO 1680
1670 LET L$="LST"
1680 GOSUB 3420
1690 PRINT "SPECIFY NUMBER OF CALCULATIONS ";
1700 INPUT NO
1710 PRINT NO
1720 PRINT
1730 PRINT "AND TIME INTERVALS ";
1740 INPUT IN
1750 PRINT IN
1760 CLS
1770 REM CALCS FOR DAYS FROM EPOCH
1780 LET EP=722895
1790 GOSUB 3480
1800 LET DE=DG
1810 LET ND=DE-EP
1820 REM CALC AND PRINTS TIME TABLE
1830 PRINT T$;" ";
1840 PRINT "TIME(S) REQUESTED"
1850 LET TC=.065753
1860 PRINT "DATE ";Y;" ";M;" ";D
1870 PRINT "LOCATION ";L$;" ZONE ";ZN-LGC/15
1880 PRINT "FOR LONGITUDE ";LO
```


TIMES (continued)

```
1890 PRINT "INTERVAL FOR TIME CALCS. ";IN;" HRS"
1900 PRINT
1910 PRINT "-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-"
1920 PRINT "          ";U$;"          ";T$
1930 PRINT " DEC.HR HR/MI/SE DEC.HR HR/MI/SE"
1940 PRINT "-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-"
1950 FOR K=1 TO NO
1960 LET H1= INT (T1)
1970 LET M1= INT ((T1-H1)*100)
1980 LET S1= INT ((((T1-H1)*100)-M1)*100)
1990 GOSUB 2270
2000 IF INT (T2)>24 THEN LET T2=T2-24
2010 LET H2= INT (T2)
2020 LET M2= INT ((T2-H2)*100)
2030 LET S2= INT ((((T2-H2)*100)-M2)*100)
2040 LET HR2= INT (T2)
2050 LET MI2=(T2- INT (T2))*60
2060 LET SE2=(MI2- INT (MI2))*60
2070 LET SE2= INT (SE2)
2080 LET MI2= INT (MI2)
2090 GOSUB 3580
2100 LET H$= STR$ (H1)
2110 LET M$= STR$ (M1)
2120 LET S$= STR$ (S1)
2130 LET P$="."
2140 LET I$=H$+P$+M$+S$
2150 LET J$= STR$ (H2)
2160 LET Q$= STR$ (M2)
2170 LET V$= STR$ (S2)
2180 LET B$=J$+P$+Q$+V$
2190 PRINT K; TAB 2;I$; TAB 8;RR;"/";IM;"/";ES;
2200 PRINT TAB 16;B$; TAB 24;HR2;"/";MI2;"/";SE2
2210 LET T1=T1+IN
2220 IF T1<24 THEN GOTO 2250
2230 LET T1=T1-24
2240 LET ND=ND+1
2250 NEXT K
2260 GOTO 2590
2270 REM PARAMETER FOR TIME CALCS
2280 REM GMST AT EPOCH 1979 MARCH 22
2290 LET GC=11.927485
2300 REM DAILY RATE OF CHANGE OF GMST
2310 LET TC=.065711
2320 IF T$="LMT" THEN GOTO 2470
2330 REM CONVERSION TO LST
2340 LET T2=TC*ND+GC+(((ZN+T1)/24)*TC)+T1
2350 REM CONVERSION FOR NON MID TIME ZONE
2360 LET T2=T2+(.0656667*LGC)
2370 IF T2<24 THEN GOTO 2400
2380 LET T2=T2-24
2390 GOTO 2370
2400 IF T2<24 THEN GOTO 2430
2410 LET T2=T2-24
2420 GOTO 2400
2430 IF T2>0 THEN GOTO 2460
2440 LET T2=T2+24
2450 GOTO 2430
```

TIMES (continued)

```
2460 RETURN
2470 REM CONV TO LST
2480 LET T2=T1-(TC*ND+GC)-(ZN/24*TC)
2490 LET T2=T2-(.0656667*LGC)
2500 IF T2<0 THEN GOTO 2530
2510 IF T2>24 THEN GOTO 2550
2520 GOTO 2570
2530 LET T2=T2+24
2540 GOTO 2500
2550 LET T2=T2-24
2560 GOTO 2510
2570 LET T2=T2-(T2/24*TC)
2580 RETURN
2590 PRINT "-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-"
2600 PRINT "WANT COPY Y/N ';
2610 INPUT A$
2620 PRINT A$
2630 IF A$="N" THEN GOTO 2650
2640 COPY
2650 PRINT "DO YOU WANT MORE"
2660 PRINT "CALCULATIONS? Y/N ";
2670 INPUT A$
2680 PRINT A$
2690 CLS
2700 PRINT
2710 PRINT
2720 IF A$="N" THEN GOTO 3040
2730 LET H$= ""
2740 LET N$= ""
2750 LET S$= ""
2760 LET I$= ""
2770 LET J$= ""
2780 LET Q$= ""
2790 LET V$= ""
2800 LET B$= ""
2810 LET T1=0
2820 LET T2=0
2830 CLS
2840 PRINT
2850 PRINT
2860 PRINT "DO YOU STILL WANT TO"
2870 PRINT "CALCULATE ';T$
2880 INPUT A$
2890 IF A$="N" AND T$="LST" THEN LET T$="LMT"
2900 GOTO 2930
2910 IF A$="N" AND T$="LMT" THEN LET T$="LST"
2920 GOTO 2930
2930 CLS
2940 PRINT
2950 PRINT
2960 PRINT "DO YOU WANT TO CHANGE DATE ";
2970 INPUT A$
2980 PRINT A$
2990 IF A$="N" THEN LET ZN=Z
3000 GOTO 1410
3010 PRINT
3020 LET ZN=Z
3030 GOTO 1200
```


TIMES (continued)

```
3040 CLS
3050 STOP
3060 CLS
3070 PRINT
3080 PRINT
3090 PRINT
3100 PRINT "DO YOU WANT INPUT IN DEC. HRS(1)"
3110 PRINT "OR IN HRS MIN SEC (2)"
3120 PRINT
3130 PRINT
3140 PRINT TAB (5);" ";
3150 INPUT R$
3160 LET PT= VAL (R$)
3170 PRINT
3180 PRINT
3190 IF PT=1 THEN GOTO 3220
3200 IF PT=2 THEN GOTO 3270
3210 IF T$="LMT" THEN LET T$="LST"
3220 PRINT "   WHAT IS THE INPUT TIME"
3230 PRINT "   HR.XXXX (24-HR CLOCK) ";
3240 INPUT T1
3250 PRINT T1
3260 GOTO 3400
3270 PRINT " WHAT IS THE INPUT TIME"
3280 PRINT " HR,MIN,SEC (24-HR CLOCK)"
3290 PRINT
3300 PRINT "TYPE HR THEN MI THEN SE WHEN ASKED"
3310 PRINT
3320 PRINT "    ";
3330 INPUT HR
3340 PRINT HR;"/";
3350 INPUT MI
3360 PRINT MI;"/";
3370 INPUT SE
3380 PRINT SE
3390 LET T1=HR+MI/60+SE/3600
3400 PRINT
3410 RETURN
3420 LET LGC=(ZN*15)-LO
3430 IF LGC<0 THEN GOTO 3460
3440 LET ZN=ZN+ ABS (LGC/15)
3450 GOTO 3470
3460 LET ZN=ZN+LGC/15
3470 RETURN
3480 REM CALC GREGORIAN DAYS TO DATE REQUESTED
3490 IF M >= 3 THEN GOTO 3550
3500 REM CALC FOR JAN AND FEB
3510 LET DG=365*Y+D
3520 LET DG=DG+((M-1)*31)+ INT ((Y-1)/4)- INT ((.75)* INT
     ((Y-1)/100+1))
3530 RETURN
3540 REM CALCS FOR MAR THRU DEC
3550 LET DG=365*Y+D+((M-1)+31)- INT (M*.4+2.3)
3560 LET DG=DG+ INT (Y/4)- INT ((.75)* INT ((Y/100)+1))
3570 RETURN
3580 REM SUB FOR PRINTING HR MI SEC
3590 LET RR= INT (T1)
```

TIMES (continued)

```
3600 LET IM=(T1- INT (T1))*60
3610 LET ES=(IM- INT (IM))*60
3620 LET ES= INT (ES)
3630 LET IM= INT (IM)
3640 RETURN
```


Program 3: JULDY

Calendar Date to Julian Day

This program provides the Julian day for any Gregorian calendar day between 1100 and 2200 (see Figure 3.1). The program takes the Julian day for the beginning of the year 1900 and calculates Gregorian days from that date to the date selected. It then either adds or subtracts, as appropriate, the number of Gregorian days to or from the Julian day at the epoch 1900. The program can be modified easily to apply to other centuries also. This program can be merged with CDATE, its complement. See Program 4 for instructions.

The listing of the JULDY program follows.



Figure 3.1: A typical display generated by the JULDY program

JULDY

```
 10 PRINT
 20 REM JULDY 4/12/83
 30 PRINT
 40 PRINT TAB 5;"-----------------"
 50 PRINT TAB 5;"I  JULIAN DAY   I"
 60 PRINT TAB 5;"-----------------"
 70 PRINT
 80 PRINT TAB 5;"ASTRONOMY PROGRAM"
 90 PRINT
100 PRINT TAB 2;"BY ERIC BURGESS F.R.A.S."
110 PRINT
120 PRINT TAB 3;"ALL RIGHTS RESERVED BY"
130 PRINT TAB 2;"S AND T SOFTWARE SERVICE"
140 PAUSE 100
150 CLS
160 PRINT
170 PRINT
180 PRINT
190 PRINT
200 PRINT "TYPE YEAR ";
210 INPUT Y1
220 PRINT Y1
230 PRINT
```


JULDY (continued)

```
240 PRINT "TYPE MONTH ";
250 INPUT M1
260 IF M1<1 OR M1>12 THEN GOTO 230
270 PRINT M1
280 PRINT
290 PRINT "TYPE DAY ";
300 INPUT D1
310 IF D1<1 OR D1>31 THEN GOTO 280
320 IF M1=2 AND D1>29 THEN GOTO 280
330 PRINT D1
340 IF Y1<0 THEN GOTO 360
350 GOTO 370
360 LET Y1=Y1-100
370 LET Y=Y1-1900
380 LET M2=M1-1
390 IF M2=0 THEN LET DY=0
400 IF M2=1 THEN LET DY=31
410 IF M2=2 THEN LET DY=59
420 IF M2=3 THEN LET DY=90
430 IF M2=4 THEN LET DY=120
440 IF M2=5 THEN LET DY=151
450 IF M2=6 THEN LET DY=181
460 IF M2=7 THEN LET DY=212
470 IF M2=8 THEN LET DY=243
480 IF M2=9 THEN LET DY=273
490 IF M2=10 THEN LET DY=304
500 IF M2=11 THEN LET DY=334
510 LET D3=D1+DY
520 LET D4=365*Y+ INT (Y/4)
530 IF Y1>1999 THEN LET D4=D4-2
540 LET D5=15020+ INT (D4)+D3
550 PRINT
560 PRINT
570 LET JD=D5
580 IF Y/4- INT (Y/4)=0 AND M1<3 THEN LET JD=JD-1
590 LET JD=JD+2400000
600 IF Y1<1583 THEN LET JD=JD+(10- INT ((1583-Y1)/100))
610 IF Y1>1999 THEN LET JD=JD+3
620 PRINT "JULIAN DAY IS ";JD
630 PRINT
640 PRINT
650 PRINT "WANT ANOTHER DATE Y/N?"
660 INPUT Y$
670 IF Y$="N" THEN GOTO 700
680 CLS
690 GOTO 160
700 CLS
710 REM "END"
```

Program 4: CDATE

Julian Day to Calendar Date

This program complements the JULDY program. It provides a Gregorian calendar date for any Julian day between 1100 and 2000 (see Figure 4.1). Again, it is easily modified for application in other centuries. These two programs, JULDY and CDATE, can be merged into one if you wish, with a selection of one or the other at the beginning of the merged program. If you intend to do this, you should increase all the line numbers in CDATE by 1000 when you key them in, so that you can merge the two programs later. Then you must add an appropriate selection and branch routine at the beginning of JULDY.

The listing of the CDATE program follows.



***Figure 4.1**: This display, generated by the CDATE program, is the converse of the display in Figure 3 1.*

CDATE

```
 10 REM CDATE 4/12/83
 20 PRINT
 30 PRINT
 40 PRINT
 50 PRINT "-------------------------------"
 60 PRINT "I JULIAN DAY TO CALENDAR DATE I"
 70 PRINT "-------------------------------"
 80 PRINT
 90 PRINT
100 PRINT
110 PRINT AT 9,4;"AN ASTRONOMY PROGRAM"
120 PRINT AT 10,3;"BY ERIC BURGESS F.R.A.S."
130 PRINT AT 12,3;"ALL RIGHTS RESERVED BY"
140 PRINT AT 13,2;"S AND T SOFTWARE SERVICE"
150 PRINT
160 LET LY=0
170 PRINT
180 PAUSE 100
190 PRINT
200 PRINT AT 16,2;"THIS PROGRAM PROVIDES THE"
210 PRINT AT 17,2;"CALENDAR DATE FOR A JULIAN"
220 PRINT AT 18,4;"DAY BETWEEN 1100-2000"
```

CDATE (continued)

```
230 PAUSE 200
240 CLS
250 PRINT
260 PRINT
270 PRINT
280 PRINT "ENTER JULIAN DAY ";
290 INPUT JD
300 PRINT JD
310 LET JD=JD-2400000
320 PRINT
330 LET ND=JD-15018
340 LET Y1=ND/365.25
350 LET Y=1900+ INT (Y1)
360 LET Y4=Y/4
370 LET Y5=Y/100
380 LET Y6=Y4- INT (Y4)
390 LET Y7=Y5- INT (Y5)
400 IF Y6=0 AND Y7 <> 0 THEN LET LY=1
410 LET D=365.25*(Y1- INT (Y1))
420 LET D= INT (D)
430 LET D=D+ INT ((Y-2000)/100)
440 IF Y<1583 THEN LET D=D-(10+ INT ((Y-1500)/100))
450 IF D-31 >= 0 THEN GOTO 490
460 LET M=1
470 LET D=D
480 GOTO 920
490 IF D-59 >= 0 THEN GOTO 530
500 LET M=2
510 LET D=D-31
520 GOTO 920
530 IF D-90 >= 0 THEN GOTO 570
540 LET M=3
550 LET D=D-59
560 GOTO 920
570 IF D-120 >= 0 THEN GOTO 610
580 LET M=4
590 LET D=D-90
600 GOTO 920
610 IF D-151 >= 0 THEN GOTO 650
620 LET M=5
630 LET D=D-120
640 GOTO 920
650 IF D-181 >= 0 THEN GOTO 690
660 LET M=6
670 LET D=D-151
680 GOTO 920
690 IF D-212 >= 0 THEN GOTO 730
700 LET M=7
710 LET D=D-181
720 GOTO 920
730 IF D-243 >= 0 THEN GOTO 770
740 LET M=8
750 LET D=D-212
760 GOTO 920
770 IF D-273 >= 0 THEN GOTO 810
780 LET M=9
790 LET D=D-243
```


CDATE (continued)

```
800 GOTO 920
810 IF D-304 >= 0 THEN GOTO 850
820 LET M=10
830 LET D=D-273
840 GOTO 920
850 IF D-334 >= 0 THEN GOTO 890
860 LET M=11
870 LET D=D-304
880 GOTO 920
890 IF D-365 >= 0 THEN GOTO 920
900 LET M=12
910 LET D=D-334
920 IF LY=1 AND M>2 THEN LET D=D-1
930 IF LY=0 AND M<3 THEN LET D=D-1
940 IF D >= 1 THEN GOTO 1170
950 IF D<1 AND LY=1 AND M=3 THEN GOTO 1010
960 IF D<1 AND LY=0 AND M=3 THEN GOTO 1030
970 IF D<1 AND M=1 THEN GOTO 1060
980 IF D<1 AND M=2 OR M=4 OR M=6 THEN GOTO 1100
990 IF D<1 AND M=9 OR M=11 THEN GOTO 1100
1000 GOTO 1130
1010 LET D=29
1020 GOTO 1040
1030 LET D=28
1040 LET M=2
1050 GOTO 1170
1060 LET D=31
1070 LET M=12
1080 LET Y=Y-1
1090 GOTO 1170
1100 LET D=31
1110 LET M=M-1
1120 GOTO 1170
1130 IF D<1 THEN GOTO 1150
1140 GOTO 1170
1150 LET D=30
1160 LET M=M-1
1170 IF D=365 THEN GOTO 1240
1180 IF D=366 THEN GOTO 1200
1190 GOTO 1260
1200 LET D=1
1210 LET M=1
1220 LET Y=Y+1
1230 GOTO 1260
1240 LET D=31
1250 LET M=12
1260 PRINT
1270 PRINT
1280 PRINT
1290 PRINT "CALENDAR DATE IS ";Y;" ";M;" ";D
1300 IF Y <> 1582 THEN GOTO 1350
1310 PRINT
1320 PRINT "(NOTE: IN 1582 DATES BEFORE OCT 15"
1330 PRINT "MUST BE DECREASED BY 10 DAYS"
1340 PRINT "TO MATCH JULIAN CALENDAR)"
1350 PRINT
1360 PRINT
```

CDATE (continued)

```
1370 PRINT
1380 PRINT "WANT ANOTHER DATE Y/N? "
1390 INPUT A$
1400 IF A$="N" THEN GOTO 1460
1410 LET LY=0
1420 LET Y=0
1430 LET D=0
1440 LET M=0
1450 GOTO 240
1460 CLS
1470 STOP
```


Program 5: EPOCH

Updating Star Coordinates

Right ascensions and declinations are listed in nebula and star tables for a given epoch, say, 1950. If you want to find a faint stellar object by setting to the circles of an equatorially mounted telescope, you will need to update positions to the current epoch of observation. This program does this for you with sufficient accuracy to position the stellar object within the field of view of a typical finder telescope (see Figure 5.1). You must input the epoch of the star table you are using, your present epoch, and then each right ascension and declination you need updated.

The listing of the EPOCH program follows.



Figure 5.1: The EPOCH program updates star coordinates to allow for the precession of the equinox. This is a typical display generated by the program.

EPOCH

```
 10 CLEAR
 20 PRINT
 30 REM EPOCH 1/23/83
 40 PRINT
 50 REM
 60 REM
 70 PRINT
 80 PRINT
 90 PRINT "AN ASTRONOMY PROGRAM"
100 PRINT
110 PRINT "BY ERIC BURGESS F.R.A.S."
120 PRINT
130 PRINT
140 PRINT "ALL RIGHTS RESERVED BY"
150 PRINT "S  AND T SOFTWARE SERVICES"
160 PRINT
170 PAUSE 150
180 PRINT
190 PRINT
200 PRINT
210 PRINT "THIS PROGRAM COMPUTES"
220 PRINT "RIGHT ASCENSION  AND DECLINATION"
```


EPOCH (continued)

```
230 PRINT "FOR AN EPOCH WHEN RA  AND DEC"
240 PRINT "ARE GIVEN FOR ANOTHER EPOCH"
250 PRINT "REDUCING FOR PRECESSION"
260 PAUSE 300
270 CLS
280 PRINT
290 PRINT
300 PRINT "WHAT IS FIRST EPOCH (YEAR)? ";
310 INPUT Y1
320 PRINT Y1
330 PRINT
340 PRINT "WHAT IS SECOND EPOCH ";
350 INPUT Y2
360 PRINT Y2
370 LET T2=((Y2+Y1)/2-1900)/100
380 LET X=3.07234+(.00186*T2)
390 LET Y=20.0468-(.0085*T2)
400 LET Z=Y/15
410 PRINT
420 PRINT
430 REM CALC DAYS BETWEEN EPOCHS
440 LET T=Y2-Y1
450 PRINT "PICK MODE OF INPUT"
460 PRINT
470 PRINT "R.A. IN DECIMAL HRS.  (1)"
480 PRINT "  OR IN HR,MI,SE  (2)"
490 INPUT RA
500 PRINT
510 PRINT "DEC. IN DECIMAL DEG.(3)"
520 PRINT "   OR IN DE,MI,SE  (4)"
530 INPUT DE
540 PRINT
550 IF RA=2 THEN GOTO 600
560 PRINT "INPUT R.A AT EPOCH ";Y1;" ";
570 INPUT R1
580 PRINT R1
590 GOTO 650
600 PRINT "INPUT R.A. AT EPOCH ";Y1;" ";
610 INPUT R1
620 INPUT M1
630 INPUT S1
640 PRINT R1;",";M1;",";S1
650 IF DE=4 THEN GOTO 680
660 PRINT
670 REM
680 IF RA=2 THEN LET R1=R1+M1/60+S1/3600
690 IF DE=4 THEN GOTO 750
700 PRINT "INPUT DECLINATION"
710 PRINT "IN DECIMAL DEGREES ";
720 INPUT D1
730 PRINT D1
740 GOTO 860
750 PRINT "INPUT DECLINATION "
760 PRINT "      DEGREES... ";
770 INPUT DC
780 PRINT DC
790 PRINT "      MINUTES... ";
```

EPOCH (continued)

```
 800 INPUT MC
 810 PRINT MC
 820 PRINT "      SECONDS... ";
 830 INPUT SC
 840 PRINT SC
 850 LET D1=DC+MC/60+SC/3600
 860 LET R4=R1
 870 LET R1=R1*15
 880 LET SR= SIN (R1/57.29878)
 890 LET TR= TAN (R1/57.29878)
 900 LET W=.0042*T*(X+(2*SR*TR))
 910 LET R2=R1+W
 920 LET D2=D1+.00028 *T*Y*  COS (R1/57.29878)
 930 CLS
 940 PRINT
 950 PRINT "FOR EPOCH ";Y1;" RA  ";R4
 960 PRINT TAB 15;"DEC ";D1
 970 PRINT
 980 PRINT "AT EPOCH ";Y2
 990 LET R2=R2/15
1000 IF R2>24 THEN LET R2=R2-24
1010 IF R2<0 THEN LET R2=R2+24
1020 LET X$= STR$ (R2)
1030 LET RC= VAL (X$( TO 6))
1040 PRINT
1050 PRINT
1060 PRINT "RIGHT ASCENSION IS... ";RC;" HRS"
1070 LET R3= INT (R2)
1080 LET M2=60*(R2- INT (R2))
1090 LET S2=60*(M2- INT (M2))
1100 LET M2= INT (M2)
1110 LET S2= INT (S2)
1120 PRINT
1130 PRINT "   OR.. ";R3;" HR ";M2;" MI ";S2;" SE"
1140 IF D2<-90 THEN LET D2=-90+ ABS (D2)- INT ( ABS (D2))
1150 IF D2<0 THEN LET DD= VAL (( STR$ (D2))( TO 6))
1160 IF D2<0 THEN GOTO 1210
1170 IF D2>90 THEN GOTO 1200
1180 LET DD= VAL (( STR$ (D2))( TO 5))
1190 GOTO 1210
1200 LET DD=90-(D2- INT (D2))
1210 PRINT
1220 PRINT
1230 PRINT "DECLINATION IS... ";DD;" DEG"
1240 IF D2<0 THEN GOTO 1270
1250 LET D4= INT (D2)
1260 GOTO 1290
1270 LET D4= INT (D2)+1
1280 GOTO 1310
1290 LET ME=60*(D2- INT (D2))
1300 GOTO 1330
1310 LET ME=1+60*(D2- INT (D2))
1320 LET ME=60-ME
1330 LET MS= INT (ME)
1340 LET SE=60*(ME- INT (ME))
1350 LET SS= INT (SE)
1360 PRINT
```


EPOCH (continued)

```
1370 PRINT "   OR.. ";D4;" DEG ";MS;" MI ";SS;" SE"
1380 PRINT
1390 PRINT
1400 PRINT
1410 PRINT "WANT COPY Y/N? "
1420 INPUT A$
1430 IF A$="Y" THEN GOTO 1640
1440 CLS
1450 PRINT
1460 PRINT "ANOTHER CONVERSION"
1470 PRINT "FOR SAME EPOCHS Y/N ? ";
1480 INPUT A$
1490 PRINT A$
1500 PRINT
1510 PAUSE 50
1520 IF A$="N" THEN GOTO 1560
1530 CLS
1540 GOTO 540
1550 PRINT
1560 CLS
1570 PRINT "ANOTHER EPOCH Y/N? ";
1580 INPUT A$
1590 PRINT A$
1600 PAUSE 50
1610 IF A$="Y" THEN GOTO 270
1620 CLS
1630 STOP
1640 COPY
1650 GOTO 1460
```

Program 6: PSTAR

Transits and Elongations of Polaris

When an equatorial mount of a telescope is set up, the polar axis should point to the celestial pole. The exact elevation of the pole at any observing site can be determined from the elevation of Polaris when it is at east or west elongation, and the azimuth of true north can be determined by observing Polaris when the star is at upper or lower transit. The elevation of the pole is also the latitude of the observer.

This program determines the times of Polaris's elongations and transits (see Figure 6.1) within ten minutes for any date, which is sufficiently accurate to set a polar axis except for a very large telescope.

Starting from epoch 1980, the program uses the right ascension of Polaris, the hour angle, and the sidereal time for a given date to determine the local time at which the transits and elongations take place.

The listing of the PSTAR program follows.



Figure 6.1: *By observing the transits and elongations of Polaris, you can ascertain true north and your latitude. This is a typical display generated by the PSTAR program.*

PSTAR

```
 10 CLS
 20 REM PSTAR 4/13/83
 30 PRINT AT 4,5;"----------------------"
 40 PRINT AT 5,5;"I      POLARIS      I"
 50 PRINT AT 6,5;"----------------------"
 60 PRINT AT 8,5;"AN ASTRONOMY PROGRAM"
 70 PRINT AT 9,4;"BY ERIC BURGESS F.R.A.S."
 80 PRINT AT 11,5;"ALL RIGHTS RESERVED BY"
 90 PRINT AT 12,4;"S AND T SOFTWARE SERVICE"
100 PRINT AT 14,8;"PROVIDES TIMES OF"
110 PRINT AT 15,5;"ELONGATIONS AND TRANSITS"
120 PRINT AT 16,5;"OF POLARIS FOR ANY DATE"
130 PRINT AT 17,8;"(WITHIN 10 MINUTES)"
140 PRINT AT 18,4;"FOR SETTING OR CHECKING THE"
150 PRINT AT 19,4;"ALIGNMENT OF THE POLAR AXIS"
160 PRINT AT 20,8;"OF YOUR TELESCOPE"
170 PAUSE 350
180 LET LY=0
190 CLS
```

PSTAR (continued)

```
200 PRINT
210 PRINT "INPUT YEAR... ";
220 INPUT Y
230 PRINT Y
240 PRINT
250 LET Y4=Y/4
260 LET Y5=Y/100
270 LET Y6=Y4- INT (Y4)
280 LET Y7=Y5- INT (Y5)
290 IF Y6=0 AND Y7 <> 0 THEN LET LY=1
300 PRINT "      MONTH... ";
310 INPUT M
320 PRINT M
330 IF M<1 OR M>12 THEN GOTO 300
340 LET D2=0
350 IF M=2 THEN LET D2=31
360 IF M=3 THEN LET D2=59
370 IF M=4 THEN LET D2=90
380 IF M=5 THEN LET D2=120
390 IF M=6 THEN LET D2=151
400 IF M=7 THEN LET D2=181
410 IF M=8 THEN LET D2=212
420 IF M=9 THEN LET D2=243
430 IF M=10 THEN LET D2=273
440 IF M=11 THEN LET D2=304
450 IF M=12 THEN LET D2=334
460 PRINT
470 PRINT TAB 7;"DAY... ";
480 INPUT D1
490 PRINT D1
500 IF D1<1 OR D1>31 THEN GOTO 470
510 IF M=2 AND D1>29 THEN GOTO 470
520 IF M=2 AND D1>28 AND LY=0 THEN GOTO 470
530 LET D=D1+D2
540 IF LY=1 AND M>2 THEN LET D=D+1
550 LET G=Y-1980
560 LET YC=-.01638889*G
570 LET GST=6.65422+YC
580 LET DG=D*.0657096
590 LET GT=GST+DG
600 IF GT>24 THEN LET GT=GT-24
610 IF GT<0 THEN LET GT=GT+24
620 PRINT
630 PRINT " LONGITUDE... ";
640 INPUT LO
650 PRINT LO
660 PAUSE 60
670 LET CF=.065556*LO/360
680 LET GT=GT+GT*CF
690 LET HA=2.183333-GT
700 IF HA<0 THEN LET HA=HA+24
710 LET MT=HA
720 LET QD=5.98362
730 LET QE=5.933333
```


PSTAR (continued)

```
740 LET QW=5.933333
750 LET QL=11.967222
760 CLS
770 PRINT
780 PRINT
790 PRINT
800 PRINT TAB 2;"  LOWER TRANSIT-NEXT(5)"
810 PRINT TAB 2;"  UPPER TRANSIT-NEXT(6)"
820 PRINT
830 PRINT
840 PRINT "SELECT BY NUMBER ";
850 INPUT S
860 PRINT S
870 PAUSE 100
880 CLS
890 PRINT AT 6,0;"ON ";Y;" ";M;
900 PRINT " ";D1;" LONGITUDE ";LO
910 PRINT
920 IF S=1 THEN GOTO 1130
930 IF S=2 THEN GOTO 1220
940 IF S=3 THEN GOTO 1320
950 IF S=4 THEN GOTO 1420
960 IF S=5 THEN GOTO 1520
970 LET NUT=HA
980 LET IN=NUT
990 GOSUB 1610
1000 GOSUB 1640
1010 LET N$= STR$ IN
1020 PRINT "NEXT UPPER TRANSIT OF POLARIS"
1030 PRINT "WILL BE AT ";N$(1 TO 5);" HRS"
1040 GOSUB 1690
1050 GOTO 1710
1060 LET WN=MT+QW
1070 LET IN=WN
1080 GOSUB 1610
1090 GOSUB 1640
1100 LET W$= STR$ IN
1110 PRINT "NEXT WEST ELONGATION OF POLARIS"
1120 PRINT "WILL BE AT ";W$(1 TO 5);" HRS"
1130 GOSUB 1690
1140 GOTO 1710
1150 LET WP=MT-QL-QE
1160 LET IN=WP
1170 GOSUB 1610
1180 GOSUB 1640
1190 LET P$= STR$ IN
1200 PRINT "PREVIOUS WEST ELONGATION"
1210 PRINT "OF POLARIS"
1220 PRINT "    WAS AT ";P$(1 TO 5);" HRS"
1230 GOSUB 1690
1240 GOTO 1710
1250 LET EN=MT+QL+QW
1260 LET IN=EN
1270 GOSUB 1610
1280 GOSUB 1640
```

PSTAR (continued)

```
1360 LET E$= STR$ IN
1370 PRINT "NEXT EAST ELONGATION"
1380 PRINT "OF POLARIS"
1390 PRINT "WILL BE AT ";E$(1 TO 5);" HRS"
1400 GOSUB 1690
1410 GOTO 1710
1420 LET EP=MT-QW
1430 LET IN=EP
1440 GOSUB 1610
1450 GOSUB 1640
1460 LET Q$= STR$ IN
1470 PRINT "PREVIOUS EAST ELONGATION"
1480 PRINT "OF POLARIS"
1490 PRINT "    WAS AT ";Q$(1 TO 5);" HRS"
1500 GOSUB 1690
1510 GOTO 1710
1520 LET NLT=MT+QL
1530 LET IN=NLT
1540 GOSUB 1610
1550 GOSUB 1640
1560 LET T$= STR$ IN
1570 PRINT "NEXT LOWER TRANSIT OF POLARIS"
1580 PRINT "WILL BE AT ";T$(1 TO 5);" HRS"
1590 GOSUB 1690
1600 GOTO 1710
1610 IF IN>24 THEN LET IN=IN-24
1620 IF IN<0 THEN LET IN=IN+24
1630 RETURN
1640 LET IN=IN+IN*.00273043
1650 LET HM= INT (IN)
1660 LET MI=60*(IN- INT (IN))
1670 LET M2= INT (MI)
1680 RETURN
1690 PRINT
1700 PRINT TAB 8;" OR ";HM;" HR ";M2;" MI"
1710 PRINT
1720 PRINT "WANT ANOTHER ELONGATION"
1730 PRINT "OR TRANSIT Y/N ";
1740 INPUT A$
1750 PRINT A$
1760 IF A$="Y" OR A$="YES" THEN GOTO 760
1770 PRINT
1780 PRINT "WANT ANOTHER DATE Y/N";
1790 INPUT A$
1800 PRINT A$
1810 IF A$="Y" OR A$="YES" THEN GOTO 180
1820 CLS
1830 STOP
```

PART 2



THE MOON

The Moon has always beckoned and intrigued mankind; its phases and cycles are recorded on bones recovered from the sites of prehistoric settlements. The Moon has provided us with light, it has served as a timekeeper, and it has been a basis for romantic ideas and flights of fanciful fiction.

With the advent of the computer age and space exploration, the Moon has become much less of a mystery to us. The programs in this section allow you to determine the phases of the Moon, the dates of lunar eclipses, the position of the Moon relative to the stars, and the times of moonrise and moonset for any date.

Program 7: RADEM

Right Ascension and Declination of Moon for Any Date

This program computes the approximate right ascension and declination of the Moon for any requested date. These data enable you to fix the Moon's position relative to the stars.

The program determines the position of Moon in its orbit by counting from its position at the epoch 1960 (instructions 760 through 780) to the current date. It eliminates complete revolutions (instructions 800 and 830). It also adjusts for the motion of the nodes and the perigee position during the period since or before the epoch (instructions 840 through 1060). Then it adjusts for the eccentricity of the Moon's orbit and for the inclination of the orbit to the plane of the ecliptic (instructions 1070 and 1080). The program prints out the right ascension and declination for the requested date (see Figure 7.1).

The Moon's orbit is perturbed in many ways, and other corrections can be added to achieve higher orders of accuracy. But for most purposes this program provides sufficient accuracy in locating the approximate position of the Moon among the stars for any date and time.

If you wish, this program can be merged with RADEC, the program for finding right ascensions and declinations of the planets. If you intend to do this, you should key in this program using line numbers starting at 2000. Then you can use it as a subroutine for RADEC.

The listing of the RADEM program follows.



Figure 7.1: The RADEM program calculates the right ascension and declination of the Moon for any date and time. It generates the type of display shown here

RADEM

```
 10 REM RADEM 4/13/83
 20 PRINT AT 4,5;"ASTRONOMY PROGRAM"
 30 PRINT AT 6,5;"----------------"
 40 PRINT AT 7,5;"I     RADEM     I"
 50 PRINT AT 8,5;"----------------"
 60 PRINT AT 10,3;"BY ERIC BURGESS F.R.A.S."
 70 PRINT AT 12,3;"ALL RIGHTS RESERVED BY"
 80 PRINT AT 13,2;"S AND T SOFTWARE SERVICE"
 90 PRINT AT 15,0;"DISPLAYS R.A. AND DECLINATION"
100 PRINT AT 16,3;"OF MOON FOR REQUESTED"
110 PRINT AT 17,6;"DATE AND TIME"
120 PAUSE 350
130 CLS
140 PRINT
150 PRINT
160 PRINT "ENTER YEAR ";
170 INPUT Y
180 PRINT Y
190 PRINT
200 PRINT "     MONTH ";
210 INPUT M
220 PRINT M
230 PRINT
```

RADEM (continued)

```
240 PRINT "        DAY ";
250 INPUT D
260 PRINT D
270 PRINT
280 PRINT " TIME ZONE ";
290 INPUT TZ
300 PRINT TZ
310 PRINT
320 PRINT " TIME(HRS) ";
330 INPUT T1
340 PRINT T1
350 PRINT
360 PRINT
370 LET LY=0
380 IF Y/4- INT (Y/4)=0 AND Y/100- INT (Y/100) <> 0 THEN
    LET LY=1
390 LET YP=Y
400 LET MP=M
410 LET DP=D
420 LET DU=D
430 LET TU=TZ+T1
440 IF TU<24 THEN GOTO 470
450 LET TU=TU-24
460 LET DU=DU+1
470 LET DU=DU+TU/24
480 IF M <> 2 THEN GOTO 540
490 IF LY=1 AND INT (DU)>29 THEN LET M=3
500 IF LY=0 AND INT DU>28 THEN LET M=3
510 IF LY=1 AND INT (DU)>29 THEN LET DU=DU-29
520 IF LY=0 AND INT (DU)>28 THEN LET DU=DU-28
530 GOTO 640
540 IF INT (DU)<31 THEN GOTO 640
550 IF M <> 4 OR M <> 6 OR M <> 9 OR M <> 11 THEN GOTO 590
560 LET M=M+1
570 LET DU=DU-30
580 GOTO 640
590 LET M=M+1
600 LET DU=DU-31
610 IF M <> 13 THEN GOTO 640
620 LET M=1
630 LET Y=Y+1
640 LET T2=TU
650 LET D2=DU
660 LET Y2=Y
670 LET M2=M
680 LET DG=365*Y+DU+((M-1)*31)
690 IF M >= 3 THEN GOTO 720
700 LET DG=DG+ INT ((Y-1)/4)- INT ((.75)* INT ((Y-1)/100+1))
710 GOTO 740
720 LET DG=DG- INT (M*.4+2.3)+ INT (Y/4)
730 LET DG=DG- INT ((.75)* INT ((Y/100)+1))
740 LET NM=DG-715875.2
750 LET NM=NM-.5
760 LET LZ=311.1687
770 LET LE=178.699
780 LET LP=255.7433
790 LET PG=.111404*NM+LP
```


RADEM (continued)

```
800 LET PG=PG/360- INT (PG/360)
810 LET PG=PG*360
820 IF PG>0 THEN GOTO 840
830 LET PG=360+PG
840 LET LMD=LZ+360*NM/27.321582
850 LET PG=LMD-PG
860 LET DR=6.2886* SIN (.0174533*PG)
870 LET LMD=LMD+DR
880 LET LMD=LMD/360- INT (LMD/360)
890 LET LMD=LMD*360
900 IF LMD>0 THEN GOTO 920
910 LET LMD=LMD+360
920 LET RM=LMD/15
930 IF RM>0 AND RM<24 THEN GOTO 980
940 IF RM<0 THEN GOTO 970
950 LET RM=RM-24
960 GOTO 980
970 LET RM=RM+24
980 LET AL=LE-NM*.052954
990 LET AL=AL/360- INT (AL/360)
1000 LET AL=AL*360
1010 IF AL>0 THEN GOTO 1030
1020 LET AL=AL+360
1030 LET AL=LMD-AL
1040 IF AL>0 THEN GOTO 1060
1050 LET AL=AL+360
1060 IF AL>360 THEN LET AL=AL-360
1070 LET HE=5.1454* SIN (AL*3.14159/180)
1080 LET DM=HE+23.1444* SIN (LMD*3.14159/180)
1090 LET R$= STR$ (RM)
1100 LET D$= STR$ (DM)
1110 LET U$=R$(1 TO 4)
1120 LET V$=D$(1 TO 4)
1130 LET MRA= VAL (U$)
1140 LET MDE= VAL (V$)
1150 IF MRA<10 THEN LET U$=R$(1 TO 3)
1160 IF MDE<10 OR MDE>-10 THEN LET V$=D$(1 TO 3)
1170 CLS
1180 PRINT
1190 PRINT
1200 PRINT
1210 PRINT "LMT ";T1;" HR:YR ";YP;" MNTH ";MP;" DY ";DP
1220 PRINT
1230 PRINT "UT ";TU;" HR:YR ";Y2;" MNTH ";M2;" DY "; INT (D2)
1240 PRINT
1250 PRINT
1260 PRINT
1270 PRINT
1280 PRINT "RA OF MOON IS ";U$;" HOURS"
1290 PRINT
1300 PRINT "DECLINATION IS ";V$;" DEGREES"
1310 PRINT "-----------------------------"
1320 PRINT
1330 PRINT "WANT ANOTHER DATE Y/N"
1340 INPUT A$
1350 IF A$="N" THEN GOTO 1410
1360 IF A$="Y" THEN GOTO 1390
```

RADEM (continued)

```
1370 PRINT "INVALID RESPONSE"
1380 GOTO 1320
1390 CLS
1400 GOTO 140
1410 CLS
1420 STOP
```


Program 8: MRISE

Time of Rising, Transit, and Setting of Moon for Any Date

This program computes approximate times of moonrise, transit, and moonset for any requested date and geographic location. The time of transit is the time when the Moon crosses the meridian.

The program calculates the right ascension and declination for the requested date using the routines of the RADEM program. It then takes a coordinate transformation to suit the observer's latitude and calculates the approximate times of moonrise and moonset for that latitude, adjusting also for longitude. Information is displayed on the monitor screen as shown in Figure 8.1.

If you wish, this program can be merged with RADEM, the program for finding right ascensions and declinations of the Moon. If you intend to do this, you should key in this program using line numbers starting at 2000. Then you can use it as a subroutine for RADEM.

The listing of the MRISE program follows.



Figure 8.1: MRISE develops this display of times of moonrise, transit, and moonset for a given date.

MRISE

```
 10 CLEAR
 20 DIM P$(4)
 30 REM MRISE 4/12/1983
 40 PRINT AT 5,5;"----------------"
 50 PRINT AT 6,5;"I     MRISE     I"
 60 PRINT AT 7,5;"----------------"
 70 LET FR=.01745329
 80 LET FD=57.299578
 90 LET LA=38.24
100 LET LO=122.49
110 REM CHANGE ABOVE TO YOUR LATITUDE AND LONGITUDE
120 PRINT AT 10,3;"BY ERIC BURGESS F.R.A.S."
130 PRINT AT 12,4;"ALL RIGHTS RESERVED BY"
140 PRINT AT 13,3;"S AND T SOFTWARE SERVICE"
150 PAUSE 350
160 CLS
170 PRINT
180 PRINT
190 PRINT
200 PRINT TAB 5;"THIS PROGRAM GIVES"
210 PRINT TAB 2;"APPROXIMATE TIMES OF THE"
220 PRINT TAB 2;"RISING, TRANSIT, AND SETTING"
```


MRISE (continued)

```
230 PRINT TAB 8;"OF THE MOON"
240 PRINT "FOR THE DATE WHICH YOU REQUEST"
250 PRINT
260 PRINT
270 PRINT "INITIAL CONDITIONS ARE SET FOR.."
280 PRINT "  LATITUDE.. ";LA
290 PRINT "  LONGITUDE  ";LO
300 PRINT
310 PRINT "WANT TO CHANGE THEM Y/N? ";
320 INPUT B$
330 PRINT B$
340 IF B$ <> "Y" THEN GOTO 410
350 PRINT "TYPE LATITUDE ";
360 INPUT LA
370 PRINT LA
380 PRINT "TYPE LONGITUDE ";
390 INPUT LO
400 PRINT LO
410 PRINT
420 PRINT
430 CLS
440 PRINT
450 PRINT
460 PRINT "ENTER THE DATE"
470 PRINT
480 PRINT "       YEAR.... ";
490 INPUT Y
500 PRINT Y
510 PRINT
520 PRINT "       MONTH... ";
530 INPUT M
540 PRINT M
550 IF M<1 OR M>12 THEN GOTO 520
560 PRINT
570 PRINT "       DAY..... ";
580 INPUT DY
590 PRINT DY
600 PRINT
610 IF DY<1 OR DY>31 THEN GOTO 570
620 IF M=2 AND DY>29 THEN GOTO 50
630 LET DX=DY
640 LET MX=M
650 LET YX=Y
660 LET DY=DY+(LO/15)/24
670 LET DG=365*Y+DY+((M-1)*31)
680 IF M >= 3 THEN GOTO 710
690 LET DG=DG+ INT ((Y-1)/4)- INT ((.75)* INT ((Y-1)/100+1))
700 GOTO 730
710 LET DG=DG- INT (M*.4+2.3)+ INT (Y/4)
720 LET DG=DG- INT ((.75)* INT ((Y/100)+1))
730 LET NI=DG-715875
740 REM GET S.T.
750 GOSUB 1580
760 GOTO 1240
770 CLS
780 FAST
790 GOSUB 1000
800 SLOW
```

MRISE (continued)

```
 810 PRINT
 820 PRINT "DATA REQUESTED FOR ";YX;" ";MX;" ";DX
 830 PRINT "-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-"
 840 IF CI=3 THEN PRINT TAB 22;"R.A.   DEC"
 850 IF CI=3 THEN PRINT TAB 22;"HRS    DEG"
 860 IF CI=3 THEN GOTO 890
 870 PRINT "         HEL. DIST.  R.A.  DEC."
 880 PRINT "         LONG. A.U.  HRS   DEG."
 890 PRINT "-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-"
 900 GOSUB 1710
 910 PRINT
 920 PRINT "WANT COPY Y/N? "
 930 INPUT C$
 940 IF C$="N" OR C$="NO" THEN GOTO 960
 950 COPY
 960 CLS
 970 PRINT
 980 PRINT
 990 GOTO 1870
1000 REM CALC TIMES
1010 LET TA= TAN (FR*LA)* TAN (FR*DM)
1020 IF TA<0 THEN GOTO 1040
1030 IF TA>0 THEN GOTO 1060
1040 LET TA= ABS (TA)
1050 GOTO 1070
1060 LET TA=-TA
1070 LET TA= ACS (TA)
1080 LET TA=FD*TA
1090 LET B=-TA/15
1100 LET TZ=0
1110 LET TR=(B+RM+TZ-T2)*.99727
1120 LET TR=TR-1/60
1130 IF TR<0 THEN LET TR=TR+24
1140 IF TR>24 THEN LET TR=TR-24
1150 LET C=TA/15
1160 LET TS=(C+RM+TZ-T2)*.99727
1170 LET TS=TS+3/60
1180 IF TS<0 THEN LET TS=TS+24
1190 IF TS>24 THEN LET TS=TS-24
1200 LET TT=RM-T2+1/60
1210 IF TT<0 THEN LET TT=TT+24
1220 IF TT>24 THEN LET TT=TT-24
1230 RETURN
1240 REM RISE, XSIT, SET OF MOON
1250 LET LZ=311.1687
1260 LET LE=178.699
1270 LET LP=255.7433
1280 LET PG=.111404*NI+LP
1290 LET PG=(PG/360- INT (PG/360))*360
1300 LET LMD=LZ+360*NI/27.32158
1310 LET LMD=(LMD/360- INT (LMD/360))*360
1320 LET PG=LMD-PG
1330 LET DR=6.2886* SIN (FR*PG)
1340 LET LMD=LMD+DR
1350 LET LMD=(LMD/360- INT (LMD/360))*360
1360 LET RM=LMD/15
1370 IF RM>24 THEN LET RM=RM-24
1380 IF RM<0 THEN LET RM=RM+24
```


MRISE (continued)

```
1390 LET AL=LE-NI*.052954
1400 LET AL=LMD-AL
1410 LET AL=(AL/360- INT (AL/360))*360
1420 LET HE=5.1454* SIN (AL*3.14159/180)
1430 LET DM=HE+23.1444* SIN (LMD*3.14159/180)
1440 LET R$=( STR$ (RM))( TO 5)
1450 LET D$=( STR$ (DM))( TO 5)
1460 CLS
1470 PRINT
1480 PRINT "MOON DATA REQUESTED FOR..."
1490 PRINT TAB 4;"YEAR ";YX;" MONTH ";MX;" DAY ";DX
1500 PRINT
1510 PRINT "AT NOON"
1520 PRINT "R.A. OF MOON IS. ";R$;" HRS"
1530 PRINT "DECLINATION IS.. ";D$;" DEG"
1540 PRINT "-----------------------------"
1550 LET P$="MOON"
1560 GOSUB 1000
1570 GOTO 1720
1580 REM SIDEREAL TIME
1590 LET GC=11.927485
1600 LET TC=.065711
1610 LET T2=TC*(NI-7020)+GC
1620 IF T2<24 THEN GOTO 1650
1630 LET T2=T2-24
1640 GOTO 1620
1650 IF T2>-24 THEN GOTO 1680
1660 LET T2=T2+24
1670 GOTO 1650
1680 IF T2>0 THEN GOTO 1700
1690 LET T2=T2+24
1700 RETURN
1710 GOSUB 1000
1720 PRINT
1730 PRINT
1740 LET TM=60*(TR- INT (TR))
1750 LET TR= INT (TR)
1760 LET TN=60*(TT- INT (TT))
1770 LET TT= INT (TT)
1780 LET TP=60*(TS- INT (TS))
1790 LET TS= INT (TS)
1800 PRINT
1810 PRINT P$
1820 PRINT "RISES AT.... ";TR;" HRS "; INT (TM);" MIN "
1830 PRINT "TRANSITS AT. ";TT;" HRS "; INT (TN);" MIN"
1840 PRINT "SETS AT..... ";TS;" HRS "; INT (TP);" MIN"
1850 GOTO 910
1860 RETURN
1870 PRINT
1880 PRINT "WANT ANOTHER DATE Y/N? ";
1890 INPUT B$
1900 PRINT B$
1910 IF B$="N" THEN GOTO 1950
1920 LET FL=0
1930 LET F=0
1940 IF B$="Y" THEN GOTO 430
1950 CLS
1960 STOP
```

Program 9: ECLIP

Umbral Lunar Eclipses for Any Year

This program computes the date of the first umbral eclipse of the Moon in any year requested and shows the magnitude of the eclipse, that is, the fraction of the Moon's disk covered by the umbral shadow of Earth. When this fraction reaches or exceeds 1, the eclipse is total. Because the obscuring effects of Earth's penumbra are barely perceptible, penumbral eclipses are not identified. If there is no umbral eclipse in the year requested, the program continues until it finds the first eclipse in subsequent years. You can also ask the computer to display subsequent umbral eclipses. If there are no more eclipses in the year requested, the computer will search for and display the first umbral eclipse in subsequent years. A screen display generated by this program is shown in Figure 9.1.

The program checks each full moon from the date requested to see whether it occurs within the required distance from a node, and it cycles until it finds the Julian day of the first full moon occurring close to the node. It calculates the amount of the Moon's disk covered by Earth's shadow, and it determines the date on which the eclipse occurs by converting Julian day to calendar date.

The listing for the ECLIP program follows.



Figure 9.1: By checking whether a full moon occurs close to a node of the Moon's orbit, the ECLIP program tells you when eclipses, if any, take place in any year. This is a typical display generated by the program.

ECLIP

```
10 REM ECLIP 4/17/83
20 PRINT AT 3,3;"-------------------------"
30 PRINT AT 4,3;"I LUNAR UMBRAL ECLIPSES I"
40 PRINT AT 5,3;"-------------------------"
50 PRINT AT 8,3;"BY ERIC BURGESS F.R.A.S."
60 PRINT AT 12,4;"ALL RIGHTS RESERVED BY"
70 PRINT AT 14,3;"S AND T SOFTWARE SERVICE"
80 PAUSE 300
90 CLS
100 PRINT AT 6,3;"THIS PROGRAM GIVES MAGNITUDE"
110 PRINT AT 7,5;"AND DATE OF LUNAR UMBRAL"
120 PRINT AT 8,5;"ECLIPSES STARTING AT ANY"
130 PRINT AT 9,8;"YEAR REQUESTED"
140 PAUSE 250
150 CLS
160 CLEAR
170 PRINT AT 10,0;"PLEASE TYPE YEAR TO START ";
180 INPUT Y
190 PRINT Y
200 LET FL=0
210 PAUSE 100
220 CLS
```

ECLIP (continued)

```
230 PRINT AT 5,0;"RUNNING .. PLEASE WAIT"
240 LET Z=Y-1900
250 LET ZD=(Z*12.368267)-2
260 LET A= INT (ZD)
270 LET A=A+1
280 LET B=29.1053561*A
290 LET C=B+13.7774
300 LET D=(25.869180*A)+138.94
310 LET E=(30.670565*A)+216.6378
320 LET F=E-( SIN (D* PI /180))*.412
330 LET G=F+( SIN (2*D* PI /180))/8.8
340 LET H=G+( SIN (C* PI /180))*2.2265
350 LET I=H+( SIN (2*E* PI /180))*.13
360 LET I= SIN (I* PI /180)
370 LET J=0.7128-( COS (D* PI /180))/36
380 LET W=I*10 ** J
390 IF W>0 THEN GOTO 420
400 LET W=1.84769+W*1.8216
410 GOTO 430
420 LET W=1.84769-W*1.8216
430 LET K=W+( COS (D* PI /180))/30
440 IF K<0 THEN GOTO 270
450 GOTO 570
460 PRINT "------------"
470 PRINT "MAGNITUDE OF ECLIPSE IS ...  "
480 LET K$= STR$ (K)
490 LET K= VAL (K$(1 TO 4))
500 PRINT
510 PRINT TAB 12;K
520 PRINT
530 IF K>1 THEN PRINT "A TOTAL ECLIPSE"
540 IF K<1 THEN PRINT "A PARTIAL ECLIPSE"
550 PRINT
560 GOTO 1000
570 LET L=2415036.025+(A*29.53058868)
580 LET L=L-(.406* SIN (D* PI /180))+(.174* SIN (C* PI /180))
590 LET L=L+( SIN (2*D* PI /180))/62
600 LET L= INT (L-( SIN (2*E* PI /180))/97)
610 IF L<2299161 THEN GOTO 670
620 LET L2= INT ((2299161-1867216.25)/36525.25)
630 LET L=L2+L
640 LET M= INT (L2/4)
650 LET L=L-M
660 LET L=L+1
670 LET N=L-1720995
680 LET O= INT ((N-122.1)/365.25)
690 LET P= INT (O*365.25)
700 LET Q= INT ((N-P)/30.6001)
710 LET R=((N-P)- INT (Q*30.6001))/10000
720 IF Q<13.7774 THEN GOTO 750
730 LET S=Q-12-1
740 GOTO 760
750 LET S=Q-1
760 LET T=S
770 LET T2=T+R
780 IF SQR (5)<S THEN GOTO 800
790 LET O=O+1
800 LET U=O+T2
810 LET R=3+R*10000
```


ECLIP (continued)

```
820 IF T <> 2 THEN GOTO 860
830 IF T=2 AND R>28 THEN LET T=T+1
840 IF T=2 AND R>28 THEN LET R=R-28
850 GOTO 910
860 IF T=3 AND R>31 THEN LET T=T+1
870 IF T=3 AND R>31 THEN LET R=R-31
880 CLS
890 PRINT
900 PRINT
910 PRINT "DATE OF ECLIPSE IS ..."
920 PRINT
930 PRINT TAB 20;"YEAR ";
940 PRINT O
950 PRINT TAB 20;"MONTH ";
960 PRINT T
970 PRINT TAB 20;"DAY  ";
980 PRINT R
990 GOTO 460
1000 PRINT
1010 PRINT "WANT COPY Y/N? ";
1020 INPUT Z$
1030 PRINT Z$
1040 IF Z$="N" THEN GOTO 1060
1050 COPY
1060 PRINT
1070 PRINT "DO YOU WANT NEXT ECLIPSE Y/N? ";
1080 INPUT A$
1090 PRINT A$
1100 PRINT
1110 IF A$="N" OR A$="NO" THEN GOTO 1210
1120 IF A$="Y" OR A$="YES" THEN LET FL=1
1130 CLS
1140 PRINT
1150 PRINT
1160 PRINT "PLEASE WAIT"
1170 IF FL=1 THEN GOTO 270
1180 PRINT
1190 PRINT "INVALID REPLY"
1200 GOTO 1070
1210 PRINT "DO YOU WANT ANOTHER YEAR Y/N? ";
1220 INPUT C$
1230 PRINT C$
1240 IF C$="N" OR C$="NO" THEN GOTO 1320
1250 IF C$="Y" OR C$="YES" THEN CLS
1260 IF C$="Y" OR C$="YES" THEN GOTO 150
1270 CLS
1280 PRINT
1290 PRINT "INVALID REPLY"
1300 PRINT
1310 GOTO 1210
1320 CLS
1330 STOP
```

Program 10: PHASE

Approximate Phase of Moon for Any Date

This program calculates the date of each new moon and then interpolates for the other phases. The date of the new moon is accurate within one day; the others are approximate only. The program changes the calendar date to a day number and then proceeds to calculate the next new moon from the date requested, making use of the synodic month (period from one new moon to the next). It derives the calendar date for this new moon and loops to calculate the calendar date for the next new moon. The program then interpolates for the other phases. It repeats this sequence for the number of months requested. Subroutines take care of month ends and year ends and the effects of leap years on the derived dates.

The program lists phases for any two months in any year requested. If the first new moon is toward the end of the month requested, you will need to ask for a month earlier to obtain all the phases within the requested month. An example of the display provided by this program is shown in Figure 10.1.

The listing for the PHASE program follows.



Figure 10.1: The PHASE program provides a table showing the approximate dates of the phases of the Moon for any period of two months.

PHASE

```
 10 REM PHASE
 20 PRINT
 30 REM LUNAR PHASE
 40 PRINT
 45 LET NS = 1
 50 PRINT
 60 PRINT
 70 LET FL=1
 80 LET XP=0
 90 PRINT "     ---------------------"
100 PRINT "     I   PHASES OF MOON  I"
110 PRINT "     ---------------------"
120 PRINT
130 PRINT
140 PRINT
150 PRINT "    BY ERIC BURGESS F.R.A.S."
160 PRINT
170 PRINT
180 PRINT
190 PRINT "     ALL RIGHTS RESERVED BY"
200 PRINT
210 PRINT "    S AND T SOFTWARE SERVICE"
220 PAUSE 200
```

PHASE (continued)

```
230 CLS
240 PRINT
250 PRINT
260 PRINT
270 PRINT "THIS PROGRM PROVIDES DATES FOR"
280 PRINT "PHASES OF THE MOON STARTING AT"
290 PRINT "ANY MONTH OR ANY DATE AND FOR"
300 PRINT "A PERIOD OF UP TO TWO MONTHS"
310 PRINT "        (WITHIN ONE DAY)"
320 PRINT
330 PRINT
340 PRINT "START DATE REQUIRED ";
350 PRINT "YEAR ";
360 INPUT Y$
370 PRINT Y$
380 LET Y= VAL (Y$)
390 PRINT
400 PRINT "MONTH ";
410 INPUT M$
420 PRINT M$
430 LET M= VAL (M$)
440 PRINT
450 LET D=0
460 PRINT "NUMBER OF MONTHS REQUIRED (1 OR 2) ";
470 INPUT N$
480 PRINT N$
490 LET N= VAL (N$)
500 IF N>2 THEN LET N=2
510 CLS
530 PRINT
540 PRINT
550 PRINT "PHASES OF MOON ARE"
560 LET MS=M
570 LET IN=0
580 LET YP=Y
590 LET TY=Y+M/100
600 IF 1582.10>TY THEN LET FL=2
610 LET P2=M
620 LET P1=Y
630 LET P3=D
640 LET P4=M/100+D/10000
650 IF SQR (5) <= P2 THEN GOTO 680
660 LET P1=P1-1
670 LET P2=P2+12
680 LET P5= INT (365.25*P1)+ INT ((P2+1)*30.6001)+P3+1720995
690 IF FL=2 THEN GOTO 720
700 LET P0= INT (P1/100)
710 LET P5=P5-P0+2+ INT (P0/4)
720 LET P6=.0338631922
730 LET Q=(P6*P5)+.67094
740 LET Q1=1-(Q- INT (Q))
750 LET P7=Q1/P6
760 LET P5=(P5+P7)*.985600267
770 LET P7=P7+(.1743*( SIN (.01745328*(P5+73.63))))
780 LET P8=(13.06499245*P5)+271.5
790 LET P7=P7-(.4089*( SIN (.01745328*(P8))))
800 LET P7=(( SIN (.01745328*(2*P8)))*.0161)+P7
810 LET P8=((P7-.5)/10000)+P4
820 IF IN <> 0 THEN GOTO 910
```


PHASE (continued)

```
830 LET P9=P8
840 LET IN=IN+1
850 LET M=M+1
860 LET P=0
870 LET P2=0
880 LET P3=0
890 LET P4=0
900 GOTO 590
910 LET M=M-1
920 LET Z=P9-P8
930 LET Z=Z*10000
940 LET Z= ABS (30.6001*Z)
950 LET PH=Z/400
960 LET PM= INT (P9*100)
970 LET PD= INT ((P9*100- INT (P9*100))*100)
990 IF PD <> 0 THEN GOTO 1020
1000 LET PM=PM-1
1010 LET PD=31
1020 IF PD <= 31 THEN GOTO 1090
1030 LET PD=PD-31
1040 IF PD>31 THEN GOTO 1030
1050 GOTO 1090
1060 LET PD=PD-6
1070 LET PM=PM+1
1080 LET Q=1
1090 IF PM <= 12 THEN GOTO 1120
1100 LET PM=PM-12
1110 LET Y=Y+1
1120 IF XP=1 THEN LET PM=PM-1
1130 GOSUB 1780
1140 PRINT
1150 PRINT "NEW MOON      * ";Y;" ";P$;" ";PD
1160 IF Q <> 1 THEN GOTO 1190
1170 LET PM=PM-1
1180 LET Q=0
1190 LET PD=PD+1
1200 IF PM=2 AND PD+PH>28 THEN LET PD=PD+3
1210 IF PD+PM <= 31 THEN GOTO 1240
1220 LET PD=PD-31
1230 LET PM=PM+1
1240 IF PM <= 12 THEN GOTO 1270
1250 LET PM=PM-12
1260 LET Y=Y+1
1270 IF PH- INT (PH)<.5 THEN LET PH= INT (PH)+1
1280 LET PH= INT (PH)
1290 GOSUB 1780
1300 PRINT
1310 PRINT "FIRST QUARTER ) ";Y;" ";P$;" ";PD+PH
1320 IF PM=2 AND PD+2*PH>28 THEN LET PD=PD+3
1330 IF PD+2*PH>31 THEN GOTO 1350
1340 GOTO 1370
1350 LET PD=PD-31
1360 LET PM=PM+1
1370 IF PM <= 12 THEN GOTO 1400
1380 LET PM=PM-12
1390 LET Y=Y+1
1400 GOSUB 1780
1410 PRINT
1420 PRINT "FULL MOON     0 ";Y;" ";P$;" ";PD+2*PH;
```

PHASE (continued)

```
1430 PRINT
1440 IF PM=2 AND PD+3*PH>28 THEN LET PD=PD+3
1450 IF PD+3*PH>31 THEN GOTO 1470
1460 GOTO 1490
1470 LET PD=PD-31
1480 LET PM=PM+1
1490 IF PM <= 12 THEN GOTO 1520
1500 LET PM=PM-12
1510 LET Y=Y+1
1520 GOSUB 1780
1530 IF XP=1 THEN LET M=M-1
1540 PRINT
1550 PRINT "LAST QUARTER  ( ";Y;" ";P$;" ";PD+3*PH
1560 PRINT 1
1570 IF PD+3*PH+7=30 OR PD+3*PH+7=31 THEN GOTO 1820
1580 LET Y=YP
1590 LET M=M+1
1600 IF NS >= N THEN GOTO 1650
1610 LET NS=NS+1
1620 LET IN=0
1630 LET P9=0
1640 GOTO 570
1650 PRINT "WANT COPY Y/N? ";
1660 INPUT Z$
1670 PRINT Z$
1680 IF Z$="N" THEN GOTO 1700
1690 COPY
1700 CLS
1710 PRINT
1720 PRINT "DO YOU WANT MORE Y/N? ";
1730 INPUT A$
1740 IF A$="N" THEN GOTO 1760
1750 GOTO 230
1760 CLS
1770 STOP
1780 LET PQ=PM*3
1790 LET M$="JANFEBMARAPRMAYJUNJULAUGSEPOCTNOVDEC"
1800 LET P$=M$(PQ-2 TO PQ)
1810 RETURN
1820 PRINT
1830 LET PZ=30
1840 PRINT "NEW MOON      * ";Y;" ";P$;" ";PZ-23
1850 IF NS=N THEN GOTO 1990
1860 LET PM=PM+1
1870 GOSUB 1780
1875 PRINT
1880 PRINT "FIRST QUARTER ) ";Y;" ";P$;" ";PZ-23
1885 PRINT
1890 PRINT "FULL MOON     O ";Y;" ";P$;PZ-15
1895 PRINT
1900 PRINT "LAST QUARTER  ( ";Y;" ";P$;PZ-8
1910 GOTO 1650
1920 LET NF=1
1930 LET PZ=30-1
1940 PRINT
1950 LET NS=NS+1
1960 GOTO 1840
1970 LET PZ=0
1980 LET PD=0
```


PHASE (continued)

```
1990 LET XP=1
2000 LET N=N+1
2010 PRINT NS
2020 GOTO 1580
```

PART 3



THE PLANETS

Planets were known as wandering stars to the ancient Greeks, because they appear to move relatively quickly with respect to the fixed background of stars. Observations of the motions of the planets led to new explanations. These new theories of planetary motion eventually caused the demise of the dogma of an Earth-centered universe and spurred the Copernican revolution of human thought. This was part of the transformation that ultimately resulted in the Industrial Revolution. Continued developments in physics and astronomy played an important part in accelerating the pace of the Industrial Revolution.

The programs contained in this section all deal with the motions of the planets. There are programs to find the planets among the stars of the zodiac, programs that determine when planets are in certain positions, such as when Mars is closest to Earth, and programs to aid in planetary observations.

Program 11: RADEC

Right Ascension and Declination for All Planets for Any Date

This program computes right ascensions and declinations of the planets Mercury, Venus, Mars, Jupiter, Saturn, Uranus, Neptune, and Pluto for any requested date. An example of the display provided by the program is shown in Figure 11.1.

Calculations are based on the 1960 epoch (about midcentury to allow forward and backward calculations without perturbations becoming too large). The program calculates the number of days from the epoch to the requested date. It uses planetary orbital data from the data strings, in which angles are expressed in radians. These data strings provide information about each planet at the epoch; such as average motion per day, position at the epoch, eccentricity of the orbit, longitude of the perihelion, length in astronomical units of the semi-major axis, inclination of the orbit, and longitude of the ascending node. The program uses standard trigonometrical formulas to derive for each planet the current heliocentric longitude, the distance from the Sun in astronomical units, the angular distance above or below the ecliptic plane, and the distance in astronomical units from Earth. The angular distance from the Sun is then derived from the three sides of the triangle of distances of Earth, planet, and Sun. From this are derived the geocentric coordinates of each planet. These are then converted to right ascension and declination by changing degrees (measured from the first point of Aries) to hours of right ascension, and by using the inclination of Earth's axis to derive declination (angular distance north or south of the celestial equator). String functions are used to prepare the data for display to an appropriate number of decimal places.

The listing of the RADEC program follows.


```
PLANETARY DATA FOR.. 1983 10 1
          HEL.  DIST.   R.A.  DEC.
          LONG.  A.U.   HRS   DEG.
MERCURY   75     0.935  11.27  7.714
VENUS     29     0.431  9.813  10.21
MARS      129    2.347  10.02  13.37
JUPITER   255    5.750  16.41  -20.9
SATURN    216    10.74  14.24  -10.5
URANUS    248    19.54  16.36  -21.2
NEPTUNE   267    30.42  17.73  -22.1
PLUTO     211    30.49  14.07  4.677

WANT COPY Y/N?

"■"
```

Figure 11.1: The program RADEC provides a table of right ascension and declination of all the planets for any requested date.

RADEC

```
 10 SLOW
 20 CLS
 30 CLEAR
 40 DIM A$(10)
 50 DIM U(9)
 60 DIM P(9,9)
 70 DIM N$(22)
 80 DIM M$(63)
 90 DIM P$(10,7)
100 REM RADEC 4/14/83
110 DIM A(9)
120 DIM D(9)
130 DIM L(9)
140 DIM Q(9)
150 DIM R(9)
160 DIM V(9)
170 DIM K(9)
180 LET FL=0
190 PRINT AT 5,5;"-----------------"
200 PRINT AT 6,5;"I     RADEC     I"
210 PRINT AT 7,5;"-----------------"
220 PRINT AT 9,5;"ASTRONOMY PROGRAM"
230 PRINT AT 10,3;"BY ERIC BURGESS F.R.A.S."
240 PRINT AT 12,4;"ALL RIGHTS RESERVED BY"
```

RADEC (continued)

```
250 PRINT AT 13,3;"S AND T SOFTWARE SERVICE"
260 PAUSE 350
270 CLS
280 PRINT
290 PRINT
300 PRINT
310 PRINT TAB 5;"THIS PROGRAM GIVES"
320 PRINT TAB 6;"RIGHT ASCENSIONS"
330 PRINT TAB 6;"AND DECLINATIONS"
340 PRINT TAB 2;"HELIOCENTRIC LONGITUDES"
350 PRINT TAB 9;"AND THE"
360 PRINT TAB 3;"DISTANCES FROM EARTH"
370 PRINT TAB 8;"OF ALL THE"
380 PRINT TAB 9;"P_ANETS"
390 PRINT "FOR THE DATE WHICH YOJ REQUEST"
400 PRINT
410 PRINT
420 PAUSE 100
430 PRINT
440 PRINT
450 PRINT
460 PRINT "PLEASE WAIT; LOADING ARRAYS"
470 GOSUB 1110
480 CLS
490 PRINT
500 PRINT
510 PRINT "ENTER THE DATE"
520 PRINT
530 PRINT "       YEAR.... ";
540 INPUT Y
550 PRINT Y
560 PRINT
570 PRINT "       MONTH... ";
580 INPUT M
590 PRINT M
600 IF M<1 OR M>12 THEN GOTO 570
610 PRINT
620 PRINT "       DAY..... ";
630 INPUT DY
640 PRINT DY
650 IF DY>31 OR DY<1 THEN GOTO 620
660 IF M=2 AND DY>29 THEN GOTO 620
670 PRINT
680 LET DG=365*Y+DY+((M-1)*31)
690 IF M >= 3 THEN GOTO 720
700 LET DG=DG+ INT ((Y-1)/4)- INT ((.75)* INT ((Y-1)/100+1))
710 GOTO 740
720 LET DG=DG- INT (M*.4+2.3)+ INT (Y/4)
730 LET DG=DG- INT ((.75)* INT ((Y/100)+1))
740 LET NI=DG-715875
750 CLS
760 FAST
770 GOSUB 1420
780 SLOW
790 PRINT
800 PRINT "PLANETARY DATA FOR.. ";Y;" ";M;" ";DY
810 PRINT "-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-"
```


RADEC (continued)

```
820 PRINT "        HEL. DIST.  R.A. DEC."
830 PRINT "        LONG. A.U.  HRS  DEG."
840 PRINT "-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-"
850 FOR I=1 TO 9
860 IF I=3 THEN NEXT I
870 PRINT P$(I);
880 LET A(I)=A(I)*57.29578
890 LET A$= STR$ (A(I))
900 LET HL= VAL A$
910 PRINT TAB 8; INT (HL);
920 LET A$= STR$ (Q(I))
930 PRINT TAB 13;A$( TO 5);
940 LET A$= STR$ (R(I))
950 PRINT TAB 19;A$( TO 5);
960 LET A$= STR$ (V(I))
970 PRINT TAB 25;A$( TO 5)
980 NEXT I
990 PRINT "-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-"
1000 PRINT
1010 PRINT "WANT COPY Y/N? "
1020 INPUT C$
1030 IF C$="N" OR C$="NO" THEN GOTO 1060
1040 PRINT AT 16,0;"                    "
1050 COPY
1060 CLS
1070 PRINT
1080 PRINT
1090 PRINT "WANT ANOTHER DATE Y/N? "
1100 GOTO 2070
1110 REM PLANETARY DATA
1120 LET U=1
1130 LET N$=".0714223.84840.3883011"
1140 LET M$=N$+".34041.387100.0797402.73514.122173.836013"
1150 GOSUB 1720
1160 LET N$=".0279623.02812.0131952"
1170 LET M$=N$+".28638.723300.0050603.85017.0593411.33168"
1180 GOSUB 1720
1190 LET N$=".0172021.74022.0320441"
1200 LET M$=N$+".785471.00000.0170003.339260000000000000"
1210 GOSUB 1720
1220 LET N$=".0091464.51234.1753015"
1230 LET M$=N$+".852091.52370.1417041.04656.031420.858702"
1240 GOSUB 1720
1250 LET N$=".0014504.53364.090478."
1260 LET M$=N$+"2391105.20280.2493741.76188.0197201.74533"
1270 GOSUB 1720
1280 LET N$=".0005844.89884.1055581"
1290 LET M$=N$+".610949.53850.5341563.12570.0436331.97746"
1300 GOSUB 1720
1310 LET N$=".0002052.46615.0885932"
1320 LET M$=N$+".9670619.1820.9015544.49084.0139601.28805"
1330 GOSUB 1720
1340 LET N$=".0001043.78556.016965."
1350 LET M$=N$+"77318130.0600.2705402.33498.0314162.29162"
1360 GOSUB 1720
1370 LET N$=".0000693.16948.4712393"
1380 LET M$=N$+".9130339.44009.860005.23114.3001971.91812"
1390 GOSUB 1720
```

RADEC (continued)

```
1400 LET M$=""
1410 RETURN
1420 REM PLANET NAMES
1430 IF FL=1 THEN GOTO 1540
1440 LET P$(1)="MERCURY"
1450 LET P$(2)="VENUS"
1460 LET P$(3)="SUN"
1470 LET P$(4)="MARS"
1480 LET P$(5)="JUPITER"
1490 LET P$(6)="SATURN"
1500 LET P$(7)="URANUS"
1510 LET P$(8)="NEPTUNE"
1520 LET P$(9)="PLUTO"
1530 LET P$(10)="MOON"
1540 LET I=1
1550 FOR J=1 TO 9
1560 GOSUB 1770
1570 LET A(I)=A
1580 LET D(I)=D
1590 LET L(I)=L
1600 LET I=I+1
1610 NEXT J
1620 FOR I=1 TO 9
1630 IF I=3 THEN NEXT I
1640 GOSUB 1890
1650 LET Q(I)=Q
1660 LET R(I)=R
1670 LET V(I)=V
1680 NEXT I
1690 FOR I=1 TO 9
1700 IF I=3 THEN NEXT I
1710 RETURN
1720 FOR K=1 TO 9
1730 LET P(U,K)= VAL (M$(1+((K-1)*7) TO (K*7)))
1740 NEXT K
1750 LET U=U+1
1760 RETURN
1770 LET A=NI*P(J,1)+P(J,2)
1780 IF A>6.28318 THEN LET A=((A/6.28318)- INT
     (A/6.28318))*6.28318
1790 IF A<0 THEN LET A=A+6.28318
1800 IF A<0 THEN GOTO 1790
1810 LET C=P(J,3)* SIN (A-P(J,4))
1820 LET A=A+C
1830 IF A>6.28318 THEN LET A=A-6.82318
1840 IF A<0 THEN LET A=A+6.82318
1850 IF A<0 THEN GOTO 1840
1860 LET D=P(J,5)+P(J,6)* SIN (A-P(J,7))
1870 LET L=P(J,8)* SIN (A-P(J,9))
1880 RETURN
1890 LET Z=A(3)-A(I)
1900 IF ABS (Z)>3.14159 AND Z<0 THEN LET Z=Z+6.28318
1910 IF ABS (Z)>3.14159 AND Z>0 THEN LET Z=Z-6.28318
1920 LET Q= SQR (D(I) ** 2+D(3) ** 2-2*D(I)*D(3)* COS (Z))
1930 LET P=(D(I)+D(3)+Q)/2
1940 LET X=2* ACS ( SQR (((P*(P-D(I)))/(D(3)*Q))))
1950 IF Z<0 THEN LET R=57.29578*(A(3)+3.14159-X)/15
```


RADEC (continued)

```
1960 IF Z>0 THEN LET R=57.29578*(A(3)+3.14159+X)/15
1970 IF R>24 THEN LET R=R-24
1980 IF R>24 THEN GOTO 1970
1990 IF R<-24 THEN LET R=R+24
2000 IF R<-24 THEN GOTO 1990
2010 IF R<0 THEN LET R=R+24
2020 IF R<0 THEN GOTO 2010
2030 IF Z<0 THEN LET V=
     SIN (A(3)+3.14159-X)*23.44194+57.29578*L(I)
2040 IF Z>0 THEN LET V=
     SIN (A(3)+3.14159+X)*23.44194+57.29578*L(I)
2050 LET X=57.29578*X
2060 RETURN
2070 INPUT B$
2080 IF B$="Y" THEN GOTO 2100
2090 GOTO 2120
2100 LET FL=1
2110 GOTO 480
2120 CLS
2130 STOP
```

Program 12: MARSP

Angular Diameter and Distance of Mars for Any Date, and Next Opposition

This program offers two options: you can find the angular diameter of Mars at any date, or you can find the next opposition following any date. Both result in a similar display of the angular diameter of Mars, its distance from Earth, its angular distance from the Sun, and a comparison of its angular diameter with the greatest and smallest possible angular diameters. Figure 12.1 shows the detailed display of information given by the program.

When you choose to find the next opposition following any date, the program counts down dates toward the opposition. While doing so it displays a graphic plot showing the Sun, Earth, and Mars relative to the direction of the vernal equinox (first point of Aries). It also shows the movement of the two planets around the Sun toward the approaching opposition, to the point where they become aligned on the same side of the Sun at the opposition.

To accomplish this, the program first calculates the positions of Mars and Earth at the beginning date. The calculations are made in terms of heliocentric longitude, which is the angle between the planet and the first point of Aries, measured at the Sun. The plot positions are derived from the heliocentric longitudes of Earth and Mars (AE and MA), and they are converted to a suitable scale for the monitor screen. The display appears as though seen from high above the Sun.

The program next compares the heliocentric longitudes of Mars and Earth. If they differ by more than 120 degrees, the program jumps a month, recalculates the positions, and replots, erasing the first plot. When the difference in longitude of the two planets is less than 120 degrees, the increments of time are reduced to five days. As the angle between the planets narrows still further, the increments are reduced to two days and finally to one day. When the two longitudes are within two degrees of each other, the program

declares an opposition and calculates the right ascension of Mars. The program then ascertains in which zodiacal constellation the opposition occurs and prints the date and the name of the constellation. The program then calculates more details about the opposition and displays them (see Figure 12.1).

The listing of the MARSP program follows.

```
FOR YEAR 1983 MONTH 5 DAY 15
------------------------------
DISTANCE OF MARS IS 2.53 A.U.
    OR 235. MILLION MILES

ANGLE FROM SUN IS 4.57 DEG

DIAMETER OF MARS IS 3.677 ARCSEC
MAX AT CLOSEST IS 25 ARCSEC

COMPARISONS ARE.....

MAX. DIAM AT CLOSEST OPPOSITION
(-------------------------)
DIAM AT DATE
(      )
MIN. DIAM AT CONJUNCTION
(-)

WANT COPY Y/N?
```

Figure 12.1: The MARSP program will display the information shown here for any date, or it will find the next opposition of Mars and display the information for that date.

MARSP

```
 10 LET LY=0
 20 LET FL=0
 30 REM MARSP 4/15/83 VERSION
 40 LET P1= PI
 50 LET P2=6.28318
 60 CLS
 70 LET I$="INVALID RESPONSE"
 80 PRINT AT 5,6;"----------------"
 90 PRINT AT 6,6;"I     MARS     I"
100 PRINT AT 7,6;"----------------"
```

MARSP (continued)

```
110 PRINT AT 9,4;"AN ASTRONOMY PROGRAM"
120 PRINT AT 9,3;"BY ERIC BURGESS F.R.A.S."
130 PRINT AT 11,4;"ALL RIGHTS RESERVED BY"
140 PRINT AT 12,3;"S AND T SOFTWARE SERVICE"
150 PRINT AT 15,1;"DO YOU WANT INSTRUCTIONS Y/N? ";
160 INPUT A$
170 PRINT A$
180 IF A$="N" OR A$="NO" THEN GOTO 360
190 IF A$="Y" OR A$="YES" THEN GOTO 230
200 PRINT I$
210 CLS
220 GOTO 150
230 CLS
240 PRINT
250 PRINT
260 PRINT
270 PRINT "THIS PROGRAM OFFERS TWO CHOICES"
280 PRINT
290 PRINT "1)CALCULATES APPROXIMATE"
300 PRINT "  ANGULAR DIAMETER"
310 PRINT "  OF MARS FOR ANY DATE"
320 PRINT
330 PRINT "2)GIVES APPROXIMATE DATE"
340 PRINT "  OF NEXT OPPOSITION"
350 PRINT "  AFTER ANY DATE"
360 PRINT
370 PRINT
380 PRINT "SELECT 1)ANGULAR DIAMETER"
390 PRINT
400 PRINT "        2)NEXT OPPOSITION"
410 INPUT B$
420 CLS
430 IF B$="1" THEN GOTO 490
440 IF B$="2" THEN GOTO 1930
450 PRINT B$;"IS";I$
460 PAUSE 100
470 CLS
480 GOTO 360
490 CLS
500 GOSUB 1030
510 CLS
520 LET M$= STR$ (9.31/Q)
530 LET MX= VAL (M$(1 TO 5))
540 PRINT
550 SLOW
560 PRINT "FOR YEAR ";Y;" MONTH ";M;" DAY ";D
570 PRINT "------------------------------"
580 LET Q$= STR$ (Q)
590 PRINT "DISTANCE OF MARS IS ";Q$(1 TO 4);" A.U."
600 LET Q$= STR$ (Q*92.96)
610 LET Q$=Q$(1 TO 4)
620 PRINT "     OR ";Q$;" MILLION MILES"
630 PRINT
640 LET W$= STR$ (W*57.29578)
650 LET W$=W$(1 TO 4)
660 PRINT "ANGLE FROM SUN IS ";W$;" DEG"
670 PRINT
```


MARSP (continued)

```
680 IF VAL (M$)>25 THEN LET D$="25"
690 PRINT "DIAMETER OF MARS IS ";MX;" ARCSEC"
700 PRINT "MAX AT CLOSEST IS 25 ARCSEC"
710 PRINT
720 PRINT "COMPARISONS ARE....."
730 PRINT
740 PRINT "MAX. DIAM AT CLOSEST OPPOSITION"
750 PRINT "(------------------------)"
760 LET QD=9.36/Q
770 IF QD>25 THEN LET OP=24
780 IF QD<26 THEN LET OP=QD
790 PRINT "DIAM AT DATE"
800 PRINT "("; TAB 1+OP;")"
810 PRINT "MIN. DIAM AT CONJUNCTION"
820 PRINT "(-)"
830 PRINT
840 PRINT "WANT COPY Y/N? ";
850 INPUT X$
860 IF X$="N" OR X$="NO" THEN GOTO 880
870 COPY
880 CLS
890 PRINT
900 PRINT
910 PRINT "WANT ANOTHER DATE Y/N?";
920 INPUT A$
930 IF A$="N" OR A$="NO" THEN GOTO 970
940 IF A$="Y" OR A$="YES" THEN GOTO 990
950 PRINT E$
960 GOTO 830
970 CLS
980 STOP
990 CLS
1000 LET FL=0
1010 IF B$="1" THEN GOTO 490
1020 GOTO 1930
1030 GOSUB 1330
1040 GOSUB 1710
1050 LET AE=NI*.017202+1.74022
1060 IF AE>P2 THEN LET AE=((AE/P2)- INT (AE/P2))*P2
1070 IF AE >= 0 THEN GOTO 1100
1080 IF AE<0 THEN LET AE=AE+P2
1090 GOTO 1070
1100 LET CE=.032044* SIN (AE-1.78547)
1110 LET AE=AE+CE
1120 IF AE>P2 THEN LET AE=AE-P2
1130 IF AE>0 THEN GOTO 1160
1140 LET AE=AE+P2
1150 GOTO 1130
1160 LET DE=1+.017* SIN (AE-3.33926)
1170 GOSUB 1790
1180 LET Z=AE-MA
1190 IF ABS (Z)>P1 AND Z<0 THEN LET Z=Z+P2
1200 IF ABS (Z)>P1 AND Z>0 THEN LET Z=Z-P2
1210 IF Z <> 0 THEN GOTO 1250
1220 LET Q=MD-DE
1230 LET W=P1
```

MARSP (continued)

```
1240 GOTO 1280
1250 IF Z <> P1 THEN GOTO 1280
1260 IF Z=P1 THEN LET Q=MD+DE
1270 IF Z=P1 THEN LET W=Q
1280 LET Q= SQR (MD ** 2+DE ** 2-2*MD*DE* COS (Z))
1290 LET P=(MD+DE+Q)/2
1300 LET V= SQR ((P*(P-MD))/(DE*Q))
1310 LET W=2* ACS (V)
1320 RETURN
1330 IF B$="2" AND FL=1 THEN GOTO 1670
1340 PRINT
1350 PRINT
1360 PRINT "ENTER THE DATE"
1370 PRINT
1380 PRINT TAB (4);"THE YEAR...";
1390 INPUT Y$
1400 LET Y= VAL (Y$)
1410 PRINT Y
1420 GOSUB 1440
1430 GOTO 1460
1440 IF Y/4- INT (Y/4)=0 AND Y/100- INT (Y/100)
     <> 0 THEN LET LY=1
1450 RETURN
1460 PRINT
1470 PRINT TAB (4);"THE MONTH..";
1480 INPUT M$
1490 LET M= VAL (M$)
1500 PRINT M
1510 IF M<1 OR M>12 THEN GOTO 1470
1520 PRINT
1530 PRINT TAB (4);"THE DAY....";
1540 INPUT D$
1550 LET D= VAL (D$)
1560 PRINT D
1570 IF D<1 OR D>31 THEN GOTO 1530
1580 IF M=2 AND LY=1 AND D>29 THEN GOTO 1530
1590 IF M=2 AND LY=0 AND D>28 THEN GOTO 1530
1600 PRINT
1610 PRINT
1620 PRINT "SCREEN WILL BE BLANK FOR A WHILE"
1630 PRINT "     PLEASE WAIT..."
1640 PAUSE 100
1650 FAST
1660 CLS
1670 LET YI=Y
1680 LET MI=M
1690 LET DI=D
1700 RETURN
1710 LET DG=365*Y+D+((M-1)*31)
1720 IF M >= 3 THEN GOTO 1750
1730 LET DG=DG+ INT ((Y-1)/4)- INT ((.75)* INT ((Y-1)/100+1))
1740 GOTO 1770
1750 LET DG=DG- INT (M*.4+2.3)+ INT (Y/4)
1760 LET DG=DG- INT ((.75)* INT ((Y/100+1)))
1770 LET NI=DG-715875
1780 RETURN
1790 LET MA=NI*.009146+4.51234
```


MARSP (continued)

```
1800 IF MA>P2 THEN LET MA=((MA/P2)- INT (MA/P2))*P2
1810 IF M >= 0 THEN GOTO 1840
1820 LET MA=MA+P2
1830 GOTO 1810
1840 LET MC=.175301* SIN (MA-5.85209)
1850 LET MA=MA+MC
1860 IF MA>P2 THEN LET MA=MA-P2
1870 IF MA>0 THEN GOTO 1900
1880 LET MA=MA+P2
1890 GOTO 1870
1900 LET MD=1.5237+.141704* SIN (MA-1.04656)
1910 RETURN
1920 FAST
1930 GOSUB 1330
1940 GOSUB 1710
1950 GOSUB 1050
1960 CLS
1970 SLOW
1980 PRINT AT 1,0;YI;" ";MI;" ";DI
1990 GOSUB 2870
2000 LET FL=1
2010 LET DS=W*57.29578
2020 IF DS>120 THEN GOTO 2090
2030 LET FL=1
2040 LET M=M+1
2050 IF M>12 THEN LET Y=Y+1
2060 IF M>12 THEN LET M=1
2070 GOSUB 1440
2080 GOTO 1930
2090 IF DS<160 THEN LET D=D+5
2100 IF DS<160 THEN GOTO 2140
2110 IF DS<170 THEN LET D=D+2
2120 IF DS<170 THEN GOTO 2140
2130 LET D=D+1
2140 GOSUB 2270
2150 LET D2=DS
2160 IF DS <= 178 THEN GOTO 2220
2170 PRINT AT 20,0;"OPPOSITION.. WANT COPY Y/N? ";
2180 INPUT J$
2190 IF J$="N" OR J$="NO" THEN GOTO 2210
2200 COPY
2210 GOTO 2490
2220 IF DS<DS THEN GOTO 2240
2230 GOTO 2420
2240 LET D=D+1
2250 GOSUB 2270
2260 GOTO 2230
2270 IF LY=1 AND M=2 AND D>29 THEN GOTO 2380
2280 IF LY=0 AND M=2 AND D>28 THEN GOTO 2400
2290 IF D<31 THEN GOTO 2320
2300 IF M=4 OR M=6 OR M=9 OR M=11 THEN GOTO 2360
2310 IF D>31 THEN GOTO 2330
2320 RETURN
2330 LET D=D-31
2340 LET M=M+1
2350 RETURN
2360 LET D=D-30
```

MARSP (continued)

```
2370 GOTO 2340
2380 LET D=D-29
2390 GOTO 2340
2400 LET D=D-28
2410 GOTO 2340
2420 GOTO 1930
2430 IF D>31 THEN GOTO 2450
2440 RETURN
2450 LET M=M+1
2460 LET D=D-31
2470 RETURN
2480 GOTO 1930
2490 IF Z<0 THEN GOTO 2520
2500 IF Z>0 THEN LET R=AE+P1+W
2510 GOTO 2530
2520 LET R=AE+P1+W
2530 LET R=57.29578*R/15
2540 IF R>24 THEN GOTO 2560
2550 GOTO 2580
2560 LET R=R-24
2570 GOTO 2540
2580 IF R>-24 THEN GOTO 2610
2590 LET R=R+24
2600 GOTO 2580
2610 IF R<0 THEN LET R=R+24
2620 IF R>0 AND R<2 THEN LET C$="PISCES"
2630 IF R>2 AND R<4 THEN LET C$="ARIES"
2640 IF R>4 AND R<6 THEN LET C$="TAURUS"
2650 IF R>6 AND R<8 THEN LET C$="GEMINI"
2660 IF R>8 AND R<10 THEN LET C$="CANCER"
2670 IF R>10 AND R<12 THEN LET C$="LEO"
2680 IF R>12 AND R<14 THEN LET C$="VIRGO"
2690 IF R>14 AND R<16 THEN LET C$="LIBRA"
2700 IF R>16 AND R<18 THEN LET C$="SCORPIO"
2710 IF R>18 AND R<20 THEN LET C$="SAGITTARIUS"
2720 IF R>20 AND R<22 THEN LET C$="CAPRICORNUS"
2730 IF R>22 AND R<24 THEN LET C$="AQUARIUS"
2740 CLS
2750 PRINT
2760 PRINT "NEXT OPPOSITION OF MARS IS"
2770 PRINT "       ",YI;" ";MI;" ";DI
2780 PRINT "IN THE CONSTELLATION OF ";C$
2790 PRINT
2800 PRINT "FOR DETAILS PRESS ANY KEY"
2810 IF INKEY$ ="" THEN GOTO 2810
2820 LET Y=YI
2830 LET M=MI
2840 LET D=DI
2850 FAST
2860 GOTO 500
2870 LET AD=7
2880 LET DM=10
2890 LET EY=AD* SIN (AE)
2900 LET EX=AD* COS (AE)
2910 LET MY=DM* SIN (MA)
2920 LET MX=DM* COS (MA)
2930 REM PLOT ORBITS
```


MARSP (continued)

```
2940 LET EX=11+EX
2950 LET EY=15+EY
2960 LET MX=MX+11
2970 LET MY=MY+15
2980 PRINT AT EX,EY;"E"
2990 PRINT AT MX,MY;"M"
3000 PRINT AT 11,11;"SUN(*)"
3010 PRINT AT 11,21;"---- ARIES>"
3020 RETURN
```

Program 13: MERVE

Elongations, Phases, Angular Diameters, and Distances of Mercury and Venus, and Next Elongations

This program offers two alternatives. It will provide information on Mercury's and Venus's distances from Earth, angular sizes, and phases for any date (see Figure 13.1). It will also calculate the date of the next elongation of Mercury or Venus following any date.

When you select the alternative to seek the next elongation, the display counts down in dates toward the elongation and shows

```
FOR YEAR 1983 MONTH 6 DAY 16
----------------------------
DIST. OF MERCURY 0.98 A.U.
         OR 91.32 MILLION MILES
DISTANCE OF VENUS 0.71 A.U.
         OR 66.13 MILLION MILES

MERCURY ANG.DIST.TO SUN 22.1 DEG
VENUS ANG.DIST. TO SUN -45. DEG

MERCURY ANG. DIAM. 6.795 ARCSEC
MAX. AT INF. CONJ. IS 12 ARCSEC

VENUS ANG.DIAM. 23.62 ARCSEC
MAX. AT INF. CONJ. IS 61 ARCEC

MERCURY PHASE NEAR HALF 96 DEG.
VENUS PHASE NEAR HALF 89 DEG

ANOTHER DATE? Y/N
```

Figure 13.1: Details of either or both planets can be displayed by the MERVE program.

the positions of Mercury and Venus relative to the Sun. On arrival at the elongation, you can ask for more details, such as distance from Earth, phase, angular distance from the Sun, and angular diameter of the planet at the date of elongation (see Figure 13.2).

Using similar routines to those used in earlier programs, the program determines the angular distance of the planets from the Sun. It finds the angular distance of the planet for which you seek the next elongation, repeats the calculation for a later date, and searches for a maximum angular distance from the Sun by repeated comparisons following increments of the date. The next elongation is identified as the first maximum of angular distance to be encountered after the start date. Next the program determines if this is east or west of the Sun. For the date of the elongation it then calculates the other parameters for the display and develops a graphic display of the Sun with the planets configured alongside as they appear from Earth.

```
NEXT ELONGATION OF MERCURY
      IS 1983 8 19

AND THE PLANET IS....
      EAST OF SUN (EVENING STAR)

CONFIGURATION ON 1983 8 19
V=VENUS, M=MERCURY, 0=SUN

        M       V   0


FOR DETAILS PRESS ANY KEY
```

Figure 13.2: Information on the next elongation of Mercury or Venus is displayed by the MERVE program

You can speed up the program by adding other discriminating lines to the routines that seek maxima. But you will have to accept trade-offs between speed in finding the elongation and accuracy in the date of elongation. This is especially true of rapidly moving Mercury with its relatively eccentric orbit and wide range of elongation angles.

The listing for the MERVE program follows.

MERVE

```
 10 REM MERVE 4/16/83
 20 LET E$=" "
 30 LET V2=0
 40 LET D2=0
 50 LET P2=2* PI
 60 PRINT AT 4,6;"ASTRONOMY PROGRAM"
 70 PRINT AT 6,5;"--------------------"
 80 PRINT AT 7,5;"I  INNER PLANETS  I"
 90 PRINT AT 8,5;"--------------------"
100 PRINT AT 10,3;"BY ERIC BURGESS F.R.A.S."
110 PRINT AT 12,4;"ALL RIGHTS RESERVED BY"
120 PRINT AT 13,3;"S AND T SOFTWARE SERVICE"
130 PRINT
140 PRINT
150 PRINT
160 PRINT
170 PRINT "WANT INSTRUCTIONS? Y/N "
180 INPUT A$
190 IF A$="N" OR A$="NO" THEN GOTO 390
200 CLS
210 PRINT
220 PRINT "THIS PROGRAM OFFERS TWO CHOICES.."
230 PRINT
240 PRINT "1) CALCULATES APPROX. ANGULAR"
250 PRINT "   DISTANCES OF MERCURY AND"
260 PRINT "   VENUS FROM THE SUN, AND"
270 PRINT "   APPROX. DISTANCES FROM EARTH,"
280 PRINT "   AND PROVIDES APPROX."
290 PRINT "   ANGULAR DIAMETER AND PHASE"
300 PRINT "   FOR ANY DATE."
310 PRINT
320 PRINT " 2) PROVIDES DATE OF THE NEXT"
330 PRINT "   ELONGATION OF MERCURY"
340 PRINT "   OR VENUS AFTER ANY DATE"
350 PRINT
360 PRINT
370 PRINT "PRESS ANY KEY"
380 IF INKEY$ ="" THEN GOTO 380
390 CLS
400 PRINT
410 PRINT
420 PRINT "SELECT 1) ANGULAR DISTANCES"
430 PRINT "       2) NEXT ELONGATION"
440 INPUT B$
450 IF B$="1" THEN GOTO 520
460 IF B$="2" THEN GOTO 2280
470 PRINT
480 PRINT
```


MERVE (continued)

```
490 PRINT "PICK 1 OR 2"
500 PAUSE 100
510 GOTO 390
520 GOSUB 1040
530 CLS
540 GOSUB 560
550 GOTO 960
560 LET J$=( STR$ (6.68/MQ))(1 TO 5)
570 LET V$=( STR$ (16.82/VQ))(1 TO 5)
580 PRINT "FOR YEAR ";Y;" MONTH ";M;" DAY ";D
590 PRINT "---------------------------"
600 LET K$=( STR$ (MQ))(1 TO 4)
610 LET U$=( STR$ (VQ))(1 TO 4)
620 IF E$="2" THEN GOTO 670
630 PRINT "DIST. OF MERCURY ";K$;" A.U."
640 PRINT TAB 10;"OR "; VAL (( STR$ (MQ*92.9))(1 TO 5));
650 PRINT " MILLION MILES"
660 IF E$="1" THEN GOTO 700
670 PRINT "DISTANCE OF VENUS ";U$;" A.U."
680 PRINT TAB 10;"OR "; VAL (( STR$ (VQ*92.9))(1 TO 5));
690 PRINT " MILLION MILES"
700 PRINT
710 LET L$=( STR$ (MW*57.29578))(1 TO 4)
720 LET W$=( STR$ (VW*57.29578))(1 TO 4)
730 IF E$="2" THEN GOTO 760
740 PRINT "MERCURY ANG.DIST.TO SUN ";L$;" DEG"
750 IF E$="1" THEN GOTO 780
760 PRINT "VENUS ANG.DIST. TO SUN ";W$;" DEG"
770 PRINT
780 IF E$="2" THEN GOTO 830
790 PRINT "MERCURY ANG. DIAM. ";J$;" ARCSEC"
800 PRINT "MAX. AT INF. CONJ. IS 12 ARCSEC"
810 IF E$="1" THEN GOTO 850
820 PRINT
830 PRINT "VENUS ANG.DIAM. ";V$;" ARCSEC"
840 PRINT "MAX. AT INF. CONJ. IS 61 ARCEC"
850 PRINT
860 LET PH= ABS (MW)+ ABS (MZ)
870 GOSUB 3700
880 IF E$="2" THEN GOTO 910
890 PRINT "MERCURY PHASE ";H$;" ";PH;" DEG."
900 IF E$="1" THEN GOTO 960
910 LET PH= ABS (VW)+ ABS (ZV)
920 GOSUB 3700
930 PRINT "VENUS PHASE ";H$;" ";PH;" DEG'
940 IF E$="2" THEN GOTO 960
950 PRINT
960 IF B$="2" THEN RETURN
970 PRINT "ANOTHER DATE? Y/N ";
980 INPUT A$
990 IF A$="N" THEN GOTO 1020
1000 CLS
1010 GOTO 520
1020 CLS
1030 STOP
1040 REM ANG.DIAM.
1050 CLS
1060 PRINT
1070 PRINT
1080 GOSUB 1370
1090 GOSUB 1610
1100 REM DIT. FROM EARTH
```

MERVE (continued)

```
1110 LET AE=NI*.017202+1.74022
1120 IF AE>P2 THEN LET AE=((AE/P2)- INT (AE/P2))*P2
1130 GOTO 1160
1140 IF AE<0 THEN LET AE=AE+2* PI
1150 IF AE<0 THEN GOTO 1140
1160 LET CE=.032044* SIN (AE-1.78547)
1170 LET AE=AE+CE
1180 IF AE>2* PI THEN LET AE=AE-2* PI
1190 IF AE<0 THEN LET A=AE+2* PI
1200 IF AE<0 THEN GOTO 1190
1210 LET DE=1+.017* SIN (AE-3.33926)
1220 GOSUB 1710
1230 LET MZ=AE-MA
1240 LET ZM=MZ
1250 IF ABS (MZ)> PI AND MZ<0 THEN LET MZ=MZ+2* PI
1260 IF ABS (MZ)> PI AND MZ>0 THEN LET MZ=MZ-2* PI
1270 REM AU FROM EARTH
1280 LET MQ= SQR (MD*MD+DE*DE-2*MD*DE* COS (MZ))
1290 LET MP=(MD+DE+MQ)/2
1300 REM ANG.DIST.
1310 LET MV= SQR ((MP*(MP-MD))/(DE*MQ))
1320 LET MW=2* ACS (MV)
1330 IF ABS (ZM)> PI THEN LET ZM=ZM+2* PI
1340 IF ZM>0 THEN LET MW=-MW
1350 GOSUB 1830
1360 RETURN
1370 REM DATE
1380 PRINT
1390 PRINT
1400 PRINT "ENTER THE DATE...."
1410 PRINT
1420 PRINT "    YEAR ";
1430 INPUT Y
1440 PRINT Y
1450 PRINT
1460 PRINT "   MONTH ";
1470 INPUT M
1480 PRINT M
1490 PRINT
1500 IF M<1 OR M>12 THEN GOTO 1460
1510 PRINT "     DAY ";
1520 INPUT D
1530 PRINT D
1540 IF D<1 OR D>31 THEN GOTO 1510
1550 IF M=2 AND D>29 THEN GOTO 1510
1560 PAUSE 100
1570 LET YI=Y
1580 LET MI=M
1590 LET DI=D
1600 RETURN
1610 REM DAYS FROM EPOCH
1620 LET DG=365*Y+D+((M-1)*31)
1630 IF M >= 3 THEN GOTO 1660
1640 LET DG=DG+ INT ((Y-1)/4)- INT ((.75)* INT ((Y-1)/100+1))
1650 GOTO 1680
1660 LET DG=DG- INT (M*.4+2.3)+ INT (Y/4)
1670 LET DG=DG- INT ((.75)* INT ((Y/100)+1))
1680 LET NI=DG-715875
1690 RETURN
1700 GOTO 520
1710 REM MERCURY
1720 LET MA=NI*.071425+3.8494
```


MERVE (continued)

```
1730 IF MA>2* PI THEN LET MA=((MA/P2)- INT (MA/P2))*P2
1740 IF MA<0 THEN LET MA=MA+2* PI
1750 IF MA<0 THEN GOTO 1740
1760 LET MC=.388301* SIN (MA-1.34041)
1770 LET MA=MA+MC
1780 IF MA>2* PI THEN LET MA=MA-2* PI
1790 IF MA<0 THEN LET MA=MA+2* PI
1800 IF MA<0 THEN GOTO 1790
1810 LET MD=.3871+.079744* SIN (MA-2.73514)
1820 RETURN
1830 REM VENUS
1840 LET VA=NI*.027962+3.02812
1850 IF VA>2* PI THEN LET VA=((VA/P2)- INT (VA/P2))*2* PI
1860 IF VA<0 THEN LET VA=VA+2* PI
1870 IF VA<0 THEN GOTO 1860
1880 LET VC=.013195* SIN (VA-2.28638)
1890 LET VA=VA+VC
1900 IF VA>2* PI THEN LET VA=VA-2* PI
1910 IF VA<0 THEN LET VA=VA+2* PI
1920 IF VA<0 THEN GOTO 1910
1930 LET VD=.7233+.00506* SIN (VA-3.85017)
1940 LET ZV=AE-VA
1950 LET VZ=ZV
1960 IF ABS (ZV)> PI AND ZV<0 THEN LET ZV=ZV+2* PI
1970 IF ABS (ZV)> PI AND ZV>0 THEN LET ZV=ZV-2* PI
1980 REM AU FROM EARTH
1990 LET VQ= SQR (VD*VD+DE*DE-2*DE*VD* COS (ZV))
2000 LET VP=(VD+DE+VQ)/2
2010 REM ANGLE FROM SUN
2020 LET VV= SQR ((VP*(VP-VD))/(DE*VQ))
2030 LET VW=2* ACS (VV)
2040 IF ABS (VZ)> PI THEN LET VZ=VZ+2* PI
2050 IF VZ>0 THEN LET VW=-VW
2060 RETURN
2070 REM DISPLAY
2080 LET S=( VAL L$(1 TO 4))
2090 LET T=( VAL W$( TO 4))
2100 LET MX=15
2110 LET M1=MX+ INT (S*.3)
2120 LET M2=MX+ INT (T*.3)
2130 RETURN
2140 PRINT "FOR DISPLAY PRESS ANY KEY"
2150 IF INKEY$ ="" THEN GOTO 2150
2160 CLS
2170 PRINT "CONFIGURATION ON ";YI;" ";MI;" ";DI
2180 PRINT "V=VENUS, M=MERCURY, O=SUN"
2190 PRINT AT 13,M1;"M"
2200 PRINT AT 13,M2;"V"
2210 PRINT AT 13,15;"O"
2220 PRINT
2230 PRINT
2240 PRINT
2250 PRINT
2260 PRINT
2270 RETURN
2280 CLS
2290 REM ELONGATIONS
2300 PRINT
2310 PRINT
2320 PRINT "SELECT..."
2330 PRINT
```

MERVE (continued)

```
2340 PRINT "      MERCURY (1)"
2350 PRINT "    OR VENUS (2)"
2360 PRINT
2370 INPUT E$
2380 IF E$="1" THEN GOSUB 2410
2390 IF E$="1" THEN GOTO 2940
2400 GOSUB 3180
2410 CLS
2420 GOSUB 1400
2430 LET FL=0
2440 GOSUB 1610
2450 GOSUB 1100
2460 LET W$= STR$ (VW*57.29578)
2470 CLS
2480 PRINT
2490 PRINT YI;" ";MI;" ";DI
2500 PRINT
2510 PRINT
2520 PRINT
2530 PRINT
2540 PRINT
2550 LET L$= STR$ (MW*57.29578)
2560 GOSUB 2070
2570 GOSUB 2190
2580 PRINT
2590 PRINT
2600 PRINT "M=MERCURY, V=VENUS, O=SUN"
2610 LET DS=MW*57.29578
2620 IF FL=0 AND ABS (DS)<15 THEN GOTO 2640
2630 GOTO 2660
2640 LET D=D+7
2650 GOTO 2750
2660 LET CK= ABS (DS)- ABS (D2)
2670 IF CK<.06 AND CK>-.06 THEN GOTO 2870
2680 IF ABS (DS)<18 THEN GOTO 2700
2690 GOTO 2720
2700 LET D=D+4
2710 GOTO 2750
2720 IF CK<.1 OR CK>-.1 THEN GOTO 2740
2730 GOTO 2750
2740 LET D=D+1
2750 IF D>30 THEN GOTO 2770
2760 GOTO 2790
2770 LET M=M+1
2780 LET D=D-30
2790 IF M>12 THEN GOTO 2810
2800 GOTO 2830
2810 LET M=1
2820 LET Y=Y+1
2830 LET D2=DS
2840 LET FL=1
2850 GOSUB 1570
2860 GOTO 2440
2870 CLS
2880 PRINT
2890 PRINT
2900 PRINT
2910 PRINT "NEXT ELONGATION OF MERCURY"
2920 PRINT "       IS ";Y;" ";M;" ";D
2930 PRINT
2940 IF MW<0 THEN LET P$="EAST OF SUN (EVENING STAR)"
```


MERVE (continued)

```
2950 IF MW<0 THEN GOTO 2970
2960 LET P$="WEST OF SUN (MORNING STAR)"
2970 PRINT "AND THE PLANET IS.... "
2980 PRINT
2990 PRINT TAB 5;P$
3000 PRINT
3010 GOSUB 2170
3020 PRINT "FOR DETAILS PRESS ANY KEY"
3030 IF INKEY$ ="" THEN GOTO 3030
3040 CLS
3050 GOSUB 560
3060 LET A$=" "
3070 PRINT
3080 PRINT "WANT ANOTHER DATE? Y/N "
3090 INPUT A$
3100 IF E$="2" AND A$="Y" THEN GOTO 3180
3110 IF E$="1" AND A$="Y" THEN GOTO 2410
3120 PRINT
3130 PRINT
3140 PRINT "WANT THE OTHER PLANET? Y/N "
3150 INPUT A$
3160 IF A$="Y" THEN GOTO 2280
3170 GOTO 1020
3180 CLS
3190 GOSUB 1400
3200 LET FL=0
3210 GOSUB 1610
3220 GOSUB 1100
3230 CLS
3240 LET L$= STR$ (MW*57.29578)
3250 PRINT
3260 PRINT YI;" ";MI;" ";DI
3270 PRINT
3280 PRINT
3290 LET W$= STR$ (VW*57.29578)
3300 GOSUB 2070
3310 GOSUB 2190
3320 PRINT
3330 PRINT "V=VENUS, M=MERCURY, O=SUN"
3340 LET VS=VW*57.29578
3350 IF FL=0 AND ABS (VS)<40 THEN GOTO 3370
3360 GOTO 3390
3370 LET D=D+20
3380 GOTO 3490
3390 LET CK= ABS (VS)- ABS (V2)
3400 IF CK<.05 AND CK>-.05 THEN GOTO 3610
3410 IF ABS (VS)<45 THEN GOTO 3430
3420 GOTO 3450
3430 LET D=D+10
3440 GOTO 3490
3450 IF CK<.1 OR CK>-.1 THEN GOTO 3470
3460 GOTO 3490
3470 LET D=D+1
3480 GOTO 3490
3490 IF D>30 THEN GOTO 3510
3500 GOTO 3570
3510 LET M=M+1
3520 LET D=D-30
3530 IF M>12 THEN GOTO 3550
3540 GOTO 3570
3550 LET M=1
```

MERVE (continued)

```
3560 LET Y=Y+1
3570 LET V2=VS
3580 LET FL=1
3590 GOSUB 1570
3600 GOTO 3210
3610 CLS
3620 PRINT
3630 PRINT
3640 PRINT
3650 PRINT "NEXT ELONGATION OF VENUS"
3660 PRINT "       IS ";Y;" ";M;" ";D
3670 LET MW=VW
3680 PRINT
3690 GOTO 2940
3700 REM PHASES
3710 LET PH=180-PH*57.2958
3720 IF PH>150 THEN LET H$="THIN CRESCENT"
3730 IF PH>120 AND PH <= 151 THEN LET H$="FAT CRESCENT"
3740 IF PH>70 AND PH <= 120 THEN LET H$="NEAR HALF"
3750 IF PH>29 AND PH <= 70 THEN LET H$="GIBBOUS"
3760 IF PH <= 21 THEN LET H$="NEAR FULL"
3770 LET PH= INT (180-PH)
3780 RETURN
```


Program 14: RISES

Times of Rising, Transit, and Setting of Planets, Sun, and Moon for Any Date

This program calculates the approximate time of rising, transit, and setting of the Sun, the Moon, and the planets for any date, at any latitude and longitude. Times of actual visibility after rising and before setting depend on local horizon and atmospheric conditions as well as on the rate of rising and setting—that is, the body's position on the ecliptic relative to the latitude of the observer. The Sun and Moon, because of their brightness, are often visible sooner after rising and closer to setting than are the planets. A typical display generated by this program is shown in Figure 14.1.

The program loads data on the planetary orbits by slicing the data string, and it calculates for each planet its right ascension and declination at the requested date. It displays these data for the selected planet together with its distance from the Sun. The program uses the calculated sidereal time for the chosen date together with the latitude to calculate the rise, transit, and set times of the first point of Aries. From this the program calculates the local times when that point in the celestial sphere with the right ascension and declination of the planet rises above the horizon, culminates, and sets below the horizon at the observer's location.

The listing of the RISES program follows.

```
DATA REQUESTED FOR 1983 6 15
-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-
          HEL.  DIST.   R.A.  DEC.
          LONG.  A.U.   HRS   DEG.
-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-
VENUS     218.  0.717  8.579  20.38
-:-:-:-:-:-:-:-:-:-:-:-:-:-:-:-


VENUS
RISES AT....  7 HRS 55 MIN
TRANSITS AT:  15 HRS  3 MIN
SETS AT.....  22 HRS 13 MIN

WANT COPY Y/N?
```

Figure 14.1: The RISES program displays the times of rising and setting for a planet, the Sun, or the Moon on a selected date.

RISES

```
 10 CLEAR
 20 DIM A$(10)
 30 DIM U(9)
 40 DIM P(9,9)
 50 DIM N$(22)
 60 DIM M$(63)
 70 DIM P$(10,7)
 80 REM RISES 4/16/83
 90 LET P2=2* PI
100 DIM A(9)
110 DIM D(9)
120 DIM L(9)
130 DIM Q(9)
140 DIM R(10)
150 DIM V(10)
160 DIM K(9)
170 LET FL=0
180 DIM V$(9)
190 PRINT AT 5,5;"-----------------"
200 DIM R$(10)
210 PRINT AT 6,5;"I     RISES     I"
220 PRINT AT 7,5;"-----------------"
```


RISES (continued)

```
230 LET FR=.01745329
240 LET FD=57.29578
250 LET LA=38.24
260 LET LO=122.49
270 REM CHANGE ABOVE TO YOUR LATITUDE
280 REM AND LONGITUDE
290 PRINT AT 10,3;"BY ERIC BURGESS F.R.A.S."
300 PRINT AT 12,4;"ALL RIGHTS RESERVED BY"
310 PRINT AT 13,3;"S AND T SOFTWARE SERVICE"
320 PAUSE 350
330 CLS
340 PRINT
350 PRINT
360 PRINT
370 PRINT TAB 5;"THIS PROGRAM GIVES"
380 PRINT TAB 2;"APPROXIMATE TIMES OF THE"
390 PRINT TAB 2;"RISING, TRANSIT, AND SETTING"
400 PRINT TAB 4;"OF A SELECTED PLANET"
410 PRINT TAB 3;"OR THE SUN OR THE MOON"
420 PRINT "FOR THE DATE WHICH YOU REQUEST"
430 PRINT
440 PRINT
450 PRINT "INITIAL CONDITIONS ARE SET FOR.."
460 PRINT "  LATITUDE.. ";LA
470 PRINT "  LONGITUDE  ";LO
480 PRINT
490 PRINT "WANT TO CHANGE THEM Y/N? ";
500 INPUT B$
510 PRINT B$
520 IF B$ <> "Y" THEN GOTO 590
530 PRINT "TYPE LATITUDE ";
540 INPUT LA
550 PRINT LA
560 PRINT "TYPE LONGITUDE ";
570 INPUT LO
580 PRINT LO
590 PRINT
600 PRINT
610 PRINT "PLEASE WAIT; LOADING ARRAYS"
620 GOSUB 1360
630 CLS
640 PRINT
650 PRINT
660 PRINT "ENTER THE DATE"
670 PRINT
680 PRINT "       YEAR.... ";
690 INPUT Y
700 PRINT Y
710 PRINT
720 PRINT "       MONTH... ";
730 INPUT M
740 PRINT M
750 IF M<1 OR M>12 THEN GOTO 720
760 PRINT
770 PRINT "       DAY..... ";
780 INPUT DY
790 PRINT DY
```

RISES (continued)

```
 800 IF DY<1 OR DY>31 THEN GOTO 770
 810 IF M=2 AND DY>29 THEN GOTO 770
 820 REM SELECT PLANET
 830 GOSUB 3040
 840 PRINT
 850 LET DX=DY
 860 LET MX=M
 870 LET YX=Y
 880 LET DY=DY+(LO/15)/24
 890 LET DG=365*Y+DY+((M-1)*31)
 900 IF M >= 3 THEN GOTO 930
 910 LET DG=DG+ INT ((Y-1)/4)- INT ((.75)* INT ((Y-1)/100+1)
 920 GOTO 950
 930 LET DG=DG- INT (M*.4+2.3)+ INT (Y/4)
 940 LET DG=DG- INT ((.75)* INT ((Y/100)+1))
 950 LET NI=DG-715875
 960 REM GET S.T.
 970 GOSUB 3280
 980 IF CI=10 THEN GOTO 2670
 990 CLS
1000 FAST
1010 GOSUB 1670
1020 SLOW
1030 PRINT
1040 PRINT "DATA REQUESTED FOR ";YX;" ";MX;" ";DX
1050 PRINT "-:-:-:-:-:-:-:-:-:-:-:-:-:-:-"
1060 IF CI=3 THEN PRINT TAB 20;"R.A.  DEC"
1070 IF CI=3 THEN PRINT TAB 20;"HRS   DEG"
1080 IF CI=3 THEN GOTO 1110
1090 PRINT "        HEL. DIST.  R.A. DEC."
1100 PRINT "        LONG. A.U.  HRS  DEG."
1110 PRINT "-:-:-:-:-:-:-:-:-:-:-:-:-:-:-"
1120 IF CI=3 THEN PRINT P$(CI); TAB 19;R$( TO 4);"  ";V$
1130 IF CI=3 THEN PRINT
1140 IF CI=3 THEN GOTO 1250
1150 PRINT P$(CI);
1160 LET A(CI)=A(CI)*57.29578
1170 LET A$= STR$ (A(CI))
1180 PRINT TAB 8;A$( TO 4);
1190 LET A$= STR$ (Q(CI))
1200 PRINT TAB 13;A$( TO 5);
1210 LET A$= STR$ (R(CI))
1220 PRINT TAB 19;A$( TO 5);
1230 LET A$= STR$ (V(CI))
1240 PRINT TAB 25;A$( TO 5)
1250 PRINT "-:-:-:-:-:-:-:-:-:-:-:-:-:-:-"
1260 GOSUB 3410
1270 PRINT
1280 PRINT "WANT COPY Y/N? "
1290 INPUT C$
1300 IF C$="N" OR C$="NO" THEN GOTO 1320
1310 COPY
1320 CLS
1330 PRINT
1340 PRINT
1350 GOTO 3570
1360 REM PLANETARY DATA
```


RISES (continued)

```
1370 LET U=1
1380 LET N$=".0714223.84840.3883011"
1390 LET M$=N$+".34041.387100.0797402.73514.122173.836013"
1400 GOSUB 1960
1410 LET N$=".0279623.02812.0131952"
1420 LET M$=N$+".28638.723300.0050603.85017.0593411.33168"
1430 GOSUB 1960
1440 LET N$=".0172021.74022.0320441"
1450 LET M$=N$+".785471.00000.0170003.339260000000000000"
1460 GOSUB 1960
1470 LET N$=".0091464.51234.1753015"
1480 LET M$=N$+".852091.52370.1417041.04656.031420.858702"
1490 GOSUB 1960
1500 LET N$=".0014504.53364.090478."
1510 LET M$=N$+"2391105.20280.2493741.76188.0197201.74533"
1520 GOSUB 1960
1530 LET N$=".0005844.89884.1055581"
1540 LET M$=N$+".610949.53850.5341563.12570.0436331.97746"
1550 GOSUB 1960
1560 LET N$=".0002052.46615.0885932"
1570 LET M$=N$+".9670619.1820.9015544.49084.0139601.28805"
1580 GOSUB 1960
1590 LET N$=".0001043.78556.016965."
1600 LET M$=N$+"77318130.0600.2705402.33498.0314162.29162"
1610 GOSUB 1960
1620 LET N$=".0000693.16948.4712393"
1630 LET M$=N$+".9130339.44009.860005.23114.3001971.91812"
1640 GOSUB 1960
1650 LET M$=""
1660 RETURN
1670 REM PLANET NAMES
1680 IF FL=1 THEN GOTO 1790
1690 LET P$(1)="MERCURY"
1700 LET P$(2)="VENUS"
1710 LET P$(3)="SUN"
1720 LET P$(4)="MARS"
1730 LET P$(5)="JUPITER"
1740 LET P$(6)="SATURN"
1750 LET P$(7)="URANUS"
1760 LET P$(8)="NEPTUNE"
1770 LET P$(9)="PLUTO"
1780 LET P$(10)="MOON"
1790 LET I=1
1800 FOR J=1 TO 9
1810 GOSUB 2010
1820 LET A(I)=A
1830 LET D(I)=D
1840 LET L(I)=L
1850 LET I=I+1
1860 NEXT J
1870 FOR I=1 TO 9
1880 IF I=3 THEN NEXT I
1890 GOSUB 2130
1900 LET Q(I)=Q
1910 LET R(I)=R
1920 LET V(I)=V
1930 NEXT I
```

RISES (continued)

```
1940 FOR I=1 TO 9
1950 RETURN
1960 FOR K=1 TO 9
1970 LET P(U,K)= VAL (M$(1+((K-1)*7) TO (K*7)))
1980 NEXT K
1990 LET U=U+1
2000 RETURN
2010 LET A=NI*P(J,1)+P(J,2)
2020 IF A>6.28318 THEN LET A=((A/6.28318)- INT
     (A/6.28318))*6.28318
2030 IF A<0 THEN LET A=A+P2
2040 IF A<0 THEN GOTO 2030
2050 LET C=P(J,3)* SIN (A-P(J,4))
2060 LET A=A+C
2070 IF A>P2 THEN LET A=A-P2
2080 IF A<0 THEN LET A=A+P2
2090 IF A<0 THEN GOTO 2080
2100 LET D=P(J,5)+P(J,6)* SIN (A-P(J,7))
2110 LET L=P(J,8)* SIN (A-P(J,9))
2120 RETURN
2130 LET Z=A(3)-A(I)
2140 IF ABS (Z)> PI AND Z<0 THEN LET Z=Z+P2
2150 IF ABS (Z)> PI AND Z>0 THEN LET Z=Z-P2
2160 LET CZ= COS (Z)
2170 LET Q=D(I) ** 2+D(3) ** 2-2*D(I)*D(3)*CZ
2180 LET Q= SQR (Q)
2190 LET P=(D(I)+D(3)+Q)/2
2200 LET X=2* ACS ( SQR (((P*(P-D(I)))/(D(3)*Q))))
2210 IF Z<0 THEN LET R=57.29578*(A(3)+3.14159-X)/15
2220 IF Z>0 THEN LET R=57.29578*(A(3)+3.14159+X)/15
2230 IF R>24 THEN LET R=R-24
2240 IF R>24 THEN GOTO 2230
2250 IF R<-24 THEN LET R=R+24
2260 IF R<-24 THEN GOTO 2250
2270 IF R<0 THEN LET R=R+24
2280 IF R<0 THEN GOTO 2270
2290 IF Z<0 THEN LET V= SIN (A(3)+ PI -X)*23.44194+FD*L(I)
2300 IF Z>0 THEN LET V= SIN (A(3)+ PI +X)*23.44194+FD*L(I)
2310 LET X=FD*X
2320 LET R(3)=FD*(A(3)+ PI )/15
2330 IF R(3) <= 24 THEN GOTO 2360
2340 LET R(3)=R(3)-24
2350 GOTO 2330
2360 IF R(3) >= 0 THEN GOTO 2390
2370 LET R(3)=R(3)+24
2380 GOTO 2360
2390 LET R$=( STR$ (R(3)))( TO 4)
2400 LET V(3)= SIN (A(3)+3.141159)*23.44194
2410 LET V$=( STR$ (V(3)))( TO 4)
2420 RETURN
2430 REM CALC TIMES
2440 LET TA= TAN (FR*LA)* TAN (FR*V(CI))
2450 IF TA<0 THEN GOTO 2470
2460 IF TA>0 THEN GOTO 2490
2470 LET TA= ABS (TA)
2480 GOTO 2500
2490 LET TA=-TA
```


RISES (continued)

```
2500 LET TA= ACS (TA)
2510 LET TA=FD*TA
2520 LET B=-TA/15
2530 LET TZ=0
2540 LET TR=(B+R(CI)+TZ-T2)*.99727
2550 LET TR=TR-1/60
2560 IF TR<0 THEN LET TR=TR+24
2570 IF TR>24 THEN LET TR=TR-24
2580 LET C=TA/15
2590 LET TS=(C+R(CI)+TZ-T2)*.99727
2600 LET TS=TS+3/60
2610 IF TS<0 THEN LET TS=TS+24
2620 IF TS>24 THEN LET TS=TS-24
2630 LET TT=R(CI)-T2+1/60
2640 IF TT<0 THEN LET TT=TT+24
2650 IF TT>24 THEN LET TT=TT-24
2660 RETURN
2670 REM RISE, XSIT, SET OF MOON
2680 LET LZ=311.1687
2690 LET LE=178.699
2700 LET LP=255.7433
2710 LET PG=.111404*NI+LP
2720 LET PG=(PG/360- INT (PG/360))*360
2730 LET LMD=LZ+360*NI/27.32158
2740 LET LMD=(LMD/360- INT (LMD/360))*360
2750 LET PG=LMD-PG
2760 LET DR=6.2886* SIN (FR*PG)
2770 LET LMD=LMD+DR
2780 LET LMD=(LMD/360- INT (LMD/360))*360
2790 LET RM=LMD/15
2800 IF RM>24 THEN LET RM=RM-24
2810 IF RM<0 THEN LET RM=RM+24
2820 LET AL=LE-NI*.052954
2830 LET AL=LMD-AL
2840 LET AL=(AL/360- INT (AL/360))*360
2850 LET HE=5.1454* SIN (AL*3.14159/180)
2860 LET DM=HE+23.1444* SIN (LMD*3.14159/180)
2870 LET R$=( STR$ (RM))( TO 5)
2880 LET D$=( STR$ (DM))( TO 5)
2890 CLS
2900 PRINT
2910 PRINT "MOON DATA REQUESTED FOR..."
2920 PRINT TAB 4;"YEAR ";YX;" MONTH ";
2930 PRINT MX;" DAY ";DX
2940 PRINT
2950 PRINT "AT NOON"
2960 PRINT "R.A. OF MOON IS.. ";R$;" HRS'
2970 PRINT "DECLINATION IS.. ";D$;" DEG'
2980 PRINT "-----------------------------"
2990 LET R(10)=RM
3000 LET V(10)=DM
3010 LET P$(CI)="MOON"
3020 GOSUB 2430
3030 GOTO 3420
3040 REM SELECT PLANET
3050 CLS
3060 PRINT
3070 PRINT
```

RISES (continued)

```
3080 PRINT
3090 PRINT "SELECT BY NUMBER"
3100 PRINT
3110 PRINT TAB 5;"MERCURY.... 1"
3120 PRINT TAB 5;"VENUS...... 2"
3130 PRINT TAB 5;"            3...SUN"
3140 PRINT TAB 5;"MARS....... 4"
3150 PRINT TAB 5;"JUPITER.... 5"
3160 PRINT TAB 5;"SATURN..... 6"
3170 PRINT TAB 5;"URANUS..... 7"
3180 PRINT TAB 5;"NEPTUNE.... 8"
3190 PRINT TAB 5;"PLUTO...... 9"
3200 PRINT TAB 5;"            10..MOON"
3210 PRINT
3220 PRINT "TYPE NUMBER... ";
3230 INPUT SS
3240 IF SS<1 OR SS>10 THEN GOTO 3050
3250 PRINT SS
3260 LET CI=SS
3270 RETURN
3280 REM SIDEREAL TIME
3290 LET GC=11.927485
3300 LET TC=.065711
3310 LET T2=TC*(NI-7020)+GC
3320 IF T2<24 THEN GOTO 3350
3330 LET T2=T2-24
3340 GOTO 3320
3350 IF T2>-24 THEN GOTO 3380
3360 LET T2=T2+24
3370 GOTO 3350
3380 IF T2>0 THEN GOTO 3400
3390 LET T2=T2+24
3400 RETURN
3410 GOSUB 2430
3420 PRINT
3430 PRINT
3440 LET TM=60*(TR- INT (TR))
3450 LET TR= INT (TR)
3460 LET TN=60*(TT- INT (TT))
3470 LET TT= INT (TT)
3480 LET TP=60*(TS- INT (TS))
3490 LET TS= INT (TS)
3500 PRINT
3510 PRINT P$(CI)
3520 PRINT "RISES AT.... ";TR;" HRS "; INT (TM);" MIN "
3530 PRINT "TRANSITS AT. ";TT;" HRS "; INT (TN);" MIN"
3540 PRINT "SETS AT..... ";TS;" HRS "; INT (TP);" MIN"
3550 IF CI=10 THEN GOTO 1270
3560 RETURN
3570 PRINT
3580 PRINT "WANT ANOTHER PLANET"
3590 PRINT "FOR SAME DATE Y/N? ";
3600 INPUT B$
3610 PRINT B$
3620 IF B$="N" THEN GOTO 3750
3630 IF CI=10 THEN GOTO 3650
3640 IF B$="Y" THEN GOTO 3690
```


RISES (continued)

```
3650 LET DY=DX
3660 LET M=MX
3670 LET Y=YX
3680 GOTO 820
3690 GOSUB 3050
3700 IF CI=3 THEN LET FL=2
3710 IF CI=10 THEN GOSUB 2670
3720 IF CI=10 THEN GOTO 3760
3730 CLS
3740 GOTO 1030
3750 PRINT
3760 PRINT "WANT ANOTHER DATE Y/N? ";
3770 INPUT B$
3780 PRINT B$
3790 IF B$="N" THEN GOTO 3830
3800 LET FL=0
3810 LET F=0
3820 IF B$="Y" THEN GOTO 630
3830 CLS
3840 STOP
```

Program 15: SKYPT

Horizon Plots of Visible Planets, Sun, and Moon for Any Date, Time, and Location

For the T/S 1000 computer this program plots the positions of the planets, Sun, and Moon if they are above the horizon at the time, date, and location chosen. The book *Celestial BASIC*, also published by Sybex, contains more advanced versions of the program, to plot the stars as well. Such programs require computers that have greater graphic resolution.

When you run the program, you will be asked for date and other input information, as well as whether you want to change the location parameters of time zone, latitude, and longitude. Next you are asked to select a horizon of 180 degrees centered on east, south, west, or north.

Next the horizon chart is generated. For the particular horizon you have requested, the azimuth is shown below the display and the elevation is shown at the right. Date and time information are displayed above the chart. Then the program plots the Sun, the Moon, and any planets above that horizon at the time and date selected. The Moon is shown as) before full, 0 when close to full, and (after full.

Variables have been set initially for your latitude, longitude, and time zone when you key in the values of LA, LO, and TZ. While running the program you can change the variables to other latitudes, longitudes, and time zones. The program should not be expected to run accurately at latitudes exceeding 85 degrees north and south.

Planets are identified by symbols: +, Mercury; V, Venus; M, Mars; J, Jupiter; S, Saturn; U, Uranus; N, Neptune; and P, Pluto. Note that if planets are within one pixel of each other, the outermost planet will overprint the innermost and only the symbol for the outer planet will be displayed.

The graphics characters in lines 1570 through 1700 are used with other characters to generate a silhouette of buildings on the horizon, much as is done in a planetarium display. You can change these, if you wish, to create a suitable horizon for your own locality.

To see how the program works, you might choose to display the sky on 26 November 1981 at 11:30 A M for the south horizon (Figure 15.1). This shows all the planets, the Sun, and the Moon above the horizon at the same time. If you then ask for the same date and time in the Southern Hemisphere (for example, for −40 degrees latitude) and request the north horizon, you will see these same planets inverted (Figure 15.2).

The listing of the SKYPT program follows.



Figure 15.1: The SKYPT program generates this display of all the planets, Sun, and Moon above the horizon on 26 November 1981 at 11:30 A M. local time.



Figure 15.2: By requesting a display for 40 degrees south latitude and selecting the north horizon, you can see the planetary configuration of November 1981 as it appeared from Australia.

SKYPT

```
 10 SLOW
 20 CLS
 30 CLEAR
 40 DIM A$(10)
 50 DIM U(9)
 60 DIM P(9,9)
 70 DIM N$(22)
 80 DIM M$(63)
 90 DIM P$(10)
100 REM SKYPT 4/16/83 VERSION
110 LET P2=2* PI
120 DIM A(9)
130 LET FD=57.29578
140 DIM D(9)
150 DIM L(9)
160 DIM Q(9)
170 DIM R(9)
180 DIM V(9)
190 DIM K(9)
200 DIM B(10)
210 DIM Y(10)
220 DIM E(10)
230 PRINT AT 5,5;"-----------------"
```


SKYPT (continued)

```
240 PRINT AT 6,5;"I    SKYPLOT    I"
250 PRINT AT 7,5;"-----------------"
260 PRINT AT 9,5;"ASTRONOMY PROGRAM"
270 PRINT AT 10,3;"BY ERIC BURGESS F.R.A.S."
280 PRINT AT 12,4;"ALL RIGHTS RESERVED BY"
290 PRINT AT 13,3;"S AND T SOFTWARE SERVICE"
300 LET FL=0
310 PAUSE 350
320 CLS
330 PRINT
340 PRINT
350 PRINT
360 PRINT TAB 5;"THIS PROGRAM SHOWS"
370 PRINT TAB 4;"THE POSITIONS OF THE"
380 PRINT TAB 3;"PLANETS, SUN, AND MOON"
390 PRINT TAB 3;"AS THEY APPEAR IN THE"
400 PRINT TAB 5;"SKY FOR ANY CHOSEN"
410 PRINT TAB 6;"LOCATION ON EARTH"
420 PRINT TAB 7;"FOR ANY TIME ON"
430 PRINT TAB 2;"THE DATE WHICH YOU REQUEST"
440 PRINT
450 PRINT
460 PAUSE 100
470 PRINT "WANT INSTRUCTIONS Y/N? ";
480 INPUT D$
490 IF D$="Y" THEN GOSUB 3960
500 PRINT
510 PRINT
520 PRINT "PLEASE WAIT; LOADING ARRAYS"
530 GOSUB 2150
540 CLS
550 LET LA=38.24
560 LET LO=122.49
570 PRINT "INITIAL CONDITIONS ARE SET FOR"
580 PRINT "LATITUDE... ";LA
590 PRINT "LONGITUDE.. ";LO
600 PRINT
610 PRINT "WANT TO CHANGE THEM Y/N? ";
620 INPUT L$
630 PRINT L$
640 IF L$="N" THEN GOTO 710
650 PRINT "TYPE LATITUDE ";
660 INPUT LA
670 PRINT LA
680 PRINT "TYPE LONGITUDE ";
690 INPUT LO
700 PRINT LO
710 LET LS=LA
720 LET LA= PI /180*LA
730 PRINT
740 CLS
750 PRINT
760 PRINT "ENTER THE DATE"
770 PRINT
780 PRINT "      YEAR.... ";
790 INPUT Y
800 PRINT Y
```

SKYPT (continued)

```
 810 PRINT
 820 PRINT "       MONTH... ";
 830 INPUT M
 840 PRINT M
 850 IF M<1 OR M>12 THEN GOTO 820
 860 PRINT
 870 PRINT "       DAY..... ";
 880 INPUT DY
 890 PRINT DY
 900 IF DY<1 OR DY>31 THEN GOTO 870
 910 IF M=2 AND DY>29 THEN GOTO 870
 920 PRINT
 930 PRINT "       TIME ZONE ";
 940 INPUT TZ
 950 PRINT TZ
 960 IF TZ<-12 OR TZ>12 THEN GOTO 930
 970 PRINT
 980 LET DS=DY
 990 PRINT "TYPE LMT (HRS) ";
1000 INPUT T1
1010 PRINT T1
1020 IF T1<0 OR T1>24 THEN GOTO 1030
1030 PRINT
1040 PRINT "SELECT HORIZON (1 THRU 4)"
1050 PRINT
1060 PRINT "  EAST    0 TO 180 (1)"
1070 PRINT "  WEST  180 TO 360 (2)"
1080 PRINT "  SOUTH  90 TO 270 (3)"
1090 PRINT "  NORTH 270 TO  90 (4)"
1100 PRINT
1110 PRINT "TYPE NUMBER SELECTED ";
1120 INPUT HZ
1130 PRINT HZ
1140 CLS
1150 IF HZ<1 OR HZ>4 THEN GOTO 1030
1160 PRINT
1170 PRINT "SCREEN WILL GO BLANK"
1180 PRINT "DURING CALCULATIONS"
1190 PRINT
1200 PRINT "PLEASE WAIT FOR DISPLAY"
1210 PAUSE 100
1220 FAST
1230 LET DG=365*Y+DY+((M-1)*31)
1240 IF M >= 3 THEN GOTO 1270
1250 LET DG=DG+ INT ((Y-1)/4)- INT ((.75)* INT ((Y-1)/100+1))
1260 GOTO 1290
1270 LET DG=DG- INT (M*.4+2.3)+ INT (Y/4)
1280 LET DG=DG- INT ((.75)* INT ((Y/100)+1))
1290 LET NS=DG-722895
1300 LET NI=DG-715875
1310 LET GC=12.064707
1320 LET SG=.065711
1330 LET T2=SG*NS+GC+(((T2+T1)/24)*SG)+T1
1340 IF T2<24 THEN GOTO 1380
1350 IF T2>240 THEN LET T2=T2-240
1360 LET T2=T2-24
1370 GOTO 1340
```


SKYPT (continued)

```
1380 IF T2>-24 THEN GOTO 1420
1390 IF T2<-240 THEN LET T2=T2+240
1400 LET T2=T2+24
1410 GOTO 1380
1420 IF T2>0 THEN GOTO 1440
1430 LET T2=T2+24
1440 CLS
1450 GOSUB 2460
1460 SLOW
1470 PRINT "SKYPLOT - ";Y;" ";M;" ";DY;" AT ";T1;" LMT'
1480 PRINT INT (T2);' LST ";LS;" LAT. ";LO;" LONG."
1490 FOR K=1 TO 17
1500 LET Z=90
1510 PRINT "I";
1520 PRINT TAB 27;"I- ";Z-K*5
1530 NEXT K
1540 IF HZ=2 THEN GOTO 1610
1550 IF HZ=3 THEN GOTO 1650
1560 IF HZ=4 THEN GOTO 1690
1570 PRINT AT 18,1;"
1580 PRINT AT 19,1;"
1590 PRINT AT 20,1;"N     NE      E     SE      S'
1600 GOTO 1720
1610 PRINT AT 18,1;"
1620 PRINT AT 19,1;"
1630 PRINT AT 20,1;"S     SW      W     NW      N"
1640 GOTO 1720
1650 PRINT AT 18,1;"
1660 PRINT AT 19,1;"
1670 PRINT AT 20,1;"E     SE      S     SW      W"
1680 GOTO 1720
1690 PRINT AT 18,1;"
1700 PRINT AT 19,1;"
1710 PRINT AT 20,1;"W     NW      N     NE      E"
1720 REM PRINT PLANETS
1730 FOR J=1 TO 10
1740 LET E(J)=B(J)
1750 IF E(J)<5 OR E(J)>85 THEN GOTO 1930
1760 IF HZ=1 THEN LET Y(J)=Y(J)
1770 IF HZ=1 THEN GOTO 1870
1780 IF HZ=2 THEN LET Y(J)=Y(J)-180
1790 IF HZ=2 THEN GOTO 1870
1800 IF HZ=3 THEN LET Y(J)=Y(J)-90
1810 IF HZ=3 THEN GOTO 1870
1820 IF HZ=4 AND Y(J)>270 AND Y(J)<360 THEN GOTO 1840
1830 IF HZ=4 AND Y(J)>0 AND Y(J)<90 THEN GOTO 1860
1840 LET Y(J)=Y(J)-270
1850 GOTO 1870
1860 LET Y(J)=Y(J)+90
1870 IF Y(J)<0 OR Y(J)>180 THEN GOTO 1930
1880 LET X1=E(J)
1890 LET Y1=Y(J)
1900 LET X2=19- INT (X1*18/90)
1910 LET Y2= INT (Y1*27/180)-1
1920 PRINT AT X2,Y2;P$(J)
1930 NEXT J
1940 PRINT AT 21,0;"WANT COPY Y/N? "
```

SKYPT (continued)

```
1950 INPUT C$
1960 IF C$="N" OR C$="NO" THEN GOTO 1990
1970 PRINT AT 21,0;"                "
1980 COPY
1990 CLS
2000 PRINT
2010 PRINT "WANT ANOTHER HORIZON Y/N? ";
2020 INPUT E$
2030 IF E$="N" THEN GOTO 2060
2040 CLS
2050 GOTO 1040
2060 CLS
2070 PRINT
2080 PRINT "WANT ANOTHER DATE Y/N? "
2090 INPUT G$
2100 IF G$="N" THEN GOTO 2130
2110 LET FL=1
2120 GOTO 740
2130 CLS
2140 STOP
2150 REM PLANETARY DATA
2160 LET U=1
2170 LET N$=".0714223.84840.3883011"
2180 LET M$=N$+".34041.387100.0797402.73514.122173.836013"
2190 GOSUB 2810
2200 LET N$=".0279623.02812.0131952"
2210 LET M$=N$+".28638.723300.0050603.85017.0593411.33168"
2220 GOSUB 2810
2230 LET N$=".0172021.74022.0320441"
2240 LET M$=N$+".785471.00000.0170003.339260000000000000"
2250 GOSUB 2810
2260 LET N$=".0091464.51234.1753015"
2270 LET M$=N$+".852091.52370.1417041.04656.031420.858702"
2280 GOSUB 2810
2290 LET N$=".0014504.53364.090478."
2300 LET M$=N$+"2391105.20280.2493741.76188.0197201.74533"
2310 GOSUB 2810
2320 LET N$=".0005844.89884.1055581"
2330 LET M$=N$+".610949.53850.5341563.12570.0436331.97746"
2340 GOSUB 2810
2350 LET N$=".0002052.46615.0885932"
2360 LET M$=N$+".9670619.1820.9015544.49084.0139601.28805"
2370 GOSUB 2810
2380 LET N$=".0001043.78556.016965."
2390 LET M$=N$+"77318130.0600.2705402.33498.0314162.29162"
2400 GOSUB 2810
2410 LET N$=".0000693.16948.4712393"
2420 LET M$=N$+".9130339.44009.860005.23114.3001971.91812"
2430 GOSUB 2810
2440 LET M$=""
2450 RETURN
2460 REM PLANET NAMES
2470 IF FL=1 THEN GOTO 2570
2480 LET P$(1)="+"
2490 LET P$(2)="V"
2500 LET P$(3)="O"
2510 LET P$(4)="M"
```


SKYPT (continued)

```
2520 LET P$(5)="J"
2530 LET P$(6)="S"
2540 LET P$(7)="U"
2550 LET P$(8)="N"
2560 LET P$(9)="P"
2570 LET I=1
2580 FOR J=1 TO 9
2590 GOSUB 2860
2600 LET A(I)=A
2610 LET D(I)=D
2620 LET L(I)=L
2630 LET I=I+1
2640 NEXT J
2650 FOR I=1 TO 9
2660 IF I=3 THEN NEXT I
2670 GOSUB 2980
2680 LET Q(I)=Q
2690 LET R(I)=R
2700 LET V(I)=V
2710 LET B(I)=AL
2720 LET Y(I)=AZ
2730 NEXT I
2740 GOSUB 3280
2750 LET B(3)=AS
2760 LET Y(3)=ZS
2770 GOSUB 3480
2780 LET B(10)=NL
2790 LET Y(10)=NZ
2800 RETURN
2810 FOR K=1 TO 9
2820 LET P(U,K)= VAL (M$(1+((K-1)*7) TO (K*7)))
2830 NEXT K
2840 LET U=U+1
2850 RETURN
2860 LET A=NI*P(J,1)+P(J,2)
2870 IF A>P2 THEN LET A=((A/P2)- INT (A/P2))*P2
2880 IF A<0 THEN LET A=A+6.28318
2890 IF A<0 THEN GOTO 2880
2900 LET C=P(J,3)* SIN (A-P(J,4))
2910 LET A=A+C
2920 IF A>6.28318 THEN LET A=A-6.82318
2930 IF A<0 THEN LET A=A+6.82318
2940 IF A<0 THEN GOTO 2930
2950 LET D=P(J,5)+P(J,6)* SIN (A-P(J,7))
2960 LET L=P(J,8)* SIN (A-P(J,9))
2970 RETURN
2980 LET Z=A(3)-A(I)
2990 IF ABS (Z)>3.14159 AND Z<0 THEN LET Z=Z+6.28318
3000 IF ABS (Z)>3.14159 AND Z>0 THEN LET Z=Z-6.28318
3010 LET Q= SQR (D(I) ** 2+D(3) ** 2-2*D(I)*D(3)* COS (Z))
3020 LET P=(D(I)+D(3)+Q)/2
3030 LET X=2* ACS ( SQR (((P*(P-D(I)))/(D(3)*Q))))
3040 IF Z<0 THEN LET R=57.29578*(A(3)+3.14159-X)/15
3050 IF Z>0 THEN LET R=57.29578*(A(3)+3.14159+X)/15
3060 IF R>24 THEN LET R=R-24
3070 IF R>24 THEN GOTO 3060
3080 IF R<0 THEN LET R=R+24
```

SKYPT (continued)

```
3090 IF R<0 THEN GOTO 3080
3100 IF Z<0 THEN LET V= SIN (A(3)+ PI -X)*23.44194+FD*L(I)
3110 IF Z>0 THEN LET V= SIN (A(3)+ PI +X)*23.44194+FD*L(I)
3120 REM COMP AZ AND EL
3130 LET HA=T2-R
3140 IF HA<-12 THEN LET HA=HA+24
3150 IF HA>12 THEN LET HA=HA-24
3160 LET HA=3.14159/180*HA*15
3170 LET V=3.14159/180*V
3180 LET CLA= COS (LA)
3190 LET SLA= SIN (LA)
3200 LET AL= SIN (V)*SLA+ COS (V)*CLA* COS (HA)
3210 LET AL= ASN (AL)
3220 LET AZ=( SIN (V)-SLA* SIN (AL))/(CLA* COS (AL))
3230 LET AZ= ACS (AZ)
3240 IF HA>0 THEN LET AZ= PI *2-AZ
3250 LET AL=180/3.14159*AL
3260 LET AZ=180/3.14159*AZ
3270 RETURN
3280 REM POSITION SUN
3290 LET RS=180/3.14159*(A(3)+ PI )/15
3300 IF RS>24 THEN LET RS=RS-24
3310 IF RS>24 THEN GOTO 3300
3320 IF RS<0 THEN LET RS=RS+24
3330 IF RS<0 THEN GOTO 3320
3340 LET VS= SIN (A(3)+ PI )*23.44194
3350 LET HS=T2-RS
3360 IF HS<-12 THEN LET HS=HS+24
3370 IF HS>12 THEN LET HS=HS-24
3380 LET HS=3.14159/180*HS*15
3390 LET VS=3.14159/180*VS
3400 LET AS= SIN (VS)*SLA+ COS (VS)*CLA* COS (HS)
3410 LET AS= ASN (AS)
3420 LET ZS=( SIN (VS)-SLA* SIN (AS))/(CLA* COS (AS))
3430 LET ZS= ACS (ZS)
3440 IF HS>0 THEN LET ZS= PI *2-ZS
3450 LET AS=180/3.14159*AS
3460 LET ZS=180/3.14159*ZS
3470 RETURN
3480 REM CALCS FOR MOON
3490 LET LP=255.7433
3500 LET LZ=311.1687
3510 LET LE=178.699
3520 LET LM=LZ+360*NI/27.32158
3530 LET LM=(LM/360- INT (LM/360))*360
3540 LET PG=.111404*NI+LP
3550 LET PG=(PG/360- INT (PG/360))*360
3560 LET PG=LM-PG
3570 LET DR=6.2886*( SIN (3.14159/180*PG))
3580 LET LM=LM+DR
3590 LET RQ=LM
3600 LET RM=LM/15
3610 IF RM>24 THEN LET RM=RM-24
3620 IF RM>24 THEN GOTO 3610
3630 IF RM<0 THEN LET RM=RM+24
3640 IF RM<0 THEN GOTO 3630
3650 LET AL=LE-NI*.052954
```


SKYPT (continued)

```
3660 LET AL=RQ-AL
3670 LET AL=(AL/360- INT (AL/360))*360
3680 LET HE=5.1333* SIN (AL* PI /180)
3690 LET DM=HE+23.1444* SIN (RQ* PI /180)
3700 LET HD=T2-RM
3710 IF HD<-12 THEN LET HD=HD+24
3720 IF HD>12 THEN LET HD=HD-24
3730 IF HD>12 OR HD<-12 THEN RETURN
3740 LET HA= PI /180*HD*15
3750 LET DM= PI /180*DM
3760 LET ML= SIN (DM)*SLA+ COS (DM)*CLA* COS (HA)
3770 LET ML= ASN (ML)
3780 LET MZ=( SIN (DM)-SLA* SIN (ML))/(CLA* COS (ML))
3790 LET MZ= ACS (MZ)
3800 IF HA>0 THEN LET MZ= PI *2-MZ
3810 LET MZ=180/ PI *MZ
3810 LET MZ=180/ PI *MZ
3820 LET ML=180/ PI *ML
3830 IF ML>85 OR ML<5 THEN RETJRN
3840 LET MX=MZ
3850 LET PM=RS+12-RQ/15
3860 IF PM>24 THEN LET PM=PM-24
3870 IF PM>12 THEN LET PM=PM-24
3880 IF PM >= -2 AND PM <= 2 THEN LET P$(10)="0"
3890 IF LA<0 THEN GOTO 3930
3900 IF PM<-2 THEN LET P$(10)="("
3910 IF PM>2 THEN LET P$(10)=")"
3920 RETURN
3930 IF PM<-2 THEN LET P$(10)=")"
3940 IF PM>2 THEN LET P$(10)="("
3950 RETURN
3960 REM INSTRUCTIONS
3970 CLS
3980 PRINT "PLANETS ETC. ARE"
3990 PRINT "IDENTIFIED AS... "
4000 PRINT
4010 PRINT "SUN..... O"
4020 PRINT "MERCURY... +"
4030 PRINT "VENUS..... V"
4040 PRINT "MARS..... M"
4050 PRINT "JUPITER... J"
4060 PRINT "SATURN.... S"
4070 PRINT "URANUS.... U"
4080 PRINT "NEPTUNE... N"
4090 PRINT "PLUTO..... P"
4100 PRINT "MOON"
4110 PRINT "  NEAR FULL   0"
4120 PRINT "  BEFORE FULL )"
4130 PRINT "  AFTER FULL  ("
4140 PRINT
4150 PRINT "PRESS ANY KEY"
4160 IF INKEY$ ="" THEN GOTO 4160
4170 CLS
4180 PRINT
4190 PRINT
4200 PRINT "YOU CAN CHANGE LATITUDE"
4210 PRINT "AND LONGITUDE WHEN ASKED."
4220 PRINT
```

SKYPT (continued)

```
4230 PRINT "THEN YOU TYPE THE DATE...."
4240 PRINT "YEAR, THEN MONTH (1 TO 12)"
4250 PRINT "THEN DAY."
4260 PRINT
4270 PRINT "NEXT YOU TYPE THE TIME ZONE"
4280 PRINT "AND THE LOCAL TIME IN DEC. HRS."
4290 PRINT
4300 PRINT "FINALLY; SELECT THE HORIZON"
4310 PRINT "YOU WANT TO OBSERVE."
4320 PRINT
4330 PRINT "NOTE:IF TWO PLANETS ARE CLOSE"
4340 PRINT "IN SKY, OUTER PLANET WILL"
4350 PRINT "OVERPRINT INNER PLANET"
4360 PRINT
4370 PRINT "PRESS ANY KEY"
4380 IF INKEY$ ="" THEN GOTO 4380
4390 CLS
4400 RETURN
```


Program 16: PLNTF

Finds and Plots Planets, Sun, and Moon in Constellations for Any Date and Time

When you select a date and a planet, the Sun, or the Moon, this program calculates where the object is located among the "fixed" stars of the celestial sphere. It selects a suitable zodiacal star chart (2 hours of right ascension and 30 degrees of declination), displays its name, and shows the object among the stars. Because of the limitations of resolution of any monitor screen, the selected object's accurate right ascension and declination is displayed before the object is shown on the star chart. The program displays other planets, the Sun, and the Moon if they are located in the same chart region on the date requested (see Figure 16.1). The planetary symbols used are the same as those used in the SKYPT program.

The program next asks you if you want another planet on the same date. If you answer 'Y', it will select an appropriate chart to display the new selection and again chart other planets in that new star chart. Other alternatives available are to ask for another date for the same planet, or another date and another planet.

The program operates by first loading planetary data from a data string, as in other programs, and then computing the right

ascension and declination of the planet you wish to find for the date you have requested. It displays this information on the monitor screen. Next the program determines which zodiacal region must be displayed. It then jumps to the part of the program containing the screen coordinates for the stars in that region of the sky, and it displays the appropriate stars on the monitor screen. Converting the right ascension and declination to screen coordinates, the program places the planet at its correct location in the constellation by plotting its character at the screen coordinate.

The program checks the right ascension of each of the other planets, the Sun, and the Moon, and if it falls within the range of right ascensions covered by the chart displayed on the monitor, the program displays the planet, Sun, or Moon at the correct location in the zodiacal constellation.

The listing of the PLNTF program follows.



Figure 16.1: *This display results if you ask the PLNTF program to find Mars on 25 February 1982. The planet (M) is plotted near to the star Spica in the constellation of Virgo. The display also shows any other planets in the same region of the zodiac. In this case Saturn (S) is the only other planet in this region of the sky.*

PLNTF

```
 10 REM PLNTF 4/16/83
 20 CLS
 30 CLEAR
 40 DIM A$(10)
 50 DIM U(9)
 60 DIM P(9,9)
 70 DIM N$(22)
 80 DIM M$(63)
 90 DIM P$(10,7)
100 DIM A(9)
110 DIM D(9)
120 DIM L(10)
130 DIM Q(9)
140 DIM R(10)
150 DIM V(10)
160 DIM K(9)
170 LET P2=2* PI
180 LET FD=57.29578
190 LET FR=.01745328
200 PRINT AT 3,5;"-----------------"
210 PRINT AT 4,5;"I PLANET FINDER I"
220 PRINT AT 5,5;"-----------------"
230 PRINT AT 7,5;"ASTRONOMY PROGRAM"
240 PRINT AT 8,3;"BY ERIC BURGESS F.R.A.S."
250 PRINT AT 10,4;"ALL RIGHTS RESERVED BY"
260 PRINT AT 11,3;"S AND T SOFTWARE SERVICE"
270 PRINT
280 PRINT
290 PRINT TAB 5;"THIS PROGRAM PLACES"
300 PRINT TAB 4;"A PLANET, OR THE SUN,"
310 PRINT TAB 4;"OR THE MOON AMONG THE"
320 PRINT TAB 3;"ZODIACAL CONSTELLATIONS"
330 PRINT TAB 8;"FOR ANY DATE."
340 PRINT
350 PRINT
360 PRINT "TYPE LONGITUDE ";
370 INPUT LO
380 PRINT LO
390 PAUSE 200
400 CLS
410 PRINT
420 PRINT
430 PRINT "BECAUSE OF LIMITATIONS OF"
440 PRINT "SCREEN RESOLUTION, RIGHT"
450 PRINT "ASCENSION AND DECLINATION OF"
460 PRINT "THE CHOSEN PLANET, SUN,"
470 PRINT "OR MOON ARE GIVEN BEFORE"
480 PRINT "ZODIACAL STAR CHART IS"
490 PRINT "DISPLAYED ON MONITOR SCREEN."
500 PRINT
510 PRINT "HORIZONTAL DOTTED LINE"
520 PRINT "ACROSS MIDDLE OF EACH STAR"
530 PRINT "CHART IS ECLIPTIC."
540 PRINT
550 PRINT
560 PRINT
570 PRINT "NOW LOADING ARRAYS"
```

PLNTF (continued)

```
580 GOSUB 1270
590 CLS
600 PRINT
610 PRINT "ENTER DATE"
620 PRINT
630 PRINT "       YEAR.... ";
640 INPUT Y
650 PRINT Y
660 PRINT
670 PRINT "       MONTH... ";
680 INPUT M
690 PRINT M
700 IF M<1 OR M>12 THEN GOTO 670
710 PRINT
720 PRINT "       DAY..... ";
730 INPUT DY
740 PRINT DY
750 IF DY<1 OR DY>31 THEN GOTO 720
760 IF M=2 AND DY>29 THEN GOTO 720
770 PRINT
780 LET DL=DY+24/LO
790 LET DG=365*Y+DL+((M-1)*31)
800 IF M >= 3 THEN GOTO 830
810 LET DG=DG+ INT ((Y-1)/4)- INT ((.75)* INT ((Y-1)/100+1))
820 GOTO 850
830 LET DG=DG- INT (M*.4+2.3)+ INT (Y/4)
840 LET DG=DG- INT ((.75)* INT ((Y/100)+1))
850 LET NI=DG-715875
860 PRINT "SELECT BY NUMBER.."
870 PRINT
880 PRINT "MERCURY(+)..1"
890 PRINT "VENUS(V)....2"
900 PRINT "             3...SUN(O)"
910 PRINT "MARS(M).....4"
920 PRINT "JUPITER(J)..5"
930 PRINT "SATURN(S)...6"
940 PRINT "URANUS(U)...7"
950 PRINT "NEPTUNE(N)..8"
960 PRINT "PLUTO(P)....9"
970 PRINT "            10..MOON())"
980 INPUT PS
990 CLS
1000 GOSUB 4550
1010 FAST
1020 GOSUB 1670
1030 SLOW
1040 PRINT
1050 PRINT P$(PS);" ON.. ";Y;" ";M;" ";DY
1060 PRINT "-:-:-:-:-:-:-:-:-:-:-"
1070 PRINT "           R.A.  DEC."
1080 PRINT "           HRS   DEG."
1090 PRINT "-:-:-:-:-:-:-:-:-:-:-"
1100 PRINT P$(PS); TAB 9;( STR$ (R(PS)))( TO 5);
1110 PRINT TAB 15;( STR$ (V(PS)))( TO 5)
1120 PRINT "-:-:-:-:-:-:-:-:-:-:-"
1130 PRINT
1140 PRINT "WANT COPY Y/N? "
```


PLNTF (continued)

```
1150 INPUT C$
1160 IF C$="N" THEN GOTO 1180
1170 COPY
1180 LET RA=R(PS)
1190 CLS
1200 GOSUB 2320
1210 LET Q$="+VOMJSUNP)"
1220 GOSUB 2520
1230 PRINT
1240 PRINT
1250 PRINT "ANOTHER DATE Y/N? "
1260 GOTO 2260
1270 LET U=1
1280 LET M$=".0714223.84840.3883011"
1290 LET M$=N$+".34041.387100.0797402.73514.122173.836013"
1300 GOSUB 1850
1310 LET N$=".0279623.02812.0131952"
1320 LET M$=N$+".28638.723300.0050603.85017.0593411.33168"
1330 GOSUB 1850
1340 LET N$=".0172021.74022.0320441"
1350 LET M$=N$+".785471.00000.0170003.339260000000000000"
1360 GOSUB 1850
1370 LET N$=".0091464.51234.1753015"
1380 LET M$=N$+".852091.52370.1417041.04656.031420.858702"
1390 GOSUB 1850
1400 LET N$=".0014504.53364.090478."
1410 LET M$=N$+"2391105.20280.2493741.76188.0197201.74533"
1420 GOSUB 1850
1430 LET N$=".0005844.89884.1055581"
1440 LET M$=N$+".610949.53850.5341563.12570.0436331.97746"
1450 GOSUB 1850
1460 LET N$=".0002052.46615.0885932"
1470 LET M$=N$+".9670619.1820.9015544.49084.0139601.28805"
1480 GOSUB 1850
1490 LET N$=".0001043.78556.016965."
1500 LET M$=N$+"77318130.0600.2705402.33498.0314162.29162"
1510 GOSUB 1850
1520 LET N$=".0000693.16948.4712393"
1530 LET M$=N$+".9130339.44009.860005.23114.3001971.91812"
1540 GOSUB 1850
1550 LET M$=""
1560 LET P$(1)="MERCURY"
1570 LET P$(2)="VENUS"
1580 LET P$(3)="SUN"
1590 LET P$(4)="MARS"
1600 LET P$(5)="JUPITER"
1610 LET P$(6)="SATURN"
1620 LET P$(7)="URANUS"
1630 LET P$(8)="NEPTUNE"
1640 LET P$(9)="PLUTO"
1650 LET P$(10)="MOON"
1660 RETURN
1670 LET I=1
1680 FOR J=1 TO 9
1690 GOSUB 1900
1700 LET A(I)=A
1710 LET D(I)=D
```

PLNTF (continued)

```
1720 LET L(I)=L
1730 LET I=I+1
1740 NEXT J
1750 FOR I=1 TO 9
1760 IF I=3 THEN NEXT I
1770 GOSUB 2020
1780 LET Q(I)=Q
1790 LET R(I)=R
1800 LET V(I)=V
1810 NEXT I
1820 FOR I=1 TO 9
1830 IF I=3 THEN NEXT I
1840 RETURN
1850 FOR K=1 TO 9
1860 LET P(U,K)= VAL (M$(1+((K-1)*7) TO (K*7)))
1870 NEXT K
1880 LET U=U+1
1890 RETURN
1900 LET A=NI*P(J,1)+P(J,2)
1910 IF A>P2 THEN LET A=((A/P2)- INT (A/P2))*P2
1920 IF A<0 THEN LET A=A+P2
1930 IF A<0 THEN GOTO 1920
1940 LET C=P(J,3)* SIN (A-P(J,4))
1950 LET A=A+C
1960 IF A>P2 THEN LET A=A-P2
1970 IF A<0 THEN LET A=A+P2
1980 IF A<0 THEN GOTO 1970
1990 LET D=P(J,5)+P(J,6)* SIN (A-P(J,7))
2000 LET L=P(J,8)* SIN (A-P(J,9))
2010 RETURN
2020 LET Z=A(3)-A(I)
2030 IF ABS (Z)> PI AND Z<0 THEN LET Z=Z+P2
2040 IF ABS (Z)> PI AND Z>0 THEN LET Z=Z-P2
2050 LET Q=D(I) ** 2+D(3) ** 2-2*D(I)*D(3)* COS (Z)
2060 LET Q= SQR (Q)
2070 LET P=(D(I)+D(3)+Q)/2
2080 LET X=2* ACS ( SQR (((P*(P-D(I)))/(D(3)*Q))))
2090 IF Z<0 THEN LET R=57.29578*(A(3)+3.14159-X)/15
2100 IF Z>0 THEN LET R=57.29578*(A(3)+3.14159+X)/15
2110 IF R>24 THEN LET R=R-24
2120 IF R>24 THEN GOTO 2110
2130 IF R<-24 THEN LET R=R+24
2140 IF R<-24 THEN GOTO 2130
2150 IF R<0 THEN LET R=R+24
2160 IF R<0 THEN GOTO 2150
2170 IF Z<0 THEN LET V= SIN (A(3)+ PI -X)*23.44194+FD*L(I)
2180 IF Z>0 THEN LET V= SIN (A(3)+ PI +X)*23.44194+FD*L(I)
2190 LET X=57.29578*X
2200 LET R(3)=(FD*A(3)-180)/15
2210 IF R(3)<0 THEN LET R(3)=R(3)+24
2220 IF R(3)>24 THEN LET R(3)=R(3)-24
2230 LET V(3)=( SIN (FR*(A(3)-180)))*23.44194
2240 LET L(3)=0
2250 RETURN
2260 LET A$=""
2270 INPUT A$
2280 IF A$="Y" THEN GOTO 2300
```


PLNTF (continued)

```
2290 GOTO 2310
2300 GOTO 590
2310 CLS
2320 IF RA >= 22 AND RA<24 THEN LET CA=22
2330 IF RA >= 20 AND RA<22 THEN LET CA=20
2340 IF RA >= 18 AND RA<20 THEN LET CA=18
2350 IF RA >= 16 AND RA<18 THEN LET CA=16
2360 IF RA >= 14 AND RA<16 THEN LET CA=14
2370 IF RA >= 12 AND RA<14 THEN LET CA=12
2380 IF RA >= 10 AND RA<12 THEN LET CA=10
2390 IF RA >= 8 AND RA<10 THEN LET CA=8
2400 IF RA >= 6 AND RA<8 THEN LET CA=6
2410 IF RA >= 4 AND RA<6 THEN LET CA=4
2420 IF RA >= 2 AND RA<4 THEN LET CA=2
2430 IF RA >= 0 AND RA<2 THEN LET CA=0
2440 LET RA=RA-CA
2450 LET CH=(CA+2)/2
2460 LET RA=RA*15
2470 IF (RA- INT (RA))>.49 THEN LET RA=1+ INT (RA)
2480 LET PL= INT (1.1*RA)
2490 LET CF=10-FD*(L(PS))
2500 LET PL=31-PL
2510 RETURN
2520 CLS
2530 FOR J=0 TO 31 STEP 2
2540 PRINT AT 10,J;"-"
2550 NEXT J
2560 IF CH=1 THEN GOTO 2680
2570 IF CH=2 THEN GOTO 2930
2580 IF CH=3 THEN GOTO 3080
2590 IF CH=4 THEN GOTO 3220
2600 IF CH=5 THEN GOTO 3370
2610 IF CH=6 THEN GOTO 3520
2620 IF CH=7 THEN GOTO 3680
2630 IF CH=8 THEN GOTO 3780
2640 IF CH=9 THEN GOTO 3840
2650 IF CH=10 THEN GOTO 3980
2660 IF CH=11 THEN GOTO 4100
2670 IF CH=12 THEN GOTO 4250
2680 LET C$="PISCES"
2690 GOSUB 2710
2700 GOTO 2840
2710 LET W$="          "
2720 IF FD*(L(PS))<-10 THEN GOTO 2750
2730 IF FD*(L(PS))>10 THEN GOTO 2770
2740 GOTO 2780
2750 LET W$="OFF CHART BOTTOM"
2760 GOTO 2780
2770 LET W$="OFF CHART TOP"
2780 PRINT AT 0,0;P$(PS);":IN ";C$;" ";W$
2790 PRINT AT 1,0;"ON ";Y;" ";M;" ";DY
2800 IF W$( TO 3)="OFF" THEN GOTO 2820
2810 PRINT AT 2,0;"AS DISPLAYED ON CHART BELOW"
2820 LET W$="          "
2830 RETURN
2840 PRINT AT 5,7;"."
2850 PRINT AT 5,28;"*"
2860 PRINT AT 8,16;"."
```

PLNTF (continued)

```
2870 PRINT AT 9,15;"."
2880 PRINT AT 11,4;"."
2890 PRINT AT 13,6;"."
2900 PRINT AT 17,4;"."
2910 PRINT AT 18,28;"."
2920 GOTO 4400
2930 LET C$="ARIES"
2940 GOSUB 2710
2950 PRINT AT 3,24;"X"
2960 PRINT AT 4,28;"*"
2970 PRINT AT 5,28;"*"
2980 PRINT AT 8,3;".:"
2990 PRINT AT 8,11;"."
3000 PRINT AT 13,28;"."
3010 PRINT AT 14,22;"."
3020 PRINT AT 17,25;"."
3030 PRINT AT 16,4;"*"
3040 PRINT AT 16,12;".."
3050 PRINT AT 19,18;"*"
3060 PRINT AT 19,24;"*"
3070 GOTO 4400
3080 LET C$="TAURUS"
3090 GOSUB 2710
3100 PRINT AT 4,8;"*"
3110 PRINT AT 11,7;"*"
3120 PRINT AT 11,15;"."
3130 PRINT AT 11,23;"."
3140 PRINT AT 12,24;"."
3150 PRINT AT 14,22;"X"
3160 PRINT AT 14,23;":"
3170 PRINT AT 14,25;"."
3180 PRINT AT 19,16;"."
3190 PRINT AT 20,8;"."
3200 PRINT AT 20,16;"."
3210 GOTO 4400
3220 LET C$="GEMINI"
3230 GOSUB 2710
3240 PRINT AT 4,5;"*"
3250 PRINT AT 5,7;"."
3260 PRINT AT 6,5;"."
3270 PRINT AT 7,17;"*"
3280 PRINT AT 9,12;"*"
3290 PRINT AT 11,15;"."
3300 PRINT AT 11,24;"*"
3310 PRINT AT 11,26;"*"
3320 PRINT AT 12,24;"."
3330 PRINT AT 14,20;"*"
3340 PRINT AT 17,17;"."
3350 PRINT AT 20,15;"*"
3360 GOTO 4400
3370 LET C$="CANCER"
3380 GOSUB 2710
3390 PRINT AT 7,20;"."
3400 PRINT AT 8,20;"."
3410 PRINT AT 12,5;"*"
3420 PRINT AT 13,15;"."
3430 PRINT AT 16,14;"*"
```


PLNTF (continued)

```
3440 PRINT AT 16,16;"*"
3450 PRINT AT 17,20;"."
3460 PRINT AT 17,23;"*"
3470 PRINT AT 18,10;"."
3480 PRINT AT 18,17;"."
3490 PRINT AT 19,15;"."
3500 PRINT AT 20,17;"."
3510 GOTO 4400
3520 LET C$="LEO"
3530 GOSUB 2710
3540 PRINT AT 3,15;"*"
3550 PRINT AT 3,28;"*"
3560 PRINT AT 4,12;"."
3570 PRINT AT 5,2;"."
3580 PRINT AT 5,29;"."
3590 PRINT AT 6,6;"."
3600 PRINT AT 8,11;"."
3610 PRINT AT 8,15;"."
3620 PRINT AT 9,2;"*"
3630 PRINT AT 9,22;"."
3640 PRINT AT 9,28;"X"
3650 PRINT AT 12,6;"."
3660 PRINT AT 16,10;"."
3670 GOTO 4400
3680 LET C$="VIRGO"
3690 GOSUB 2710
3700 PRINT AT 3,11;"*"
3710 PRINT AT 3,20;"*"
3720 PRINT AT 7,21;"*"
3730 PRINT AT 9,14;"."
3740 PRINT AT 9,26;"*"
3750 PRINT AT 12,11;"X"
3760 PRINT AT 19,19;"*"
3770 GOTO 4400
3780 LET C$="LIBRA"
3790 GOSUB 2710
3800 PRINT AT 4,15;"*"
3810 PRINT AT 9,17;"*"
3820 PRINT AT 17,13;"."
3830 GOTO 4400
3840 LET C$="SCORPIO"
3850 GOSUB 2710
3860 PRINT AT 3,7;"."
3870 PRINT AT 3,11;"."
3880 PRINT AT 5,8;"*"
3890 PRINT AT 5,14;"*"
3900 PRINT AT 8,28;"*"
3910 PRINT AT 11,29;"*"
3920 PRINT AT 12,23;"*"
3930 PRINT AT 13,21;"X"
3940 PRINT AT 14,29;"."
3950 PRINT AT 15,20;"*"
3960 PRINT AT 19,16;"*"
3970 GOTO 4400
3980 LET C$="SAGITTARIUS"
3990 GOSUB 2710
4000 PRINT AT 8,23;"."
```

PLNTF (continued)

```
4010 PRINT AT 9,14;"."
4020 PRINT AT 9,16;"."
4030 PRINT AT 12,24;"*"
4040 PRINT AT 13,17;"*"
4050 PRINT AT 14,19;"."
4060 PRINT AT 15,19;"*"
4070 PRINT AT 16,15;"*"
4080 PRINT AT 20,24;"*"
4090 GOTO 4400
4100 LET C$="CAPRICORNUS"
4110 GOSUB 2710
4120 PRINT AT 3,5;"*"
4130 PRINT AT 4,15;"."
4140 PRINT AT 4,23;"*"
4150 PRINT AT 6,23;"*"
4160 PRINT AT 11,11;"."
4170 PRINT AT 11,15;"*"
4180 PRINT AT 12,5;"*"
4190 PRINT AT 12,7;"*"
4200 PRINT AT 15,10;".."
4210 PRINT AT 16,20;"."
4220 PRINT AT 17,15;"."
4230 PRINT AT 17,19;"."
4240 GOTO 4400
4250 LET C$="AQUARIUS"
4260 GOSUB 2710
4270 PRINT AT 3,6;"."
4280 PRINT AT 3,11;"."
4290 PRINT AT 3,25;"*"
4300 PRINT AT 4,4;"."
4310 PRINT AT 4,9;"."
4320 PRINT AT 4,19;"*"
4330 PRINT AT 4,22;"*"
4340 PRINT AT 11,17;"."
4350 PRINT AT 14,14;".."
4360 PRINT AT 15,20;"."
4370 PRINT AT 17,20;"*"
4380 PRINT AT 18,1;"*"
4390 GOTO 4400
4400 IF CF<0 OR CF>20 THEN GOTO 4420
4410 PRINT AT CF,PL;Q$(PS)
4420 FOR N=1 TO 10
4430 IF N=11 THEN GOTO 4540
4440 IF N=PS THEN GOTO 4530
4450 LET R2=R(N)-(CH*2-2)
4460 IF R2<0 OR R2>1.9999 THEN GOTO 4530
4470 LET PL=R2*15
4480 IF (PL- INT (PL))>.49 THEN LET PL=1+ INT (PL)
4490 LET PL=31- INT (1.1*PL)
4500 LET CF=10-FD*(L(N))
4510 IF CF<0 OR CF>20 THEN GOTO 4530
4520 PRINT AT CF,PL;Q$(N)
4530 NEXT N
4540 GOTO 4730
4550 LET NM=NI-.4
4560 LET PG=.111404*NM+255.7433
4570 LET LMD=311.1687+360*NM/27.32158
```


PLNTF (continued)

```
4580 LET PG=LMD-PG
4590 LET PG=(PG/360- INT (PG/360))*360
4600 LET DR=6.2886* SIN (FR*PG)
4610 LET LMD=LMD+DR
4620 LET LMD=(LMD/360- INT (LMD/360))*360
4630 LET R(10)=LMD/15
4640 LET AL=178.699-NM*.052954
4650 LET AL=(AL/360- INT (AL/360))*360
4660 LET AL=LMD-AL
4670 IF AL<0 THEN LET AL=AL+360
4680 IF AL>360 THEN LET AL=AL-360
4690 LET HE=5.1454*( SIN (AL*FR))
4700 LET L(10)=HE*FR
4710 LET V(10)=HE+23.1444* SIN (LMD*FR)
4720 RETURN
4730 PRINT AT 21,0;"WANT COPY Y/N?"
4740 INPUT G$
4750 IF G$="N" THEN GOTO 4770
4760 COPY
4770 CLS
4780 PRINT "ANOTHER PLANET Y/N?"
4790 INPUT H$
4800 IF H$="Y" THEN GOTO 860
4810 PRINT
4820 PRINT "ANOTHER DATE Y/N?"
4830 INPUT I$
4840 IF I$="Y" THEN GOTO 590
4850 CLS
4860 STOP
```

Program 17: JSATS

Positions of Galilean Satellites of Jupiter for Any Date and Time

This program allows you to compute the positions of Jupiter's four largest satellites, Io, Europa, Ganymede, and Callisto, relative to Jupiter as seen from Earth, and it identifies the satellites.

With the program you will never be in doubt as to the identity of each of the four starlike objects. The program can be used worldwide; it asks for your time zone and local time. It will display several time intervals on the screen. The display can be oriented with either north or south at the top to suit observers with different types of telescopes and for Northern or Southern Hemisphere viewing.

When you run this program, the computer first displays the distances of the satellites in terms of Jupiter radii for exact observations (see Figure 17.1). When you are ready to continue, it displays the satellites and the planet, identified as follows: J, Jupiter; I, Io; E, Europa; G, Ganymede; and C, Callisto (see Figure 17.2).

Jupiter is always located centrally on the display line. It also shows which format, north or south at top, is being used. The

program does not show positions out of the horizontal plane, so sometimes if two of the satellites are visible, one slightly above the other at the same angular distance from Jupiter, the display for the outermost replaces the symbol for the inner satellite. You can quickly ascertain which is which by asking for displays a few hours on either side. The display loses the satellites when they are in occultation or transit—that is, they are erased from the screen by Jupiter. Again, by asking for other displays at earlier times you can identify transits and occultations. With north at top, satellites moving toward Jupiter from the right of the planet will go into occultation behind Jupiter; those moving in the opposite direction will go into transit across Jupiter. With south at top, satellite movement from the right results in transit and from the left it results in occultation.

You can add to the program and simplify the sorting out of occultations and transits by showing an orbit projection, using the plotting technique given in the MARSP program.

```
DISTANCES IN JUPITER RADII
 WHICH IS 71,100 KM(44,180 MI)

FOR 1983 9 24 AT 6 HRS UT
    1983 9 23 AT 22 HRS ZN 8

   IO(I)..... -2.0
   EUROPA(E).. 8.90
   GANYMEDE(G) 14.8
   CALLISTO(C) 12.9

MINUS IS TO LEFT OF JUPITER
   PLUS IS TO THE RIGHT

  WITH S AT TOP

WANT COPY Y/N?

"■"
```

Figure 17.1: The JSATS program provides this information in its first display. It gives the distances of the Galilean satellites from Jupiter in terms of Jovian radii (44,180 miles).

The program first calculates the number of days from the epoch of 1900, adjusting for time and time zone. It uses this calculation to determine the position of each satellite in its orbit around Jupiter at the requested date. Next it determines the relative positions of Earth and Jupiter at the requested date and corrects for the viewing angle from Earth. Then it calculates the radial distance of each satellite from Jupiter (in terms of the radius of Jupiter) as viewed from Earth's position, and it displays this information on the monitor screen. When you are ready to continue, the program displays the satellites together with Jupiter on one line across the screen, at proportional distances from Jupiter.

Next the program increments the time and date as requested and develops second and third displays. Housekeeping routines adjust for month and year ends in the display of dates for each configuration of the satellites.

The listing of the JSATS program follows.



Figure 17.2: Next the JSATS program generates a graphic display showing the positions of the satellites relative to Jupiter For this display three configurations were selected, 24 hours apart. The movements of the satellites relative to the planet are clearly shown

JSATS

```
 10 CLEAR
 20 REM JSATS 4/16/83
 30 CLS
 40 LET FL=0
 50 DIM Z$(4)
 60 LET MQ=3
 70 DIM B$(4)
 80 LET RD= PI /180
 90 PRINT AT 4,6;"ASTRONOMY PROGRAM"
100 PRINT AT 6,2;"------------------------"
110 PRINT AT 7,2;"I  GALILEAN SATELLITES  I"
120 PRINT AT 8,2;"------------------------"
130 PRINT AT 10,3;"BY ERIC BURGESS F.R.A.S."
140 PRINT AT 12,4;"ALL RIGHTS RESERVED BY"
150 PRINT AT 13,3;"S AND T SOFTWARE SERVICE"
160 PAUSE 250
170 CLS
180 PRINT AT 4,0;"SHOWS THE GALILEAN SATELLITES"
190 PRINT AT 5,0;"POSITIONED RELATIVE TO JUPITER"
200 PRINT AT 6,2;"IN THEIR RELATIVE POSITIONS"
210 PRINT AT 9,9;"SELECT"
220 PRINT AT 10,0;"NORTH OR SOUTH AT TOP (N/S)?"
230 INPUT S$
240 PRINT AT 15,2;"DISPLAY WILL BE ";S$;" AT TOP"
250 PAUSE 200
260 IF S$="S" THEN LET FL=1
270 CLS
280 PRINT
290 LET LY=0
300 PRINT
310 PRINT "ENTER YEAR... ";
320 INPUT Y
330 PRINT Y
340 PRINT
350 PRINT "ENTER MONTH.. ";
360 INPUT M
370 PRINT M
380 IF M<1 OR M>12 THEN GOTO 350
390 PRINT
400 PRINT "ENTER DAY.... ";
410 INPUT DA
420 PRINT DA
430 IF DA<1 OR DA>31 THEN GOTO 400
440 LET Y4=Y/4
450 LET Y5=Y/100
460 IF Y4- INT Y4=0 AND Y5- INT Y5 <> 0 THEN LET LY=1
470 IF M=2 AND DA=29 AND LY=0 THEN GOTO 500
480 IF M=2 AND DA=30 AND LY=1 THEN GOTO 500
490 GOTO 520
500 PRINT "INVALID DAY"
510 GOTO 400
520 PRINT
530 PRINT "TIME ZONE..... ";
540 INPUT TZ
550 PRINT TZ
560 PRINT
570 PRINT "LOCAL TIME DEC.HR. ";
580 INPUT LT
590 PRINT LT
```

JSATS (continued)

```
 600 PRINT
 610 PRINT "NUMBER OF DISPLAYS (1 TO 4) ";
 620 INPUT ND
 630 PRINT ND
 640 PRINT
 650 PRINT "TIME INTERVAL IN HRS ";
 660 INPUT IN
 670 PRINT IN
 680 PRINT
 690 PRINT
 700 CLS
 710 FAST
 720 LET L=1
 730 LET YP=Y
 740 LET MP=M
 750 LET DP=DA
 760 IF L>1 THEN GOTO 780
 770 LET UT=LT+TZ
 780 IF UT<0 THEN GOTO 990
 790 IF UT >= 0 AND UT<24 THEN GOTO 1000
 800 LET UT=UT-24
 810 LET DA=DA+1
 820 IF (LY=1 AND M=2 AND DA=30) THEN GOTO 850
 830 IF (LY=0 AND M=2 AND DA=29) THEN GOTO 850
 840 GOTO 880
 850 LET M=3
 860 LET DA=1
 870 GOTO 960
 880 IF DA <= 31 THEN GOTO 960
 890 IF (DA>30 AND M=4 OR M=6 OR M=9 OR M=11) THEN GOTO 910
 900 GOTO 940
 910 LET M=M+1
 920 LET DA=1
 930 GOTO 960
 940 IF DA>31 THEN LET M=M+1
 950 LET DA=1
 960 IF M=13 THEN LET Y=Y+1
 970 IF M=13 THEN LET M=1
 980 GOTO 1000
 990 GOSUB 2720
1000 IF FL=1 THEN GOTO 1040
1010 LET YP=Y
1020 LET MP=M
1030 LET DP=DA
1040 LET T=UT/24
1050 LET YC=Y
1060 LET MC=M
1070 LET DC=DA+T
1080 IF MC>2 THEN GOTO 1110
1090 LET YC=Y-1
1100 LET MC=M+12
1110 LET TD= INT (365.25*YC)+ INT ((MC+1)*30.6001)+DC
1120 LET YZ= INT (YC/100)
1130 LET I=TD-YZ+2+ INT (YZ/4)-694025.5
1140 LET A=I*203.40586+84.55061
1150 LET A=(A/360- INT (A/360))*360
1160 LET B=I*101.2916323+41.50155
1170 LET B=(B/360- INT (B/360))*360
1180 LET C=I*50.23451687+109.97702
1190 LET C=(C/360- INT (C/360))*360
```


JSATS (continued)

```
1200 LET D=I*21.48798021+176.35864
1210 LET D=(D/360- INT (D/360))*360
1220 LET P1=I*.9856003+358.476
1230 LET P1=(P1/360- INT (P1/360))*360
1240 LET I2=I*.0830853+225.328
1250 LET I2=(I2/360- INT (I2/360))*360
1260 LET P2=I*.9025179+221.647
1270 LET P2=(P2/360- INT (P2/360))*360
1280 LET P2=P2+( SIN (2*P1*RD)/50)+( SIN (P1*RD)/.521)
1290 LET P3=( SIN (2*I2*RD)*.1673)+( SIN (I2*RD)*5.5372)
1300 LET P2=P2-P3
1310 LET P1= SQR (28.07-( COS (P2*RD)*10.4056))
1320 LET J=( SIN (P2*RD)/P1)
1330 LET SN= SIN (RD*J)
1340 LET SN=SN*SN
1350 LET SN=1-SN
1360 LET SN= SQR SN
1370 LET I3=57.2958*( ATN (( SIN (RD*J))/SN))
1380 LET I3=57.2958*I3
1390 LET P1=P1/173
1400 LET IO=5.906* SIN (RD*(A+I3-P3-(P1*203.405863)))
1410 LET EU=9.397* SIN (RD*(B+I3-P3-(P1*101.2916323)))
1420 LET GA=14.989* SIN (RD*(C+I3-P3-(P1*50.23451684)))
1430 LET CA=26.364* SIN (RD*(D+I3-P3-(P1*21.48798021)))
1440 IF FL=0 THEN GOTO 1490
1450 LET IO=0-IO
1460 LET EU=0-EU
1470 LET GA=0-GA
1480 LET CA=0-CA
1490 IF L>1 THEN GOTO 1560
1500 SLOW
1510 CLS
1520 PRINT AT 4,0;"DISTANCES IN JUPITER RADII"
1530 PRINT AT 5,1;"WHICH IS 71,100 KM(44,180 MI)"
1540 PRINT AT 7,0;"FOR ";Y;" ";M;" ";DA;" AT ";UT;" HRS UT"
1550 PRINT AT 8,4;YP;" ";MP;" ";DP;" AT ";LT;" HRS ZN ";TZ
1560 LET EM=M
1570 LET S$= STR$ (IO)
1580 LET T$= STR$ (EU)
1590 LET U$= STR$ (GA)
1600 LET V$= STR$ (CA)
1610 IF L>1 THEN GOTO 1830
1620 PRINT AT 10,3;"IO(I)..... ";S$(1 TO 4)
1630 PRINT AT 11,3;"EUROPA(E).. ";T$(1 TO 4)
1640 PRINT AT 12,3;"GANYMEDE(G) ";U$(1 TO 4)
1650 PRINT AT 13,3;"CALLISTO(C) ";V$(1 TO 4)
1660 PRINT AT 15,0;"MINUS IS TO LEFT OF JUPITER"
1670 PRINT AT 16,3;"PLUS IS TO THE RIGHT"
1680 IF FL <> 1 THEN GOTO 1720
1690 PRINT AT 18,2;"WITH S AT TOP"
1700 PRINT
1710 GOTO 1740
1720 PRINT AT 18,2;"WITH N AT TOP"
1730 PRINT
1740 PRINT "WANT COPY Y/N? ";
1750 INPUT X$
1760 PRINT X$
1770 IF X$="N" THEN GOTO 1800
1780 COPY
1790 PRINT
```

JSATS (continued)

```
1800 PRINT AT 20,0;"PRESS ANY KEY FOR DISPLAY"
1810 IF INKEY$ ="" THEN GOTO 1810
1820 CLS
1830 GOSUB 2580
1840 LET LT=UT-TZ
1850 IF LT>0 THEN GOTO 2010
1860 LET DP=DA-1
1870 LET LT=LT+24
1880 IF DP>0 THEN GOTO 2010
1890 IF DP<1 THEN LET MP=EM-1
1900 IF (LY=1 AND MP=2) THEN GOTO 1940
1910 IF (LY=0 AND MP=2) THEN GOTO 1960
1920 IF (MP=4 OR MP=6 OR MP=9 OR MP=11) THEN GOTO 1980
1930 GOTO 2000
1940 LET DP=29
1950 GOTO 2010
1960 LET DP=28
1970 GOTO 2010
1980 LET DP=30
1990 GOTO 2010
2000 LET DP=31
2010 IF MP>0 THEN GOTO 2040
2020 LET MP=12
2030 LET YP=Y-1
2040 IF L>1 THEN GOTO 2120
2050 IF FL=0 THEN GOTO 2080
2060 PRINT TAB 15;"S"
2070 GOTO 2090
2080 PRINT TAB 15;"N"
2090 PRINT "JUPITER..J"; TAB 15;":"; TAB 21;"IO.....I"
2100 PRINT "EUROPA...E"; TAB 15;":"; TAB 21;"GANYMEDE G"
2110 PRINT "CALLISTO C"; TAB 15;":"
2120 PRINT "----------------------------"
2130 PRINT YP;" ";MP;" ";DP;" AT ";LT;" HR ZONE ";TZ
2140 PRINT
2150 PRINT AT MQ,M1;"I"
2160 PRINT AT MQ,M2;"E"
2170 PRINT AT MQ,M3;"G"
2180 PRINT AT MQ,M4;"C"
2190 PRINT AT MQ,M5;"J"
2200 IF L=ND THEN GOTO 2420
2210 LET UT=UT+IN
2220 IF UT>24 THEN GOTO 2240
2230 GOTO 2260
2240 LET UT=UT-24
2250 LET DA=DA+1
2260 IF (LY=1 AND M=2 AND DA=30) THEN GOTO 2310
2270 IF (LY=0 AND M=2 AND DA=29) THEN GOTO 2310
2280 IF DA<31 THEN GOTO 2360
2290 IF (DA>30 AND M=4 OR M=6 OR M=9 OR M=11) THEN GOTO 2340
2300 IF DA>31 THEN GOTO 2340
2310 LET M=3
2320 LET DA=1
2330 GOTO 2360
2340 LET M=M+1
2350 LET DA=1
2360 IF M=13 THEN GOTO 2380
2370 GOTO 2400
2380 LET M=1
```


JSATS (continued)

```
2390 LET Y=Y+1
2400 LET L=L+1
2410 GOTO 730
2420 PRINT
2430 PRINT "WANT COPY Y/N? ";
2440 INPUT H$
2450 PRINT H$
2460 IF H$="Y" THEN COPY
2470 PRINT
2480 PRINT "WANT ANOTHER DATE Y/N?";
2490 INPUT Y$
2500 IF Y$="Y" OR Y$="N" THEN GOTO 2530
2510 PRINT "INCORRECT RESPONSE"
2520 GOTO 2480
2530 IF Y$="N" THEN GOTO 2700
2540 CLS
2550 LET FL=0
2560 LET MQ=3
2570 GOTO 220
2580 LET S=( VAL S$(1 TO 4))
2590 LET T=( VAL T$(1 TO 4))
2600 LET U=( VAL U$(1 TO 4))
2610 LET V=( VAL V$(1 TO 4))
2620 LET MX=15
2630 LET MQ=MQ+4
2640 LET M1=MX+ INT (S*.55)
2650 LET M2=MX+ INT (T*.55)
2660 LET M3=MX+ INT (U*.55)
2670 LET M4=MX+ INT (V*.55)
2680 LET M5=MX
2690 RETURN
2700 CLS
2710 STOP
2720 LET UT=UT+24
2730 LET DA=DA-1
2740 IF DA<1 THEN GOTO 2760
2750 RETURN
2760 LET M=M-1
2770 IF M=4 OR M=6 OR M=9 OR M=11 THEN GOTO 2800
2780 LET DA=31
2790 GOTO 2820
2800 LET DA=30
2810 RETURN
2820 IF M <> 0 THEN GOTO 2860
2830 LET M=12
2840 LET Y=Y-1
2850 RETURN
2860 IF M=2 THEN GOTO 2880
2870 RETURN
2880 IF LY=0 THEN LET DA=28
2890 IF LY=1 THEN LET DA=29
2900 RETURN
```

PART 4



GENERAL AND TUTORIAL

There is a basic human fascination in the night sky, which is especially captivating when our view is unhindered by the glaring lights of the cities. The stars beckon—bright diamonds sparkling in the blackness, some glaring to our dark-adjusted night vision, others elusively winking in and out at the limit of visibility. At first the stars and planets seem distant and aloof, but once you learn to recognize them, they represent both a comforting presence and a new horizon.

This group of programs helps you find and recognize constellations and bright stars. One of the programs gives helpful information for photographing planets, and another provides useful astronomical conversions.

Program 18: ACONV

Useful Astronomical Conversions

This program is a straightforward conversion routine useful when astronomical measurements need to be expressed in other units. It includes light years, astronomical units, parsecs, telescope resolving power, and metric conversions. It allows conversions either way. If you desire other conversions, these can be added easily, following the pattern shown.

The program offers a menu (see Figure 18.1), from which you select the conversion needed. You can then continue with that conversion as long as you wish or return to the menu after any calculation.

The listing of the ACONV program follows.


```
ASTRONOMY CONVERSION PROGRAM

YOU CAN HAVE...

KM - MI....(1)  OR MI - KM....(2)
LY - MI....(3)  OR MI - LY....(4)
PARS. - LY.(5)  OR LY - PARS..(6)
C - F......(7)  OR F - C......(8)
A.U. - MI..(9)  OR MI - A.U. (10)
A.U. - KM..(9)  OR KM - A.U. (10)
DEG.MIN.SEC. - DECIMAL DEG. (11)
DECIMAL DEG. - DEG.MIN.SEC. (12)
HRS.MIN.SEC. - DECIMAL HRS. (13)
DECIMAL HRS. - HRS.MI.SEC.. (14)
RES.PWR. - INS. OF APERTURE (15)
INS. OF APERTURE - RES.PWR. (16)

TYPE 999 TO END OR TO CHANGE
TO ANOTHER CONVERSION

SELECT CONVERSION REQUIRED 1-16
```

Figure 18.1: The menu of astronomical conversions available on the ACONV program

ACONV

```
10 REM ACONV 4/16/83
20 PRINT AT 5,1;"--------------------------"
30 PRINT AT 6,1;"I       CONVERSIONS      I"
40 PRINT AT 7,1;"--------------------------"
50 PRINT
60 PRINT
70 PRINT TAB 5;"AN ASTRONOMY PROGRAM"
80 PRINT TAB 3;"BY ERIC BURGESS F.R.A.S."
90 PRINT
100 PRINT TAB 4;"ALL RIGHTS RESERVED BY"
110 PRINT TAB 3;"S AND T SOFTWARE SERVICE"
120 PRINT
130 PAUSE 250
140 CLS
150 PRINT "ASTRONOMY CONVERSION PROGRAM"
160 PRINT
170 PRINT "YOU CAN HAVE..."
180 PRINT
190 PRINT "KM - MI....(1) OR MI - KM....(2)"
200 PRINT "LY - MI....(3) OR MI - LY....(4)"
210 PRINT "PARS. - LY.(5) OR LY - PARS..(6)"
220 PRINT "C - F......(7) OR F - C......(8)"
230 PRINT "A.U. - MI..(9) OR MI - A.U. (10)"
```

ACONV (continued)

```
240 PRINT "A.U. - KM..(9) OR KM - A.U. (10)"
250 PRINT "DEG.MIN.SEC. - DECIMAL DEG. (11)"
260 PRINT "DECIMAL DEG. - DEG.MIN.SEC. (12)"
270 PRINT "HRS.MIN.SEC. - DECIMAL HRS. (13)"
280 PRINT "DECIMAL HRS. - HRS.MI.SEC.. (14)"
290 PRINT "RES.PWR. - INS. OF APERTURE (15)"
300 PRINT "INS. OF APERTURE - RES.PWR. (16)"
310 PRINT
320 PRINT "TYPE 999 TO END OR TO CHANGE"
330 PRINT "TO ANOTHER CONVERSION"
340 PRINT
350 PRINT "SELECT CONVERSION REQUIRED 1-16 ";
360 INPUT D
370 PRINT D
380 IF D>0 AND D<17 OR D=999 THEN GOTO 410
390 CLS
400 GOTO 140
410 CLS
420 IF D=999 THEN GOTO 2770
430 IF D=1 THEN GOTO 630
440 IF D=2 THEN GOTO 740
450 IF D=3 THEN GOTO 850
460 IF D=4 THEN GOTO 960
470 IF D=5 THEN GOTO 1070
480 IF D=6 THEN GOTO 1180
490 IF D=7 THEN GOTO 1290
500 IF D=8 THEN GOTO 1400
510 IF D=9 THEN GOTO 1510
520 IF D=10 THEN GOTO 1640
530 IF D=11 THEN GOTO 1880
540 IF D=12 THEN GOTO 2060
550 IF D=13 THEN GOTO 2200
560 IF D=14 THEN GOTO 2380
570 IF D=15 THEN GOTO 2520
580 IF D=16 THEN GOTO 2640
590 CLS
600 PRINT "PLEASE SELECT 1 THRU 16 OR 999"
610 PAUSE 100
620 GOTO 160
630 PRINT
640 PRINT "KM TO MI CONVERSION"
650 PRINT "TO STOP ENTER 999"
660 PRINT
670 PRINT "KM? ";
680 INPUT X
690 PRINT X
700 IF X=999 THEN GOTO 140
710 LET Y=X*(1/1.60934)
720 PRINT "MI = ";Y
730 GOTO 660
740 PRINT
750 PRINT "MI TO KM CONVERSION"
760 PRINT "TO STOP ENTER 999"
770 PRINT
780 PRINT "MI? ";
790 INPUT X
800 PRINT X
```


ACONV (continued)

```
810 IF X=999 THEN GOTO 140
820 LET Y=X*1.609344
830 PRINT "KM = ";Y
840 GOTO 770
850 PRINT
860 PRINT "LY TO TRILLION MI CONVERSION"
870 PRINT "TO STOP ENTER 999"
880 PRINT
890 PRINT "LY? ";
900 INPUT X
910 PRINT X
920 IF X=999 THEN GOTO 140
930 LET Y$= STR$ (X*5.88)
940 PRINT "TRILLION MI = "; VAL Y$(1 TO 5)
950 GOTO 880
960 PRINT
970 PRINT "TRILLION MI TO LY CONVERSION"
980 PRINT "TO STOP ENTER 999"
990 PRINT
1000 PRINT "TRILLION MI? ";
1010 INPUT X
1020 PRINT X
1030 IF X=999 THEN GOTO 140
1040 LET Y$= STR$ (X*1/5.88)
1050 PRINT "LY = ";Y$(1 TO 5)
1060 GOTO 990
1070 PRINT
1080 PRINT "PARSECS TO LY CONVERSION"
1090 PRINT "TO STOP ENTER 999"
1100 PRINT
1110 PRINT "PARSECS? ";
1120 INPUT X
1130 PRINT X
1140 IF X=999 THEN GOTO 140
1150 LET Y$= STR$ (X*3.26001)
1160 PRINT "LY = ";Y$(1 TO 5)
1170 GOTO 1100
1180 PRINT
1190 PRINT "LY TO PARSECS CONVERSION"
1200 PRINT "TO STOP ENTER 999"
1210 PRINT
1220 PRINT "LY? ";
1230 INPUT X
1240 PRINT X
1250 IF X=999 THEN GOTO 140
1260 LET Y$= STR$ (X*1/3.26000001)
1270 PRINT "PARSECS = ";Y$(1 TO 5)
1280 GOTO 1210
1290 PRINT
1300 PRINT "C - F CONVERSION"
1310 PRINT "TO STOP ENTER 999"
1320 PRINT
1330 PRINT "C? ";
1340 INPUT X
1350 PRINT X
1360 IF X=999 THEN GOTO 140
```

ACONV (continued)

```
1370 LET Y=X*1.8+32
1380 PRINT "F = ";Y
1390 GOTO 1320
1400 PRINT
1410 PRINT "F - C CONVERSION"
1420 PRINT "TO STOP ENTER 999"
1430 PRINT
1440 PRINT "F? ";
1450 INPUT X
1460 PRINT X
1470 IF X=999 THEN GOTO 140
1480 LET Y=(X-32)*.555556
1490 PRINT "C = ";Y
1500 GOTO 1430
1510 PRINT
1520 PRINT "A.U. TO MI/KM CONVERSION"
1530 PRINT "TO STOP ENTER 999"
1540 PRINT
1550 PRINT "A.U.? ";
1560 INPUT X
1570 PRINT X
1580 IF X=999 THEN GOTO 140
1590 LET Y$= STR$ (X*92.95721)
1600 LET Z$= STR$ (X*149.589)
1610 PRINT "MI = ";Y$(1 TO 5)
1620 PRINT "KM = ";Z$(1 TO 5)
1630 GOTO 1540
1640 PRINT
1650 PRINT "MI/KM TO A.U. CONVERSION"
1660 PRINT "TO STOP ENTER 999"
1670 PRINT
1680 PRINT "TYPE KM OR MI ";
1690 INPUT Q$
1700 PRINT Q$
1710 IF Q$="KM" THEN GOTO 1740
1720 IF Q$="MI" THEN GOTO 1810
1730 GOTO 1680
1740 PRINT "KM (MILLIONS) ";
1750 INPUT X
1760 PRINT X
1770 IF X=999 THEN GOTO 140
1780 LET Y$= STR$ (X*(1/149.598))
1790 PRINT "A.U. = ";Y$(1 TO 5)
1800 GOTO 1740
1810 PRINT "MI (MILLIONS) ";
1820 INPUT X
1830 PRINT X
1840 IF X=999 THEN GOTO 140
1850 LET Y$= STR$ (X*(1/92.15721))
1860 PRINT "A.U. = ";Y$(1 TO 5)
1870 GOTO 1810
1880 PRINT
1890 PRINT "DEG.MIN.SEC. TO DECIMAL DEG."
1900 PRINT "TO STOP ENTER 999"
1910 PRINT
1920 PRINT "DEGREES ? ";
```


ACONV (continued)

```
1930 INPUT D
1940 IF D=999 THEN GOTO 140
1950 PRINT D
1960 PRINT "MINUTES ? ";
1970 INPUT M
1980 PRINT M
1990 PRINT "SECONDS ? ";
2000 INPUT S
2010 PRINT S
2020 LET Y=D+M/60+S/3600
2030 PRINT "DECIMAL DEG = ";Y
2040 PRINT
2050 GOTO 1910
2060 PRINT
2070 PRINT "DECIMAL DEG. TO DEG.MIN.SEC."
2080 PRINT "TO STOP TYPE 999"
2090 PRINT
2100 PRINT "DECIMAL DEG. ? ";
2110 INPUT X
2120 PRINT X
2130 IF X=999 THEN GOTO 140
2140 LET Y= INT (X)
2150 LET M=60*(X- INT (X))
2160 LET S=60*(M- INT (M))
2170 PRINT "DEG.MIN.SEC.= ";Y;" "; INT (M);" "; INT S
2180 PRINT
2190 GOTO 2100
2200 PRINT
2210 PRINT "HRS.MIN.SEC. TO DECIMAL HRS."
2220 PRINT "TO STOP TYPE 999"
2230 PRINT
2240 PRINT "HOURS ? ";
2250 INPUT X
2260 PRINT X
2270 IF X=999 THEN GOTO 140
2280 PRINT "MINUTES ? ";
2290 INPUT M
2300 PRINT M
2310 PRINT "SECONDS ? ";
2320 INPUT S
2330 PRINT S
2340 LET Y=X+M/60+S/3600
2350 PRINT "DECIMAL HRS. = ";Y
2360 PRINT
2370 GOTO 2240
2380 PRINT
2390 PRINT "DECIMAL HRS. TO HRS.MIN.SEC."
2400 PRINT "TO STOP TYPE 999"
2410 PRINT
2420 PRINT "DECIMAL HRS. ? ";
2430 INPUT X
2440 PRINT X
2450 IF X=999 THEN GOTO 140
2460 LET Y= INT (X)
2470 LET M=60*(X- INT (X))
2480 LET S=60*(M- INT (M))
2490 PRINT "HRS.MIN.SEC. = ";Y;" "; INT (M);" "; INT (S)
```

ACONV (continued)

```
2500 PRINT
2510 GOTO 2420
2520 PRINT
2530 PRINT "RESOLVING POWER TO APERTURE"
2540 PRINT "TO STOP ENTER 999"
2550 PRINT
2560 PRINT "RES.POWER ARCSEC ? ";
2570 INPUT X
2580 PRINT X
2590 IF X=999 THEN GOTO 140
2600 LET Y=4.56/X
2610 PRINT "APERTURE IN INCHES = ";Y
2620 PRINT
2630 GOTO 2560
2640 PRINT
2650 PRINT "APERTURE TO RESOLVING POWER"
2660 PRINT "TO STOP ENTER 999"
2670 PRINT
2680 PRINT "APERTURE IN INCHES ? ";
2690 INPUT X
2700 PRINT X
2710 IF X=999 THEN GOTO 140
2720 LET Y=4.56/X
2730 PRINT "RESOLVING POWER"
2740 PRINT "   IN ARCSEC.= ";Y
2750 PRINT
2760 GOTO 2680
2770 CLS
2780 STOP
```


Program 19: CONST

Recognizing the Constellations

This program is designed to help you learn the constellations by pattern recognition. It does not show the imaginary lines of the star charts. But when you remember the patterns of bright and faint stars, you should have no problem in identifying the constellation in the night sky. It does not take long to become familiar with the constellations. Unfortunately, not all are visible at any one time or season. With this program you can learn to recognize many major constellations before going out to search for them in the night sky. Readers in the Southern Hemisphere should refer to the book *Celestial BASIC* for a supplementary program to identify the southern circumpolar constellations.

The program offers two alternatives. You can be shown a constellation and asked to name it, and then be asked to name the bright star within it. Or you can have questions mingling stars and constellations unrelated to each other.

The program randomly selects the constellations for display and gives you two tries at naming the chosen constellation correctly. Then it tells you the name, so you should remember to associate it with the star pattern the next time you see it. You can continue being tested as long as you wish, or you can quit at any time. Figure 19.1, which shows Cassiopeia, is typical of the displays generated by this program.

There are no complex algorithms in the program. It selects the constellations by use of the random function and jumps to the appropriate section of the program. Each section is a straightforward routine that prints stars on the monitor screen and then uses a subroutine to ask you to name the constellation. It checks if your answer is correct. If your answer is incorrect it allows you one more try before printing the name of the constellation. A similar subroutine is used to quiz you on the name of a bright star within the constellation.

If you wish you can add a second subroutine to display other stars for a quiz. You can also use the same techniques to display nebulae when you want a more advanced tutorial, and you want to learn where such objects are located. These would be interesting projects for an astronomy class. If you are in the Southern Hemisphere, you probably will find it preferable to invert all the constellations when you key in the routines for each constellation by changing the PRINT AT instructions.



***Figure 19.1:** Here the CONST program displays the constellation Cassiopeia. If you do not name it correctly after two tries, the name is displayed and you are asked if you would like another quiz.*

Because the ON X GOTO instruction of other BASICs is not available for the T/S 1000, a string has been used to generate the line numbers for random jumps to different constellations. This complicates the renumbering of the program. Do not attempt to renumber after you have keyed in the program and made any changes you might specifically need, unless you rewrite the GOTO string after renumbering.

The listing of the CONST program follows.

CONST

```
 05 REM CONST 4/16/83
 08 DIM E$(30)
 10 DIM D$(60)
 15 LET FL=0
 20 PRINT
 23 PRINT
 24 LET E$="060810121416182022242628303234"
 25 LET D$=E$+"363840424446485052545658606264"
 30 PRINT "AN ASTRONOMY PROGRAM"
 40 PRINT TAB 5;"TUTORIAL"
 55 PRINT
 60 PRINT "------------------"
 70 PRINT "I CONSTELLATIONS I"
 80 PRINT "------------------"
 90 PRINT
 92 PRINT
100 PRINT "BY ERIC BURGESS F.R.A.S."
110 PRINT
120 PRINT "ALL RIGHTS RESERVED BY"
130 PRINT "S AND T SOFTWARE SERVICES"
140 PAUSE 100
150 PRINT
160 PRINT
170 PRINT
180 PRINT "WANT INSTRUCTIONS Y/N?"
185 INPUT A$
190 IF A$="N" THEN GOTO 370
195 IF A$="Y" THEN GOTO 210
201 PRINT
202 GOTO 180
210 CLS
212 PRINT
220 PRINT "THIS PROGRAM DISPLAYS"
225 PRINT "A CONSTELLATION SELECTED"
230 PRINT "AT RANDOM AND ASKS FOR ITS NAME"
235 PRINT
240 PRINT "YOU HAVE TWO TRIES BEFORE"
250 PRINT "THE NAME IS REVEALED"
260 PRINT
270 PRINT "NEXT THE PROGRAM ASKS FOR"
275 PRINT "THE NAME OF A BRIGHT STAR"
280 PRINT "IN THE CONSTELLATION"
285 PRINT
```

CONST (continued)

```
290 PRINT "AGAIN YOU HAVE TWO CHANCES"
295 PRINT "TO NAME IT BEFORE BEING TOLD"
310 PRINT
320 PRINT "PICK FROM TWO ALTERNATIVES"
330 PRINT "CONSTELLATIONS THEN THE STARS"
340 PRINT "OR STARS AND CONSTELLATIONS"
345 PRINT TAB 6;"MIXED"
355 PRINT
360 PRINT "PRESS ANY KEY WHEN READY"
365 IF INKEY$ ="" THEN GOTO 365
370 CLS
371 PRINT
372 PRINT
373 PRINT
374 PRINT
380 PRINT "SELECT CONSTELLATIONS    (1)"
385 PRINT TAB 8;"OR"
390 PRINT "STARS AND CONSTELLATIONS (2)"
395 PRINT
397 INPUT AL
420 LET AL=AL-1
430 LET C= INT ( RND *30)+1
452 CLS
455 GOTO (100* VAL (D$((2*C-1) TO 2*C)))
600 LET C$="PISCES"
620 PRINT AT 5,5;"."
630 PRINT AT 9,3;"."
640 PRINT AT 12,7;"."; TAB 11;"."
645 PRINT AT 13,19;"."; TAB 26;"."; TAB 30;"."
650 PRINT AT 14,0;"."; TAB 23;"."
670 PRINT AT 15,28;"."
680 PRINT AT 16,23;"."; TAB 26;"."
690 GOSUB 7450
720 GOTO 7810
800 LET C$="ARIES"
810 LET S$="HAMAL"
820 IF FL=1 THEN GOSUB 7640
830 PRINT AT 6,23;"."
840 PRINT AT 7,22;"."
850 PRINT AT 9,26;"."
860 PRINT AT 10,12;"."
870 PRINT AT 11,13;"."
880 PRINT AT 12,11;"."
890 IF FL=0 THEN GOTO 920
895 FOR K=1 TO 6
900 PRINT AT 15,25;"X"
905 PRINT AT 15,25;" "
908 NEXT K
920 PRINT AT 15,25;"X"
930 PRINT AT 16,9;"."; TAB 28;"*"
940 PRINT AT 17,5;"."; TAB 28;"."
945 IF FL=1 THEN GOTO 7660
950 GOSUB 7450
980 GOTO 820
1000 LET C$="TAURUS"
1010 LET S$="ALDEBARAN"
1020 IF FL=1 THEN GOSUB 7640
```


CONST (continued)

```
1050 PRINT AT 3,0;"*"
1060 PRINT AT 6,24;".::"
1070 PRINT AT 8,6;"."
1080 PRINT AT 9,0;"*"
1090 PRINT AT 10,13;"."
1100 PRINT AT 11,14;".."
1110 IF FL=0 THEN GOTO 1130
1115 FOR K=1 TO 6
1120 PRINT AT 12,11;"X"
1122 PRINT AT 12,11;" "
1125 NEXT K
1130 PRINT AT 12,11;"X"
1140 PRINT AT 14,25;"."; TAB 30;"."
1150 PRINT AT 16,29;"."
1160 PRINT AT 17,23;"."; TAB 30;"."
1163 IF FL=1 THEN GOTO 7660
1165 GOSUB 7450
1195 GOTO 1020
1200 LET C$="GEMINI"
1210 LET S$="CASTOR"
1220 IF FL=1 THEN GOSUB 7640
1240 PRINT AT 4,19;"."
1245 IF FL=0 THEN GOTO 1260
1250 FOR K=1 TO 6
1252 PRINT AT 5,6;"X"
1254 PRINT AT 5,6;" "
1256 NEXT K
1260 PRINT AT 5,6;"X"
1270 PRINT AT 7,14;"."
1280 PRINT AT 9,3;"X"
1290 PRINT AT 10,7;"."
1300 PRINT AT 11,22;"*"
1310 PRINT AT 12,4;"."
1320 PRINT AT 13,12;"."; TAB 29;"*  *"
1330 PRINT AT 14,17;"."; TAB 28;"."
1340 PRINT AT 16,12;"*"
1350 PRINT AT 17,26;"X"
1360 PRINT AT 19,24;"*"
1365 IF FL=1 THEN GOTO 7660
1370 GOSUB 7450
1395 GOTO 1220
1400 LET C$="CANCER"
1420 PRINT AT 2,15;"."
1430 PRINT AT 8,15;"."
1440 PRINT AT 9,19;"."
1450 PRINT AT 12,15;"."; TAB 26;"."
1460 PRINT AT 16,11;"."
1470 PRINT AT 18,25;"."
1490 GOSUB 7450
1520 GOTO 7810
1600 LET C$="URSA MAJOR"
1610 LET S$="DUBHE"
1615 IF FL=1 THEN GOSUB 7640
1620 PRINT AT 8,31;"."
1630 IF FL=0 THEN GOTO 1660
1635 FOR K=1 TO 6
1640 PRINT AT 10,25;"*"
```

CONST (continued)

```
1645 PRINT AT 10,25;" "
1650 NEXT K
1660 PRINT AT 10,25;"*"
1670 PRINT AT 11,8;"*"
1680 PRINT AT 12,12;"*"; TAB 17;"*"
1690 PRINT AT 13,3;"*"
1700 PRINT AT 14,25;"*"
1710 PRINT AT 15,18;"*"
1720 PRINT AT 17,18;"."
1730 PRINT AT 19,25;"."
1740 IF FL=1 THEN GOTO 7660
1750 GOSUB 7450
1770 GOTO 1615
1800 LET C$="LEO"
1810 LET S$="REGULUS"
1820 IF FL=1 THEN GOSUB 7640
1830 PRINT AT 4,26;"."
1840 PRINT AT 5,30;"*"
1850 PRINT AT 6,21;"."
1860 PRINT AT 8,6;"*"; TAB 9;"."; TAB 21;"*"
1870 PRINT AT 9,0;"."
1880 PRINT AT 11,23;"."
1890 PRINT AT 12,6;"*"
1900 PRINT AT 13,0;"*"
1905 IF FL=0 THEN GOTO 1920
1908 FOR K=1 TO 6
1910 PRINT AT 14,22;"X"
1912 PRINT AT 14,22;" "
1915 NEXT K
1920 PRINT AT 14,22;"X"
1930 PRINT AT 16,4;"."; TAB 17;"."; TAB 28;"."
1940 PRINT AT 17,10;"."
1950 PRINT AT 19,6;"."
1955 IF FL=1 THEN GOTO 7660
1960 GOSUB 7450
1990 GOTO 1820
2000 LET C$="VIRGO"
2010 LET S$="SPICA"
2020 IF FL=1 THEN GOSUB 7640
2030 PRINT AT 5,12;"*"
2040 PRINT AT 7,25;"."
2050 PRINT AT 8,26;"."; TAB 30;"."
2060 PRINT AT 10,14;"*"
2070 PRINT AT 11,30;"*"
2080 PRINT AT 12,6;"."
2090 PRINT AT 13,16;"*"; TAB 21;"*"
2100 PRINT AT 15,6;"."
2105 IF FL=0 THEN GOTO 2135
2110 FOR K=1 TO 6
2115 PRINT AT 19,4;"X"
2120 PRINT AT 19,4;" "
2130 NEXT K
2135 PRINT AT 19,4;"X"
2140 IF FL=1 THEN GOTO 7660
2150 GOSUB 7450
2160 GOTO 2020
```


CONST (continued)

```
2200 LET C$="URSA MINOR"
2210 LET S$="POLARIS"
2215 IF FL=1 THEN GOSUB 7640
2230 IF FL=0 THEN GOTO 2260
2240 FOR K=1 TO 6
2245 PRINT AT 7,22;"X"
2250 PRINT AT 7,22;" "
2255 NEXT K
2260 PRINT AT 7,22;"X"
2270 PRINT AT 9,19;"."
2280 PRINT AT 11,17;"."
2290 PRINT AT 14,14;"."; TAB 17;"*"
2300 PRINT AT 15,20;"."
2310 PRINT AT 16,18;"*"
2320 PRINT AT 17,15;"*"
2330 IF FL=1 THEN GOTO 7660
2340 GOSUB 7450
2350 GOTO 2215
2400 LET C$="LIBRA"
2420 PRINT AT 4,22;"."
2430 PRINT AT 9,18;"*"
2440 PRINT AT 11,7;"."
2450 PRINT AT 12,13;"."
2460 PRINT AT 13,8;"."
2470 PRINT AT 14,23;"*"
2480 PRINT AT 19,18;"."
2500 GOSUB 7450
2530 GOTO 7810
2600 LET C$="SCORPIO"
2610 LET S$="ANTARES"
2620 IF FL=1 THEN GOSUB 7640
2630 PRINT AT 2,26;".*"
2640 PRINT AT 3,29;"."
2650 PRINT AT 4,30;"*"
2660 PRINT AT 7,23;"*"; TAB 30;"*"
2665 IF FL=0 THEN GOTO 2680
2670 FOR K=1 TO 6
2672 PRINT AT 8,21;"X"
2676 NEXT K
2680 PRINT AT 8,21;"X"
2690 PRINT AT 9,20;"*"; TAB 30;"."
2700 PRINT AT 13,4;"*"; TAB 7;"**"; TAB 17;"*"; TAB 21;"."
2710 PRINT AT 14,20;"."
2720 PRINT AT 15,6;"*"
2730 PRINT AT 16,5;"*"; TAB 17;"*"
2740 PRINT AT 19,8;"*"; TAB 14;"*"
2750 IF FL=1 THEN GOTO 7660
2760 GOSUB 7450
2790 GOTO 2620
2800 LET C$="SAGITTARIUS"
2820 PRINT AT 4,9;"."
2830 PRINT AT 5,9;"."
2840 PRINT AT 7,12;"*"; TAB 15;"."
2850 PRINT AT 8,14;"."
2860 PRINT AT 10,15;"*"; TAB 25;"*"
2870 PRINT AT 11,17;"*"; TAB 27;"."
2880 PRINT AT 12,12;"*"; TAB 29;"."
```

CONST (continued)

```
2890 PRINT AT 13,13;"*"; TAB 26;"*"
2900 PRINT AT 14,29;"*"
2910 PRINT AT 17,24;"*"
2920 PRINT AT 19,25;"*"
2940 GOSUB 7450
2970 GOTO 7810
3000 LET C$="CAPRICORNUS"
3020 PRINT AT 4,30;"*"
3030 PRINT AT 6,29;"*"
3040 PRINT AT 8,4;"*  ."; TAB 12;"."; TAB 16;"."
3050 PRINT AT 12,7;"."
3060 PRINT AT 13,12;":"
3070 PRINT AT 14,17;"."; TAB 24;"."
3080 PRINT AT 15,15;"*"
3090 PRINT AT 19,16;"*"
3110 GOSUB 7450
3140 GOTO 7810
3200 LET C$="AQUARIUS"
3220 PRINT
3230 PRINT AT 7,4;"*  *"; TAB 12;"*"
3240 PRINT AT 8,8;"*"
3250 PRINT AT 10,21;"*"; TAB 29;"."
3260 PRINT AT 12,2;"."
3270 PRINT AT 13,30;"."
3280 PRINT AT 14,25;"."
3290 PRINT AT 16,3;"."
3300 PRINT AT 18,1;"*"
3330 GOSUB 7450
3360 GOTO 7810
3400 LET C$="ORION"
3410 LET S$="RIGEL"
3420 IF FL=1 THEN GOSUB 7640
3450 PRINT AT 3,7;"."; TAB 15;".:"; TAB 29;"."
3460 PRINT AT 4,30;"."
3470 PRINT AT 5,9;"X"; TAB 30;"."
3480 PRINT AT 6,18;"."; TAB 20;"*"; TAB 30;"."
3490 PRINT AT 8,15;"."
3495 PRINT AT 9,29;". ."
3500 PRINT AT 12,18;"*"
3510 PRINT AT 13,16;"*"
3515 PRINT AT 14,14;"*"; TAB 20;"."
3520 PRINT AT 16,15;"."
3530 PRINT AT 17,15;":"
3535 IF FL=0 THEN GOTO 3560
3540 FOR K=1 TO 6
3542 PRINT AT 18,23;"X"
3545 PRINT AT 18,23;" "
3548 NEXT K
3560 PRINT AT 18,23;"X"
3570 PRINT AT 19,11;"*"
3575 IF FL=1 THEN GOTO 7660
3580 GOSUB 7450
3595 GOTO 3420
3600 GOTO 430
3800 LET C$="CANIS MAJOR"
3810 LET S$="SIRIUS"
3820 IF FL=1 THEN GOSUB 7640
```


CONST (continued)

```
3830 PRINT AT 5,17;"."
3832 IF FL=0 THEN GOTO 3845
3834 FOR K=1 TO 6
3836 PRINT AT 6,21;"X"
3838 PRINT AT 6,21;" "
3840 NEXT K
3845 PRINT AT 6,21;"X"
3850 PRINT AT 8,23;"."; TAB 27;"*"
3860 PRINT AT 10,16;"*"
3870 PRINT AT 11,12;"."; TAB 18;"."
3880 PRINT AT 12,14;"*"
3890 PRINT AT 13,9;"*"
3900 PRINT AT 14,15;"*"
3910 PRINT AT 16,15;"."; TAB 24;"*"
3920 PRINT AT 18,20;"."
3930 IF FL=1 THEN GOTO 7660
3960 GOSUB 7450
3990 GOTO 3820
4000 LET C$="CANIS MINOR"
4010 LET S$="PROCYON"
4020 IF FL=1 THEN GOSUB 7640
4030 PRINT AT 8,21;"."
4040 PRINT AT 9,22;"*"
4045 IF FL=0 THEN GOTO 4060
4047 FOR K=1 TO 6
4049 PRINT AT 12,16;"X"
4052 PRINT AT 12,16;" "
4055 NEXT K
4060 PRINT AT 12,16;"X"
4070 PRINT AT 15,6;"."
4080 IF FL=1 THEN GOTO 7660
4110 GOSUB 7450
4140 GOTO 4020
4200 GOTO 430
4400 LET C$="AQUILA"
4410 LET S$="ALTAIR"
4420 IF FL=1 THEN GOSUB 7640
4430 PRINT AT 4,27;"."
4440 PRINT AT 5,25;"*"
4450 PRINT AT 7,16;"*"
4460 IF FL=0 THEN GOTO 4500
4470 FOR K=1 TO 6
4480 PRINT AT 8,15;"X"
4490 PRINT AT 8,15;" "
4495 NEXT K
4500 PRINT AT 8,15;"X"
4510 PRINT AT 9,20;"."
4520 PRINT AT 10,14;"*"
4530 PRINT AT 14,14;"."; TAB 22;"*"
4540 PRINT AT 15,9;"*"
4550 PRINT AT 19,27;"*"
4560 PRINT AT 20,29;"*"
4570 IF FL=1 THEN GOTO 7660
4580 GOSUB 7450
4595 GOTO 4420
4600 LET C$="LYRA"
4610 LET S$="VEGA"
```

CONST (continued)

```
4620 IF FL=1 THEN GOSUB 7640
4630 PRINT AT 5,15;"."
4640 PRINT AT 9,19;"."
4650 PRINT AT 10,8;"."
4660 IF FL=0 THEN GOTO 4700
4670 FOR K=1 TO 6
4680 PRINT AT 10,21;"X"
4685 PRINT AT 10,21;" "
4690 NEXT K
4700 PRINT AT 10,21;"X"
4710 PRINT AT 11,7;"."; TAB 18;"."
4720 PRINT AT 12,15;"."; TAB 27;"."
4730 PRINT AT 16,17;"*"
4740 PRINT AT 17,14;"*"
4750 IF FL=1 THEN GOTO 7660
4760 GOSUB 7450
4790 GOTO 4620
4800 LET C$="CYGNUS"
4810 LET S$="DENEB"
4820 IF FL=1 THEN GOSUB 7640
4830 PRINT AT 2,30;"."
4840 PRINT AT 3,28;"."
4850 PRINT AT 6,20;"."
4860 PRINT AT 7,20;"."
4865 PRINT AT 8,6;"."
4870 IF FL=0 THEN GOTO 4900
4880 FOR K=1 TO 6
4885 PRINT AT 8,15;"X"
4890 PRINT AT 8,15;" "
4895 NEXT K
4900 PRINT AT 8,15;"X"
4910 PRINT AT 8,26;"*"
4920 PRINT AT 9,10;"."
4930 PRINT AT 10,11;"."
4940 PRINT AT 11,19;"*"
4950 PRINT AT 12,7;"."
4955 PRINT AT 14,24;"."
4960 PRINT AT 15,14;"*"; TAB 20;"."
4965 PRINT AT 19,30;"*"
4970 IF FL=1 THEN GOTO 7660
4975 GOSUB 7450
4990 GOTO 4820
5000 LET C$="HERCULES"
5020 PRINT AT 5,18;"*"; TAB 29;". ."
5030 PRINT AT 6,30;"."
5040 PRINT AT 7,28;"."
5050 PRINT AT 9,26;"*"
5060 PRINT AT 10,13;"*"; TAB 20;"*"
5070 PRINT AT 13,26;"*"
5080 PRINT AT 14,11;"."; TAB 22;"*"
5090 PRINT AT 15,9;". ."
5100 PRINT AT 16,13;"*"
5110 PRINT AT 17,19;"*"
5120 PRINT AT 19,4;"."; TAB 30;"*"
5130 GOSUB 7450
5160 GOTO 7810
5200 LET C$="CEPHEUS"
```


CONST (continued)

```
5210 PRINT AT 5,18;"*"
5220 PRINT AT 11,24;"*"
5230 PRINT AT 13,16;"*"
5240 PRINT AT 17,25;"*"
5250 PRINT AT 18,30;"*"
5260 PRINT AT 19,17;"*"
5270 GOSUB 7450
5280 GOTO 7810
5400 LET C$="CASSIOPEIA"
5410 PRINT AT 6,15;"."
5420 PRINT AT 7,14;"."
5430 PRINT AT 12,11;"*"
5440 PRINT AT 16,20;"*"; TAB 30;"*"
5450 PRINT AT 17,15;"*"
5460 PRINT AT 18,22;"."
5470 PRINT AT 19,24;"*"
5480 GOSUB 7450
5490 GOTO 7810
5600 LET C$="PERSEUS"
5610 LET S$="ALGOL"
5615 IF FL=1 THEN GOSUB 7640
5620 PRINT AT 5,23;"."
5630 PRINT AT 7,21;"*  ."
5640 PRINT AT 10,6;"."; TAB 18;"X"; TAB 21;"."; TAB 25;"."
5650 PRINT AT 11,6;"."
5660 PRINT AT 12,8;"."; TAB 13;"*"
5670 PRINT AT 13,22;"."
5680 PRINT AT 16,14;"."
5690 IF FL=0 THEN GOTO 5720
5695 FOR K=1 TO 6
5700 PRINT AT 17,23;"*"
5710 PRINT AT 17,23;" "
5715 NEXT K
5720 PRINT AT 17,23;"*"
5730 PRINT AT 18,11;"*"
5740 PRINT AT 19,24;"."
5750 IF FL=1 THEN GOTO 7660
5760 GOSUB 7450
5770 GOTO 5615
5800 LET C$="ANDROMEDA"
5810 PRINT AT 5,14;"."
5820 PRINT AT 6,15;"."
5830 PRINT AT 10,9;"*"
5840 PRINT AT 11,14;"."; TAB 24;"."
5850 PRINT AT 12,23;"."
5860 PRINT AT 14,20;"*"; TAB 27;"."
5870 PRINT AT 15,26;"."
5880 PRINT AT 16,31;"*"
5890 PRINT AT 17,26;"."
5900 GOSUB 7450
5910 GOTO 7810
6000 LET C$="PEGASUS"
6010 PRINT AT 6,22;"."
6020 PRINT AT 7,19;"*"
6030 PRINT AT 9,21;"."
6040 PRINT AT 10,22;"."
```

CONST (continued)

```
6050 PRINT AT 13,8;"*"; TAB 19;"*"
6060 PRINT AT 16,23;"*"
6070 PRINT AT 18,28;"*"
6080 GOSUB 7450
6090 GOTO 7810
6200 LET C$="BOOTES"
6210 LET S$="ARCTURUS"
6220 IF FL=1 THEN GOSUB 7640
6230 PRINT AT 3,21;". ."
6240 PRINT AT 5,22;"."
6250 PRINT AT 7,17;"*"
6260 PRINT AT 8,20;"*"
6270 PRINT AT 10,14;"*"
6280 PRINT AT 12,21;"."
6290 PRINT AT 14,19;"*"
6300 IF FL=0 THEN GOTO 6330
6305 FOR K=1 TO 6
6310 PRINT AT 17,25;"X"
6320 PRINT AT 17,25;" "
6325 NEXT K
6330 PRINT AT 17,25;"X"
6340 IF FL=1 THEN GOTO 7660
6350 GOSUB 7450
6360 GOTO 6220
6400 LET C$="AURIGA"
6410 LET S$="CAPELLA"
6415 IF FL=1 THEN GOSUB 7640
6420 PRINT AT 4,12;"."
6430 IF FL=0 THEN GOTO 6470
6440 FOR K=1 TO 6
6445 PRINT AT 9,18;"X"
6450 PRINT AT 9,18;" "
6460 NEXT K
6470 PRINT AT 9,18;"X"
6480 PRINT AT 10,10;"*"
6490 PRINT AT 11,21;"."
6500 PRINT AT 14,20;".."
6510 PRINT AT 14,12;".."
6520 PRINT AT 15,10;"*"
6530 PRINT AT 19,22;"*"
6540 IF FL=1 THEN GOTO 7660
6550 GOSUB 7450
6560 GOTO 6415
7450 REM QUERY CONST
7460 PRINT AT 20,0;"NAME THIS CONSTELLATION"
7470 INPUT A$
7480 IF A$=C$ THEN GOTO 7560
7490 PRINT AT 20,0;"INCORRECT, TRY AGAIN    "
7495 LET A$=""
7500 INPUT A$
7510 IF A$=C$ THEN GOTO 7560
7520 PRINT AT 20,0;"    YOU ARE STILL WRONG"
7525 PAUSE 50
7530 PRINT AT 1,0;"IT IS ";C$
7540 LET FL=1
7550 PRINT AT 20,0;"                        "
```


CONST (continued)

```
7555 PRINT AT 21,0;"                        "
7556 PRINT AT 21,0;"PRESS ANY KEY           "
7557 IF INKEY$ ="" THEN GOTO 7557
7558 PRINT AT 21,0;"                  "
7559 GOTO 7610
7560 PRINT AT 20,0;"                        "
7565 FOR K=1 TO 6
7570 PRINT AT 1,2;"CORRECT"
7580 PAUSE 5
7590 PRINT AT 1,2;"       "
7600 NEXT K
7605 LET FL=1
7607 PRINT AT 20,0;"                        "
7610 RETURN
7640 REM  STAR QUERY
7650 PRINT AT 1,0;"NAME THIS STAR IN ";C$
7652 PAUSE 50
7655 RETURN
7660 LET FL=0
7665 INPUT A$
7670 IF A$=S$ THEN GOTO 7740
7680 PRINT AT 2,0;"INCORRECT, TRY AGAIN"
7685 INPUT A$
7686 PRINT AT 2,0;"                         "
7690 IF A$=S$ THEN GOTO 7740
7700 PRINT AT 2,0;" YOU ARE STILL WRONG"
7710 PRINT AT 3,0;"THE STAR IS ";S$
7730 GOTO 7810
7740 PRINT AT 2,0;S$;" IS"
7750 PRINT AT 3,2;"CORRECT"
7810 PRINT AT 20,0;"PRESS ANY KEY TO CONTINUE"
7812 IF INKEY$ ="" THEN GOTO 7812
7814 CLS
7820 PRINT AT 4,0;"ANOTHER TEST Y/N? "
7830 INPUT B$
7840 IF B$="N" THEN GOTO 7920
7842 IF AL=0 AND FL=1 THEN LET FL=0
7850 LET AL= ABS (AL-1)
7910 GOTO 430
7920 CLS
7930 STOP
```

Program 20: PHOTO

Photographing the Planets

Three common methods of planetary photography are as follows:

1. Placing the photographic film at the prime focus of the telescope's objective
2. Attaching a camera (with lens) to the eyepiece of the telescope
3. Attaching a camera (without lens) to the eyepiece of the telescope (projection photography)

This program allows you to calculate exposures for the various planets for each of these three methods of photography. You have to input the diameter (in inches) of the telescope's objective (clear aperture), the focal length (in inches) of the objective, the focal length (in mm) of the eyepiece used, the focal length (in mm) of the camera lens used, and the distance (in inches) from the eyepiece to the film plane if using projection photography. Also, you must give the name of the planet and its diameter in arc seconds at the date you are making the photograph. You can determine planetary diameters in arc seconds from distance data. You must divide the diameter at unit distance of one astronomical unit by the distance of the planet in astronomical units (given by Program 11). Diameters of the planets at unit distance are: Mercury, 6.68; Venus, 16.82; Mars, 9.36; Jupiter, 190; Saturn, 158; Uranus, 68; and Neptune, 73. Pluto is too small to be of significance. In addition, you must select an appropriate film and provide the ASA rating.

The program provides a guide to the exposure you will need, the resolution to be expected, and the size of the image on the film plane. With this information you can effectively plan a photographic session in advance and thus avoid many disappointments. A typical display is shown in Figure 20.1.

The listing for the PHOTO program follows.



Figure 20.1: To help you obtain better planetary photographs, PHOTO develops displays such as this for three types of camera/telescope arrangements.

PHOTO

```
10 REM PHOTO 4/16/83
20 PRINT AT 7,6;"------------------"
30 PRINT AT 8,6;"I      PHOTO     I"
40 PRINT AT 9,6;"------------------"
50 PRINT AT 11,5;"AN ASTRONOMY PROGRAM"
60 PRINT AT 12,4;"BY ERIC BURGESS F.R.A.S."
70 PRINT AT 14,5;"ALL RIGHTS RESERVED BY"
80 PRINT AT 15,4;"S AND T SOFTWARE SERVICE"
90 PRINT
100 PRINT AT 18,4;"WANT INTRUCTIONS Y/N? ";
110 INPUT A$
120 PRINT A$
130 IF A$="Y" THEN GOTO 200
```

PHOTO (continued)

```
140 CLS
150 PRINT
160 PRINT
170 IF A$="N" THEN GOTO 450
180 PRINT "INVALID RESPONSE"
190 GOTO 100
200 CLS
210 PRINT
220 PRINT
230 PRINT "THIS PROGRAM ASK PARTICULARS OF"
240 PRINT "YOUR ASTROPHOTO INSTALLATION AND"
250 PRINT "THEN COMPUTES THE SIZE OF THE"
260 PRINT "IMAGE OF ANY SELECTED PLANET ON"
270 PRINT "THE FILM PLANE"
280 PRINT
290 PRINT
300 PRINT "FOR ASA RATING WHICH YOU INPUT"
310 PRINT "IT CALCULATES AN AVERAGE"
320 PRINT "EXPOSURE FOR THE PLANET AND THE"
330 PRINT "INSTRUMENT CONFIGURATION YOU ARE USING"
340 PRINT
350 PRINT
360 PRINT
370 PRINT "PRESS ANY KEY TO CONTINUE"
380 IF INKEY$ ="" THEN GOTO 380
390 CLS
400 PRINT
410 PRINT
420 PRINT "WHICH METHOD ARE YOU USING?"
430 PRINT
440 PRINT
450 PRINT "1) FILM AT PRIME FOCUS OF"
460 PRINT "   OBJECTIVE OF TELESCOPE"
470 PRINT
480 PRINT "2) FILM IN CAMERA AND CAMERA"
490 PRINT "   AT EYEPIECE OF TELESCOPE"
500 PRINT "   WITH CAMERA LENS IN USE"
510 PRINT
520 PRINT "3) IMAGE PROJECTED FROM EYEPIECE"
530 PRINT "   OF TELESCOPE INTO CAMERA"
540 PRINT "   WITHOUT A CAMERA LENS"
550 PRINT
560 PRINT
570 PRINT "SELECT 1,2 OR 3 ";
580 INPUT A
590 PRINT A
600 PAUSE 150
610 CLS
620 PRINT
630 PRINT "GIVE FOLLOWING DATA"
640 PRINT
650 PRINT "CLEAR APERTURE OF OBJECTIVE (IN) ";
660 INPUT AO
670 PRINT AO
680 LET AO=AO*25.4
690 PRINT
```


PHOTO (continued)

```
700 PRINT "FOCAL LENGTH OF OBJECTIVE (IN)"
710 INPUT FO
720 PRINT FO
730 LET FO=FO*25.4
740 LET NF=FO/AO
750 IF A <> 2 THEN GOTO 800
760 PRINT
770 PRINT "FOCAL LENGTH OF CAMERA LENS (MM) ";
780 INPUT FC
790 PRINT FC
800 IF A=1 THEN GOTO 930
810 PRINT
820 PRINT "FOCAL LENGTH OF EYEPIECE (MM) ";
830 INPUT FE
840 PRINT FE
850 IF A=1 THEN GOTO 930
860 IF A=2 THEN GOTO 930
870 PRINT
880 PRINT "DISTANCE EYEPIECE TO FILM PLANE (IN) ";
890 INPUT DE
900 PRINT DE
910 LET DE=DE*25.4
920 REM CALC. EFFECTIVE F NUMBER
930 IF A=1 THEN LET EFL=NF
940 IF A=2 THEN LET EFL=NF*FC/FE
950 IF A=3 THEN LET EFL=NF*DE/FE
 960 CLS
 970 PRINT
 980 PRINT
 990 PRINT "YOU MUST NOW ENTER THE NAME OF"
1000 PRINT "THE PLANET YOU ARE PHOTOGRAPHING"
1010 PRINT "AND ITS DIAMETER IN ARCSECS"
1020 PRINT
1030 PRINT "(YOU CAN FIND THIS FROM ONE OF"
1040 PRINT "THE PROGRAMS IN THIS SERIES)"
1050 PRINT
1060 PRINT
1070 PRINT "WHAT IS THE PLANETS NAME ";
1080 INPUT P$
1090 PRINT P$
1100 IF P$ <> "PLUTO" THEN GOTO 1130
1110 PRINT "NO DATA ON PLUTO; PICK ANOTHER PLANET"
1120 GOTO 1070
1130 PRINT
1140 PRINT "WHAT IS THE DIAMETER IN ARCSECS? ";
1150 INPUT DP
1160 PRINT DP
1170 LET F=EFL*AO
1180 CLS
1190 PRINT
1200 PRINT
1210 PRINT
1220 LET DP=DP/3600
1230 LET DP= TAN (DP/57.29578)
1240 LET I=F*DP
1250 LET R=1450/EFL
1260 PRINT
```

PHOTO (continued)

```
1270 PRINT
1280 PRINT "RESOLUTION FOR ";P$;" AT"
1290 PRINT "FILM PLANE WILL BE "; INT (R);" LINES/MM"
1300 PRINT
1310 PRINT
1320 PRINT "YOU ARE ADVISED TO PICK A FILM"
1330 PRINT "WITH AT LEAST THREE TIMES"
1340 PRINT "THIS RESOLUTION"
1350 LET R=3*R
1360 PRINT
1370 PRINT
1380 PRINT "NAMELY "; INT (R);" LINES/MM TO"
1390 PRINT "AVOID GRAIN SPOILING THE IMAGE"
1400 PRINT
1410 PRINT
1420 PRINT "INPUT ASA RATING OF CHOSEN FILM? ";
1430 INPUT ASA
1440 PRINT ASA
1450 CLS
1460 PRINT
1470 PRINT
1480 IF P$="MERCURY" THEN LET N=260
1490 IF P$="VENUS" THEN LET N=960
1500 IF P$="MARS" THEN LET N=40
1510 IF P$="JUPITER" THEN LET N=12
1520 IF P$="SATURN" THEN LET N=3.5
1530 IF P$="URANUS" THEN LET N=.81
1540 IF P$="NEPTUNE" THEN LET N=.38
1550 LET N=N*ASA
1560 LET EX=EFL*EFL/N
1570 CLS
1580 PRINT
1590 PRINT "TO PHOTOGRAPH ";P$;" ON FILM "
1600 PRINT "RATED AT ";ASA;" ASA"
1610 PRINT
1620 IF A=1 THEN LET Q$= "FILM AT PRIME FOCUS OF OBJECTIVE"
1630 IF A=2 THEN LET Q$= "CAMERA EYEPIECE COMBINATION"
1640 IF A=3 THEN LET Q$= "PROJECTION FROM EYEPIECE"
1650 PRINT "WITH ";Q$
1660 PRINT
1670 LET X$= STR$ (EX)
1680 IF EX<.01 THEN GOTO 1710
1690 PRINT "EXPOSURE IS ABOUT ";X$(1 TO 3);" SEC"
1700 GOTO 1720
1710 PRINT "EXPOSURE IS ABOUT ";X$(1 TO 4);" SEC"
1720 PRINT
1730 PRINT "  (NOTE EXPOSURE GREATER THAN"
1740 PRINT "   1 SEC NOT RECOMMENDED)"
1750 PRINT
1760 PRINT "AND THE IMAGE ON THE FILM PLANE"
1770 PRINT "WILL BE ";
1780 LET I$= STR$ (I)
1790 PRINT I$(1 TO 3);" MM IN DIAM."
1800 PRINT
1810 PRINT "  (NOTE: IMAGE ON THE FILM PLANE"
1820 PRINT "SHOULD EXCEED 3 MM TO BE USEFUL)"
1830 PRINT
```


PHOTO (continued)

```
1840 PRINT "WANT COPY Y/N? ";
1850 INPUT Z$
1860 PRINT Z$
1870 IF Z$="N" THEN GOTO 1890
1880 COPY
1890 PRINT
1900 PRINT "WANT ANOTHER CALCULATION? Y/N? ";
1910 INPUT Y$
1920 PRINT Y$
1930 CLS
1940 IF Y$="N" THEN GOTO 2000
1950 IF Y$="Y" THEN GOTO 400
1960 PRINT "INVALID RESPONSE"
1970 PAUSE 150
1980 GOTO 1900
1990 CLS
2000 STOP
```

Bibliography

The following books are recommended for further information about astronomical calculations. Some of them provide equations that can be used to obtain much greater precision of almanac data than that provided by the programs in this book. While the programs given here are sufficient for most practical purposes, more specialized routines are sometimes needed for predictions of stellar occultations by the Moon, close conjunctions of planets, or rigorous observations of celestial objects relative to star positions. If you need such precision in computations, many of the programs provided in this book can be upgraded within the limits of your computer's capability to handle numbers.

The American Ephemeris and Nautical Almanac. U.S. Government Printing Office: Washington D.C., annually to 1980.

The Astronomical Ephemeris. H. M. Nautical Almanac Office: Royal Greenwich Observatory, H. M. Stationery Office: London, annually to 1980. (British edition of the above.)

The Astronomical Almanac. U.S. Government Printing Office: Washington, D.C., and H. M. Stationery Office: London, annually from 1981. (Combines *The American Ephemeris* and *The Astronomical Ephemeris.*)

Explanatory Supplement to the Nautical Almanac. H. M. Stationery Office: London, 1977.

Astrodynamics. R. M. L. Baker and M. W. Makemson. Academic Press: New York, 1960.

Celestial BASIC. Eric Burgess. Sybex: Berkeley, Calif., 1982.

Mathematical Astronomy with a Pocket Calculator. Aubrey Jones. Wiley: New York, 1978.

Positional Astronomy. D. McNally. Muller: London, 1974.

Practical Astronomy With Your Calculator. Second Edition. P. Duffett-Smith. Cambridge University Press: London, 1981.

Index

MORE USES FOR YOUR

TIMEX/SINCLAIR 1000: ASTRONOMY

ON YOUR COMPUTER

Let your Timex/Sinclair 1000 take you to the stars!

Explore the heavens on your computer—no more hunting for your almanac, digging up that elusive astronomical magazine, or spending hours using a pocket calculator for laborious calculations. For the serious astronomer or the occasional stargazer, this book provides ready-to-run programs that will:

- convert time from one system of measurement to another
- calculate and display the positions of the Sun, Moon, planets, and stars for any date or location
- teach you to recognize the constellations

and more.

Use your Timex/Sinclair 1000 to have fun while learning about the stars and the solar system, with this exciting new book.

ABOUT THE AUTHORS:

Eric Burgess's byline has appeared internationally in newspapers and magazines for over 40 years. He has written extensively for NASA, providing educational materials and books on the missions and their results. He has recently become an enthusiastic participant in the personal computer revolution. He is also the author of the SYBEX publication, *Celestial BASIC.*

Howard J. Burgess is an industrial photographer at a high-technology microwave company. He has done extensive work in the area of integrating computers into our everyday lives.

ISBN 0-89588-112-8